
はじめに
パソコンが登場してから今日まで，計算機とそれを取り巻く環境の劇的な進歩には目を見張るばかり
です．計算機の利用目的も多様になり，それぞれの目的に応じて様々なプログラミング言語が開発され
ました．しかし，数値計算やコンピュータシミュレーションにおいては，コンピュータ本来の能力であ
る「高速計算」ができればいいので，それほど新しい言語は必要ありません．
現在，数値計算をするためによく使われているプログラミング言語には，C言語と Fortranがありま
す．最近の大学では C言語で計算機の講義を教えるところが増えてきて，Fortranは影が薄くなってい
るようですが，コンピュータシミュレーションの分野では Fortranの方がよく使われています．これは，
Fortranが科学技術計算用言語として発展してきているので，数値計算に便利な書式が多数採用されて
いるからです．例えば，複素数計算や行列計算などは文法に組み込まれているので簡単に書くことがで
きます．数値計算向けのプログラムを書くには Fortranの方が書きやすいし，完成したプログラムをそ
のまま大型計算機で高速に実行させることも可能です．

Fortranは，計算機の歴史の初期に開発された由緒ある言語で，“FORmula TRANslation ”が語源で
す．当初から数値計算用言語として開発されており，筆者が計算機を利用し始めた数十年前には，数値
計算をするといえば，ほとんどが Fortranでした．しかし，科学技術計算に特化するということはユー
ザーの専門分野が限られるということであり，その結果コンパイラの値段があまり下がらなかったのが
Fortranをマイナーにした最大の要因だと思います．C言語は誕生時点から比較的安価に供給され，こ
れが広く使われている理由の一つです．C言語はそもそもOSを記述するために開発された言語なので，
計算機のハードウェア寄りの書式が多く，必ずしも初心者向きの言語だとは思えないのですが，プログ
ラミングの専門家にすれば FortranもC言語も大して変わりません．安い方にシフトしたのはある意味
やむを得ないことだと思います．
さて，1990年代に入って無料のOSである Linuxや FreeBSDの開発が開始され，これが非常な勢い
で普及しました．普及した最大の要因は，インターネットを通じて，世界中のアマチュアプログラマが
OSやその上で動作するアプリケーションの開発に参加したことです．コンパイラも同様で，このフリー
OSには，開発環境として有志が作成した無料の C言語 (gcc)が付属し，さらにこれを利用した無料の
Fortran(g77，gfortranなど)も付属しています．このため，Linuxなどをインストールすれば，パソコ
ンでも無料で Fortranが使えます．また，これらフリーの FortranはWindowsやMacOS上で動作する
ものも作られていているので，価格で Fortranを敬遠する理由はなくなったと言えます．
本書は，数値計算やコンピュータシミュレーションを目的として，これから計算機プログラムの勉強
を始めようとする人を対象にした Fortranプログラム作成法の解説書です．単に文法だけを説明するの
ではなく，コンピュータの構造の話なども交えながら，効率が良くてエラーを見つけやすい，“スマー
トな Fortranプログラム ” の書き方を説明します．
長い歴史を持っているFortranは，古いバージョンとの互換性を保つ必要性から，若干緩い文法体系に
なっています．例えば，デフォルトでは変数宣言をしなくてもかまわないため，変数名を書き間違えた
ときに発見するのが難しく，思わぬエラーを引き起こす可能性があります．また，コンピュータシミュ
レーションのような長いプログラムを書くときには，プログラムのチェックや修正がしやすいように書
いておかないと，メンテナンスに苦労します．このため，多少冗長になっても，読みやすく，後でデバッ
グしやすいように書くべきです．本書では，これを念頭に置いて説明をしています．
もちろん，数値計算やコンピュータシミュレーションにとっては計算速度が命ですから，できる限り
計算が速くなるようなプログラムにすることも必要です．プログラムとはコンピュータでの計算手順を
記述するものなので，コンピュータの構造を考慮してちょっと計算順序を変えるだけでも高速化できる
ことがあります．本書では，このようなコンピュータの内部構造で知っておいた方が良い部分もいくつ
か説明しました．孫子曰く，「彼を知り己を知れば百戦殆うからず」と．

Fortranは古くからある言語なのですが，1991年に策定された Fortran90を境に大きく変貌しまし



た [1]．配列演算などの新しい機能を導入すると同時に，現代的な書式で書けるようになったのです．そ
れまで 1行に 72文字までしか書けなかったという制限は無くなったし，1行に複数の文を書くこともで
きるようになりました．Fortran90以降も何度か改版されて，現在の最新規格はFortran2013であり，オ
ブジェクト指向プログラミングや並列処理の記述も可能になっています．しかし，パソコンを利用した
数値計算にはそのような新しい概念は必要ないので，本書では Fortran90の改訂版である Fortran95レ
ベルの文法で説明をしました．このレベルでプログラムを書いておけばフリーの Fortranでコンパイル
できるというメリットもあります．
本書には 6章ありますが，基本的なプログラミングに関する知識は第 4章までで十分です．そこで，第

4章までには演習問題を付けています．ただし，第 3章までの演習問題は各章の内容に合わせた問題に
なっていますが，第 4章の演習問題はそれまでのプログラミング知識全般を応用して実用的なプログラ
ムを書くための練習になるような問題になっています．また，第 3章までの章末に「スマートテクニッ
ク」と称して，プログラムを書くときの心構え，高速化のこつ，知っておくと便利な文法について紹介
しています．これらも合わせて利用すれば，より質の高いプログラムを書くことができると思います．
これに対し，第 5章と第 6章はある程度プログラミングに慣れてきた人向けです．初心者の方は，ま
ず第 4章までの内容を使って色々なプログラムを書き，プログラミングというものを良く理解して下さ
い．第 5章以降は，ある程度の経験を積んでから必要に応じて読まれれば良いと思います．また，実際
にパソコンで Fortranプログラムを動作させる方法や，その際の注意事項などをまとめて付録にしてい
ます．こちらも必要に応じて活用して下さい．
本書は，筆者が摂南大学の卒業研究生向けに書いた Fortranのテキストがベースになっています．こ
れを，大阪大学レーザーエネルギー学研究センターの福田優子氏がWebで公開できる Fortranテキス
トを探しているというので提供したところ，結構気に入ってくれて，学生に配ったり，他の計算機セン
ターに紹介してくれました．その後，紹介先の一つである東北大学サイバーサイエンスセンターから広
報誌 SENACへテキストの内容を投稿してくれないかとの依頼があったので，加筆・修正をして 2012年
度に 3回の連載として掲載されました．ただ，3回の連載ではページ数が限られていたので演習問題を
カットせざるを得ず，卒研生を教育するために配る資料としてはもの足りないものになってしまいまし
た．そこで，SENACの原稿を再編成して説明が不十分の箇所を補い，演習問題を加えてまとめ直した
のが本書です．
その後，阪大のWebで公開していたテキストを見られた技術評論社の方から，「Fortranの解説とそれ
を利用した数値計算の本として出版しないか」という提案をいただきました．これまで勉強してきたこと
を活かす良い機会だと思って引き受けてから 2年近くかかりましたが，なんとか 2015年 7月に “Fortran
ハンドブック ”という題で出版されました [7]．本書の 2015年度版は，本を出版するために行った確認
チェックにより気がついた問題点を踏まえて改訂したものです．
最後に，筆者が適当に書いたテキストをこのようなまともな文書にまとめ直すきっかけを与えてくれ
た福田優子氏に深い感謝の意を表します．また，連載するに当たって，つたない文章を熱心に読んで多
数の有益なご意見を下さった東北大学サイバーメディアセンターの方々にも深く感謝致します．

第1章 Fortranプログラミングの基本
計算機プログラム (あるいは，単にプログラム)とはコンピュータへの動作指示 (命令)を記述した文の
集まりのことです．コンピュータを動作させる本来の命令は “機械語 ”と呼ばれる数値で表されていま
すが，これで直接プログラムを書くことはまず不可能です．そこで，人間が理解しやすいように，単語
や記号を使った書式で計算機の動作を書けるようにしたものが，Fortranのような “プログラミング言
語 ”です．プログラミング言語を使ってプログラムを書いておけば，これを “コンパイラ”という機械語
への翻訳ソフトを使って，コンピュータで実行可能な機械語 (実行形式)に変換することができます．
しかし，コンパイラは翻訳機に過ぎません．プログラミング言語という，我々には理解しやすい書式
で書けるといっても，あくまでもコンピュータを動作させる手順になるように書かねばならないのです．
本章では，Fortranにおける基本的なプログラムの書式とデータの取り扱いについて説明します．

Fortranプログラムは，基本的にコンピュータに与える命令を次の形をした “文 ”で記述します．

　　動作指示語　動作制御パラメータ

すなわち，先頭に “動作指示語 ”と呼ばれる計算機の動作を指定する単語を書き，それに続けて，動作
を制御するための数値や予約語を “動作制御パラメータ ”として記述します．ただし動作制御パラメー
タを記述する必要のない，動作指示語だけの命令も存在します．
実行形式に変換されたプログラムを起動すると，コンピュータは，

　　実行開始処理　→　プログラムに記述された命令を順に実行　→　実行終了処理

という流れで動作します．実行を開始した時に最初に実行する部分を “メインプログラム”，または “メ
インルーチン ”といいます．言わばプログラムの本体です．本章と次章では，メインプログラムだけを
使った Fortranの基本的プログラムの書き方について説明します．メインプログラムと同レベルの完結
したプログラムには，他のプログラムの中から実行開始を指示してその機能を利用する “サブルーチン”
がありますが，これについては第 3章で説明します．

1.1 メインプログラムの開始と終了
Fortranでは特別な手続きをせずにプログラムを書くと，それがメインプログラムと仮定されます．し
かし，それではサブルーチンと区別しにくいので，program文を用いて最初にプログラムの名前を書き
ます．program文は以下の形式です．

　　　　 program プログラム名

これに対し，メインプログラムの終了は end program文で指定します．

　　　　 end program プログラム名

ここで，program文と end program文で指定する “プログラム名 ”は同じでなければなりません．この
ため，メインプログラムは次のような構造になります．

program code_name
.......
.......
.......

end program code_name

この例のようにプログラム名にはアンダースコア ( _ )も使えるので，これをスペースの代わりに使えば
単語をつないだ長いプログラム名を付けることもできます．

program文と end program文の間に Fortranの文法に従った文を並べて，動作させたい計算手順を記
述します．動作手順を記述する文を “実行文 ”といいます．しかし，プログラムに記述するのは実行文
だけではありません．計算途中で必要となる変数領域を確保するための宣言文も書かねばなりません．
このような計算動作に直接携わらない文を “非実行文 ”といいます．Fortranでは，非実行文をプログラ
ムの最初に集約して，実行文はその後に書きます．このため，動作の開始は program文の次の文ではな
く，最初の実行文になります．
完結したメインプログラムの一例を示します1．

program test_code
implicit none
real x,y,z
x = 5
y = 100
z = x + y*100
print *,x,y
print *,’z =’,z

end program test_code

このプログラムにおける実行文・非実行文の区分と，動作開始から終了までの実行の流れを図 1.1に
示します．実行形式のプログラムを起動すると，最初の実行文から動作が開始し，上から順に一行ずつ
実行して，end program文に到達した段階で動作終了となります．第 2章で述べる do文を使った繰り
返しや，if文による条件分岐など，動作指定によっては所定の位置にバックしたり，条件に応じて実行
するかどうかの選択ができますが，それでもそれぞれの文が終了した後に次の文が実行されるという基
本的動作は変わりません．
これは計算機が 1回に一つの動作しかできないためです．プログラム全体を見渡して実行するのでは
なく，1行ずつ順に実行していくので，バックするような動作指定をしない限り，下方に書いた実行文
は上方に影響を及ぼしません．プログラムはこのことを常に念頭に置いて書かなければなりません．

program test_code

  implicit none

  real x,y,z

  x = 5

  y = 100

  z = x + y*100

  print *,x,y

  print *,’z =’,z

end program test_code

プログラム領域

メインプログラム

非実行文

実行文実行の流れ

実行開始

実行終了

図 1.1 プログラム開始から終了までの流れ

なお，2.2節で述べる if文を使って，条件によって途中でプログラムを終了するときや，第 3章で説
明するサブルーチン内でプログラムを終了するときには stop文を用います．stop文を実行すると，そ
の時点でプログラムが終了します．例えば，

1この例のように，プログラム内部の文は，先頭にスペースを数個入れて，program文と end program文よりも少し右に
ずらして書きます．そうすれば，メインプログラムの範囲が明確になります．

program fluid_code
implicit none
real x,y
integer m,n
.......

if (m < 0) stop
.......

end program fluid_code

と書けば，m< 0の時にプログラムは終了し，それ以降は実行しません．

1.2 基本的な実行文の書き方
数値計算における基本的要素は “定数 ”と “変数 ”で，実行文はこれらを使って数学的に記述します．
ただし，あくまでもコンピュータを実行させるための記述ですから，数学における表現とは異なる意味
を持つこともあります．そのあたりの区別が理解できれば，本節の知識だけでもプログラム電卓的な計
算をさせることが可能です．

1.2.1 代入文と演算の書式
実行文には，stop文のように常に同じ動作を指示する文と，いくつかの “数値”を伴って，その数値に
応じた動作を指示する文とがあります．プログラム中で “数値 ”を与える方法は 3種類あります．−500
とか 3.14のような数字を直接書く “定数 ”，xとか yrzのような単語で示した “変数 ”，およびそれらを
使って x+3や sin(y-5)のような計算手順を表した “計算式 ”です．計算式に書かれている手順も計算
機の動作なのですが，プログラム中の計算式は，その計算結果を動作命令に与える “数値 ”として位置
づけられています．このため，計算式を書いただけでは実行文になりません．
プログラムで最も良く使う実行文は “代入文 ”です．これは，

　　　　変数 = 計算式

という形をしています．計算式の代わりに，定数や変数を書くこともできます．プログラムにおける “変
数 ”とはコンピュータのメモリ領域のことで，数値を入れる箱だと考ればいいでしょう．代入文は右辺
の “数値 ”を左辺で示した変数メモリに格納する動作を表します．よって，

　　　　計算式 = 計算式　　　! これはエラー

という形は使えません．Fortranにおいて “=”という記号は，“左辺と右辺が等しい”という意味ではな
いからです．例えば，

x = 4 + 2
y = 9 - x
y = y + 1

のようなプログラムを考えてみましょう．プログラムは上から順に実行されるので，まず 4 + 2の結果
である 6が変数 xに代入されます．次に，9から変数 xに保存されている数値を引いた値が変数 yに代
入されるので，yには 3が代入されます．最後の文は数学の等式と見れば変ですが，イコールが “代入
する動作”だと理解できれば特に不思議な文ではありません．yはその前の文で 3が代入されていますか
ら，最後の文によって変数 yに y+1の結果である 4が代入されます．

Fortranにおける基本演算の書き方と使い方を表 1.1に示します．

表 1.1 演算記号の書き方と使い方
演算記号 演算の意味 使用例 使用例の意味

+ 足し算 x+y x + y

- 引き算 x-y x − y

* 掛け算 x*y x × y

/ 割り算 x/y x ÷ y

** べき乗 x**y xy

- マイナス -x −x

べき乗までの二項演算には以下のような優先順位があり，優先順位の高い方が低い方より先に計算さ
れます．これは数学における演算順序と同じです．

　　　　べき乗 ＞　掛け算または割り算　＞　足し算または引き算

掛け算と割り算のような同レベルの演算は左から順に計算されます．さらに，かっこを使えばかっこの
中が優先的に計算されます．例えば，

f = x + (y-3)/z**2

という文では，一番最初に y-3を計算し，次に zの 2乗を計算し，次に y-3の結果を zの 2乗の結果で
割り，その結果を xに加えて，最後に fに代入する，という動作になります．
なお，時々見かける間違いに，

　　　　 f = x + y/ab

という数式を次のようなプログラムにするものがあります．

f = x + y/a*b

数学では，y/abの記述を y

ab
と解釈することが多いのですが，プログラムでは掛け算と割り算が同じ

レベルなので，y/a*bは “yを aで割って，その結果に bを掛ける ”です．よって，数学的解釈になるよ
うにするには，

f = x + y/(a*b)

と書かなければなりません．

1.2.2 数値の型
コンピュータという機械が取り扱う数値は 2進数で表されていて，基本的には整数です．例えば，1byte
の数は，8bit，つまり 0か 1のどちらかを選択できる数が 8個並んだもので，10進数では 0～255 (=28−1)
までの整数を表すことができます2．Fortranでは小数点のない数字，56とか −1112などの数字を “整
数型”といいます．Fortranの整数型数は 4byte (32bit)で表現されていて，−231～231 − 1の整数を扱う
ことができます．
しかし，数値計算やシミュレーションをする場合には，3.14のように小数点以下を含んだり，3×1019

とか，1.6×10−27とかいった様々なスケールの数値を取り扱うため，整数型だけでは不便です．特に，整
数型数は小数点以下の表現ができないので，割り算をすると全て切り捨てになります．これを忘れてプ
ログラムを書くと，よく間違いを起こします．例えば，

2bitや byteの詳細と実数型の表現については 1.4.1節参照．

x = (5/2)**2
y = x**(3/2)
t = 2/3*y*x

と書くと，x，y，zはいくつになると思いますか？ 答えは x = 4，y = 4，t = 0です．これは整数の
割り算が切り捨てになるため，5/2 → 2，3/2 → 1，2/3 → 0になるからです．
そこで，整数型の他に，3.14のような少数点以下を含んだり，1.6×10−27のような，A×10Bという形
式の数値を取り扱うことができます．これを “実数型 ”といい，Aの部分を “仮数部 ”，Bの部分を “指
数部 ”といいます．通常，数値計算は実数型で計算しなければなりません．
実数型には有効数字の違う 2種類が用意されています．単精度実数と倍精度実数です2．単精度実数
とは 4byteで表される実数のことで，仮数部の有効数字は 7桁程度です．これに対し，倍精度実数とは
8byte (64bit)で表される実数のことで，仮数部の有効数字は 15桁程度です．7桁あれば問題ないと思わ
れるかもしれませんが，何百万個もの数字の合計を計算したり，次数の高い多項式の計算をするときな
どでは，計算回数の増加につれて有効桁数が減少することを考慮しなければなりません．このため，通
常は倍精度実数で計算をします．

Fortranにおける基本的な実数型は単精度であり，倍精度実数型を使用する時にはそれを意識した書
式で書かなければなりません．しかし，最近のコンパイラはオプションを指定すればデフォルトの実数
型を倍精度にする “自動倍精度化機能 ”を持っているので，本書では，単精度実数と倍精度実数を使い
分けることはせず，単に “実数型 ”と表現します3．よって，特に単精度で計算する必要がなければ，コ
ンパイル時に自動倍精度化オプションを付加して倍精度で計算して下さい．多くの計算機は倍精度実数
計算を高速で行えるハードウェアを内蔵しているので，倍精度計算をしてもさほど実行速度は下がりま
せん．
プログラム上で，整数型の定数と実数型の定数は小数点の有無で区別します．例えば，100とか，-12345
と書けば整数型ですが，100.0とか，-12345.とか，-0.0314とか書けば実数型になります．また，1.6×1023

を入力したい時には，1.6e23と書きます．すなわち，A×10B はAeBと書きます．Bが負の場合でも
1.6e-19のように eの後に続けて書きます．また，6e20のように eBを付加した場合には小数点が無く
ても実数型になります．例えば，

　　　　 a = 3.141592 r2 + 3x5 + 6.5×10−5 x − 105

という式をプログラムで書けば以下のようになります．

a = 3.141592*r**2 + 3.0*x**5 + 6.5e-5*x - 1e5

この例のように，r**2とか x**5のように整数べき乗の指数 (2とか 5)には整数型を使います．
先ほど整数型を使ったら期待どおりの結果が出ない例を挙げましたが，以下のように実数型を使えば
正しい結果が得られます．

x = (5.0/2.0)**2
y = x**(3.0/2.0)
y = 2.0/3.0*y*x

Fortranの便利な機能の一つは，複素数が使えることです．複素数にも単精度複素数型と倍精度複素
数型がありますが，自動倍精度化機能を使えばデフォルトが倍精度複素数形になるので，本書では単に
“複素数型 ”と表現します．複素数型の定数は，

3自動倍精度化については付録 C参照．精度の変更については付録 Dで説明しています．

(0.0,1.0)
(1e-5,-5.2e3)
(-3200.0,0.005)

のように，2個の実数をコンマでつないで，かっこで囲みます．前半が実部，後半が虚部です．つまり
この例は，それぞれ，i，10−5 − 5.2×103 i，−3200 + 0.005 iを表した複素数型定数です．
なお，ここまで読むと整数型は必要がないと思われるかもしれませんが，そうではありません．1.3節
で説明する配列の要素を指定する数や，2.1節で説明する do文のカウンタ変数は整数型でなければなり
ません．また，オン・オフを表すだけの変数を作るときにも整数型を使います．すなわち，整数型数は
“順番 ”や “区別 ”を表すときに不可欠です．また実数の整数部を取り出したいときや，剰余 (あまり)を
計算するときも整数型を利用します．数値計算プログラムの中でも整数型の使い道は多いのです．
計算式中に異なる数値型が混在する時は，情報量の多い方の型に合わせて計算し，その型の値が結果
になります．例えば，実数型と整数型の計算は整数型の数値を実数型に変換して実数型と実数型の計算
を行い，実数型の結果になります．複素数型と実数型の計算の場合にはその実数型の数値を実部とした
複素数型にして複素数計算をし，結果は複素数型になります．ただし，掛け算と割り算の計算は左から
順に行うので注意が必要です．最初の整数型を使った計算例で，

t = 2/3*y*x

が 0になるのは，最初に計算されるのが 2/3という整数型÷整数型だからです．これを，

t = y*x*2/3

と書けば，切り捨ては起こりません．

1.2.3 変数の宣言
次に，数式の計算結果を保存する変数の使い方を説明します．変数の名前は，頭文字が a～zのどれか
であれば，後は a～z，0～9をどの様な順序で並べたものでも使えます．例えば，abcとか k10xy等で
す．Fortranでは大文字と小文字を区別しないため，abcと ABCは同じ変数になります．変数にも型が
あり，計算結果に応じた型の変数を使わなければ正確に保存することができません．この変数の型を決
める文を “宣言文 ”といいます．型を決めると同時に，変数のメモリ領域を確保します．このため，計
算に用いる変数は全て宣言しなければなりません．宣言は一度しかできず，プログラムの実行時に変更
することはできません．宣言文は非実行文です．
ところが Fortranには “暗黙の宣言 ”があり，通常は宣言しなくても文法的な間違いにならないので，
タイプミスなどで思わぬエラーが発生する可能性があります．これを防ぐため，プログラムの 2行目，
すなわち program文の次の行に必ず以下の implicit文を書いておきます．

implicit none

この文を入れておけば，宣言せずに使用した変数があるとコンパイルエラーになり，タイプミスのチェッ
クができます．
数値計算は基本的に実数で行いますが，実数型変数は次のように宣言します．

real 変数 1, 変数 2，...

これに対し，整数型の変数は，次のように宣言します．

integer 変数 1, 変数 2，...

宣言文は非実行文なので，全ての実行文より前に書かなければなりません．例えば，メインプログラ
ムの始まりは以下のようになります．

program test1
implicit none
real x,y,z,omega,wave,area
integer i,k,n,imin,imax,kmax
.......

realや integer等の型宣言文は何行書いても良いし，順番も無関係です．
Fortranの暗黙宣言では，名前の頭文字が a～hか o～zの変数は実数型で，i～nの変数は整数型でし
た．このため，整数型変数の頭文字を i～nにする習慣があります．全てを宣言する以上，基本的に変数
名の付け方に制限はないのですが，整数型は用途が限られているので，頭文字を限定する方が良いと思
います．実数型数の名前はあまり頭文字にこだわる必要はありませんが，原則として整数型をイメージ
する i,j,k,l,m,nの 1文字変数は使わない方が無難です．
複素数型変数の宣言文は，

complex 変数 1, 変数 2，...

です．複素数型も用途が限定されているので名前の付け方に規則をつける方が良いでしょう．複素数型
変数名は，頭文字を cか zにすることが多いようです．
プログラムにおいて，数値を変数に保存する意義は大別して二つあります．一つはプログラム全域に
わたって共通してその値を利用するためであり，もう一つは狭い範囲で計算結果を一時的に保存するた
めです．この二つは意識して使い分けるようにします．特に，前者の大域的に利用する変数には長めで
意味のある名前を付けるべきです．これは，1文字のような短い変数名を使うと，プログラムが長くな
るにつれて，どこでその変数を使っているかを検索するのが難しくなるからです．
変数に数値を代入する時は，右辺の計算結果を左辺の変数の型に変換して代入します．このため，実
数の計算結果を整数型の変数に代入すると，小数点以下は切り捨てられます．例えば，

real x
integer n
x = 3.14
n = x + 6

の結果，nに代入される値は 9です．
また，複素数型の計算結果を実数型の変数に代入すると，その複素数の実部が代入されます．例えば，

real x
complex c
c=(1.0,-2.0)
x=c**2

とすると，x= −3.0になります．逆に，複素数型の変数に実数型の数値を代入すると，実部に結果が代
入され，虚部は 0になります．例えば，

real x
complex c
x=5
c=x**2

とすると，c= (25.0, 0.0)になります．

1.2.4 組み込み関数
数値計算上よく使う数学関数はあらかじめ用意されています．この用意された関数を “組み込み関数”
といいます．表 1.2に，代表的な組み込み数学関数を示します．

表 1.2 数学関数の書き方と使い方
組み込み関数 名　称 数学的表現 必要条件 関数値 f の範囲
sqrt(x) 平方根*

√
x x = 0

sin(x) 正弦関数* sin x

cos(x) 余弦関数* cosx

tan(x) 正接関数 tan x

asin(x) 逆正弦関数 sin−1 x −1 5 x 5 1 −π
2 5 f 5 π

2

acos(x) 逆余弦関数 cos−1 x −1 5 x 5 1 0 5 f 5 π

atan(x) 逆正接関数 tan−1 x −π
2 < f < π

2

atan2(y,x) 逆正接関数4 tan−1(y/x) −π < f 5 π

exp(x) 指数関数* ex

log(x) 自然対数* loge x x > 0
log10(x) 常用対数 log10 x x > 0
sinh(x) 双曲線正弦関数 sinh x

cosh(x) 双曲線余弦関数 coshx

tanh(x) 双曲線正接関数 tanh x

関数名の後のかっこは必ず必要です．かっこの中の xや yを “引数 (ひきすう) ”といいます．関数を計
算式中に記述すると，引数に対する関数値が結果となってその計算式を実行します．例えば，

c = exp(-x**2) + sin(10*x+3) - 2*tan(-2*log(x))**3

のように書くことができます．この例のように関数の引数に関数を使った計算式を与えることも可能
です．
表 1.2の数学関数は，引数に実数型数を与えると実数型の結果になる “実数型関数 ”ですが，名称に

“*”の付いている関数は，引数が複素数型でも使えます．Fortranの組み込み関数には，引数の型や精度
に応じて計算をする “総称名 ”機能があるので，複素数を引数に与えた場合の関数値は複素数型の結果
になります5．
ただし，総称名機能では引数の数値型に対応した関数が用意されていないとコンパイルエラーになり
ます．例えば，

√
2を計算したくて，sqrt(2)と書くとエラーです．“2”は整数型の定数であり，整数型

数の平方根は用意されていないからです．sqrt(2.0)のように実数型を引数に書かなければなりません．
この他，絶対値や余りを計算するのにも組み込み関数を使うし，実数型を複素数型に変換するなどの
型変換も組み込み関数を使って処理します．その中の代表的なものを表 1.3に示します．
型変換関数は，数値型が限定されている場所に，その数値型以外の値を指定したいときなどに使いま
す．例えば，整数型の変数 nの平方根を計算したいときには，上記のように sqrt(n)と書くことはで
きません．このときは，sqrt(real(n))と書きます．また，整数化は切り捨て演算を含んでいるので，
それを積極的に利用するときにも使います．例えば，正の実数 xの小数点以下を四捨五入した整数は，
int(x+0.5)で計算することができます．

4逆正接関数 atan2(y,x)は，座標点 (x, y)の偏角を計算する関数です (xと y の順序に注意)．よって，表の数学的表現に
は “ tan−1(y/x) ”と記述してありますが，実際には xと y の値に応じて −π から π の間の角度が結果になります．

5表 1.2の必要条件と関数値の範囲は引数が実数型の場合です．この必要条件を満たさない実数型数を与えると実行時エラー
になります．しかし複素数型を与えた場合には必ずしもエラーにはなりません．

複素数型の定数は，(1.0,-2.0)のような表現で指定することができますが，変数や計算式で実部と
虚部を指定した複素数を作る時には関数 cmplxを使います．例えば，

complex z,z1
real x,y
x = 10.0
y = -3.5
z = cmplx(x,y)
z1 = cmplx(x**2,y+2.0)

のように書きます．この場合，z= x + iyであり，z1= x2 + i(y + 2)になります．

表 1.3 型変換関数などの組み込み関数
組み込み関数 名　称 引数の数値型 関数値の数値型 関数の意味
real(n) 実数化 整数 実数 実数型に変換
abs(n) 絶対値 整数 整数 nの絶対値

mod(m,n) 剰余6 2個の整数 整数 mを nで割った余り
int(x) 整数化 実数 整数 整数型に変換 (切り捨て)
nint(x) 整数化 実数 整数 整数型に変換 (四捨五入)

sign(x,s) 符号の変更 実数 実数 s= 0なら |x|，さもなくば−|x|
abs(x) 絶対値 実数または複素数 実数 xの絶対値

mod(x,y) 剰余 2個の実数 実数 xを yで割った余り
real(z) 複素数の実部 複素数 実数 zの実部
imag(z) 複素数の虚部 複素数 実数 zの虚部

cmplx(x,y) 複素数化 2個の実数 複素数 x + iy

conjg(z) 共役複素数 複素数 複素数 zの共役複素数

1.2.5 print文による簡易出力
数値計算を目的としたプログラムでは，得られた計算結果を表示したりファイルに保存する必要があ
ります．入出力の詳細については第 4章で説明しますが，最低限，print文を使った標準形式による数
値出力は覚えておいて下さい．print文の一例を以下に示します．

integer n
n = 3
print *, 4+5, n, n*2, 2*n-11

このように，“ print *, ”に続いて変数や計算式をコンマで区切って並べると，それらの計算結果が
横に並んで出力されます．上例の場合には，4+5の結果である 9から 2*n-11の結果である−5までが以
下のように出力されます．

9 3 6 -5

なお，“ print *, ”の “ * ”は，標準形式で出力することを意味しているのですが，とりあえずは形式
的に書くものだと覚えて下さい7．
複数の数値を出力するときに数字だけを出力すると，どれがどの数値かわからなくなる可能性があり
ます．このような場合は，文字列を併用して変数の意味を同時に出力します．Fortranにおける “文字

6剰余を計算する関数 modの利用例として繰り返しの制御があります．2.2節を参照して下さい．
7出力形式の指定方法は 4.3節参照．

列 ”とは，2個のアポストロフィ「’」または 2個のダブルクォーテーション「"」ではさんだ文字の並び
のことで，print文中で数値といっしょに並べると，その文字の並びがそのまま出力されます．例えば，

real x
x = 3
print *,’x = ’,x,’ x**3 = ’,x**3

というプログラムの出力は，

x = 3.00000000000000 x**3 = 27.0000000000000

となります．

1.3 配列
ここまで出てきた変数は型に応じた 1個の数値を記憶することしかできませんでした．しかし，数値
計算やシミュレーションでは，数万個のデータを保存して，それぞれを条件に応じて変化させていく，
なんていうのが当たり前のように行われます．そこで，数学で a1, a2, · · · , anのように変数に添字を付け
て区別するように，数字で区別した変数を作ることができます．これを “配列 ”といいます．本節では
配列の使い方について説明します．

1.3.1 配列宣言
配列は変数の一種なので，型宣言文を使って宣言しておかねばなりません．単一変数と異なるのは，
宣言時に添字の範囲を示す整数値をかっこを使って付加することです．例えば，次のように宣言します．

real a(10),b(20,30)
complex cint(10,10)
integer node(100)

ここで，aや nodeのように数字が 1個の配列を 1次元配列，bや cintのように数字が 2個の配列を 2
次元配列といいます．3次元以上の配列を作ることもできます．配列の名前の付け方，頭文字の選択な
どの規則は単一変数名の付け方と同じにします．宣言した配列の各部の名称を表 1.4に示します．

表 1.4 配列の各部の名称
real a(10) と宣言した配列について
　　記　号 　　名　称
a 配列名
a(3)など 配列要素
かっこ中の数字 要素番号，添字

配列宣言で指定した数値は要素番号の最大値を表し，要素番号の使用可能範囲は 1から指定した最大
値までです．例えば a(10)の宣言では a(1)から a(10)までの 10個の配列要素が使用可能になります．
また，2次元以上の配列の場合には各次元ごとの最大値を指定しているので，b(20,30)の宣言では全部
で 20×30 = 600個の配列要素が使用可能になります．
問題によっては，要素番号として 0や負数を使いたい時があります．このような時には “: ”を間に入
れて，使用可能な要素番号の最小値と最大値を同時に指定します．例えば，

real ac(-3:5),bc(-20:20,0:100)

と宣言すると，1次元配列 acは，ac(-3)から ac(5)までの 9個が使用可能であり，2次元配列 bcは，
bc(-20,0)から bc(20,100)までの (20×2 + 1)×(100 + 1) = 4141個が使用可能です．

–10–



1.3.2 メモリ上での配列の並び
配列はコンピュータ内部において連続したメモリ領域で実現されています．例えば，real a(10)と
宣言された配列は，図 1.2左のように，a(1),a(2),...,a(10)の順で並んだ実数型のメモリです．

101:

102:

103:

104:

105:

106:

107:

108:

109:

110:

内部メモリ

....

a( 1)

a( 2)

a( 3)

a( 4)

a( 5)

a( 6)

a( 7)

a( 8)

a( 9)

a(10)

a 201:

202:

203:

204:

205:

206:

207:

208:

209:

210:

内部メモリ

....

ac(-3)

ac(-2)

ac(-1)

ac( 0)

ac( 1)

ac( 2)

ac( 3)

ac( 4)

ac( 5)

ac

real a(10) real ac(-3:5)

図 1.2 1次元配列のメモリ並び

図からわかるように，a(3)とは a(1)から数えて 3番目のメモリのことです．また，real a(10)の宣
言では 10個しかメモリを確保していないのですから，a(10000)のように範囲外の要素番号を指定する
と問題が起こります．10000番目の要素 a(10000)がどのメモリを示すのか不明だし，そもそも存在す
るという保証さえありません．

Fortranでの配列名は配列を代表する名称であると同時に，配列の先頭メモリを示します．例えば，配
列名 aは図 1.2左のように a(1)を示します．また，ac(-3:5)のように最小値も指定して宣言した場合
には，図 1.2右のように並んでいて，配列名 acは ac(-3)を示します．配列名が先頭要素を示すことは
配列をサブルーチンの引数として与える時に意識する必要があります．

2次元以上の配列の場合は，左の方の要素番号から先に進むようにメモリ上で並んでいます．例えば，

real b(3,2)

と宣言した場合，メモリ上での並びは図 1.3のようになります．2次元以上の配列も配列名は先頭要素を
示しています．配列 bの場合，配列名 bは b(1,1)を示します．　

301:

302:

303:

304:

305:

306:

307:

....

b(1,1)

b(2,1)

b(3,1)

b(1,2)

b(2,2)

b(3,2)

b

real b(3,2)

内部メモリ

図 1.3 2次元配列のメモリ並び

2次元と 3次元の配列の並び方を式で表せば，

　　 real b(m,n)の配列宣言で，b(i,j)は先頭から数えて i + m*(j-1)番目
　　 real b(`,m,n)の配列宣言で，b(i,j,k)は先頭から数えて i + `*(j-1 + m*(k-1))番目

となります．どちらの公式も一番右側の要素数 nに依存していません．一般的に，どんな次元の配列の
順番を与える公式も一番右側の要素数には依存しません．

1.4 プログラミング スマートテクニック (その 1)

計算手順が複雑になったり大量のデータを処理するようになると，数式を単純にプログラムに置きか
えるのではなく，計算機の特性を考慮したプログラムにすることで計算効率や精度を高めることができ
ます．ここでは，計算速度を向上させるテクニックや，コメントを入れてプログラムのメンテナンスを
容易にする手法などについて説明します．

1.4.1 コンピュータで表現可能な数値の大きさ
我々が日常的に使っている 10進数は，10のべき乗が基本です．すなわち，数字を 10の多項式で表し
たときの係数を次数の大きい方から並べたものです．ただし，係数は 10より小さい整数 (0～9)でなけ
ればなりません．例えば，1203という数は，10の多項式で表せば，

　　　　 1203 = 1 × 103 + 2 × 102 + 0 × 101 + 3 × 100

となるので，表現が “ 1203 ”なのです．
これに対し，コンピュータ内部で使われている 2進数は，2のべき乗を基本として表された数です．こ
の場合，多項式の係数は 2より小さい整数でなければならないので，0か 1です．例えば，10進数で 123
と表示される数を 2の多項式で表せば，

　　　　 123 = 1 × 26 + 1 × 25 + 1 × 24 + 1 × 23 + 0 × 22 + 1 × 21 + 1 × 20

となるので，2進数で表すと，1111011となります．2進数は 0または 1だけで表すことができるので，
電子回路のON/OFF回路を使って実現することができます．これがコンピュータで 2進数が使われて
いる理由です．2進数の一桁を bitといいます．現在のコンピュータは 2進数 8桁 (8bit)を基本に作られ
ていて，これを byteといいます．1byteは 8桁の 2進数ですから 28 = 256個の数字をあつかうことがで
きます．これを 0以上の整数と考える場合には，0～255となります．
さて，Fortranの整数型は 4byteなので，32桁 (32bit)の 2進数です．すなわち，232個の数字があつかえ
ます．しかし負の整数を含ませるために，整数型の範囲は−231～231−1になります．231は 2,147,483,648
ですから，取り扱える数の上限はおよそ 21億です．整数型の計算をするときは，絶対値がこれを超えな
いようにしなければなりません．
これに対し，実数型数は数値の bitを仮数部 f と指数部 e，および正負を決める符号 bitに分割して，

±f×2eのような形式で表現します．指数部 eは負数も含んでいるので，10−19のような絶対値の小さい
数も表現できます．実数型における bit分割とそれによる数の表現には規格がいくつかありますが，現
在最もよく使われているのは IEEE 754と呼ばれている規格です．IEEE 754における実数型の bit分割
数と，それによる表現可能な絶対値の上限を表 1.5に示します8．
例えば，倍精度実数型は仮数部が 52bitなので，252 ; 4.5×1015個の区別ができ，これが “有効数字

15桁程度 ”の根拠です．また，指数部の最大が 1023なので，最大 10308程度まで表現することができま

8IEEE 754に関しては「http://ja.wikipedia.org/wiki/浮動小数点数」を参考にしました．詳しくは，このWebページを
見て下さい．なお，表現できる正数の下限は，およそ「1/上限数」程度，という見当で良いと思いますが，IEEE 754の規格
ではもう少し小さい桁の数まで表せます．もっとも，実際にパソコンで上限と下限をチェックしてみたところ，下限は動作環
境やコンパイラによって異なりました．また，4倍精度に関しては上限が倍精度と同じ 10308 程度しかないものもありました．
すなわち，必ずしも IEEE 754の規格通りに実装されているとは限らないようです．

表 1.5 実数型の bit分割数と表現可能な絶対値の上限 (IEEE 754)

数値型 (bit) 符号 bit 指数部 bit 指数部 eの範囲 仮数部 bit 絶対値の上限
単精度 (32) 1 8 −126～127 23 3.4×1038程度
倍精度 (64) 1 11 −1022～1023 52 1.7×10308程度

4倍精度 (128) 1 15 −16382～16383 112 1.1×104932程度

す．これに対し，単精度は指数部の最大が 127なので，最大は 1038程度です．1038のような大きな数字
を扱うことはあまりないかもしれませんが，計算の途中で現れる可能性は考慮する必要があります．例
えば，プランク定数は h ; 6.6×10−34 Jsなので，hの逆 2乗が公式の中に入っていると，途中計算で
1038を越えてしまいます．倍精度実数を使う意義はここにもあります．
なお，IEEE 754での指数部には 2eという以外にも意味があり，数字の組み合わせによっては，“無
限大 ”や “非数 ”を表します．もし，print文で数値を出力したときに，Infinityや INFという文字が出
力された場合には無限大です．つまり，計算した結果が表現できる上限を超えたという意味です．これ
を “オーバーフロー ”といいます．
これに対し，NaNと出力された場合は非数です．非数 (Not a Number)というのは，「数字じゃない」
という意味ですが，具体的には，0を 0で割ったときや，−1の平方根を計算したときの結果がこれに相
当します．ただ，結果が非数になっても計算が継続することもあるので，どの時点で発生したのかを特
定するのは結構面倒です．

1.4.2 演算の速度を考える
コンピュータの計算動作は a+bのような加算が基本です．a-bのような減算は bを負数に変換して加
算するだけなので加算とそれほど実行時間は変わりませんが，a*bのような乗算は加算の繰り返し動作
ですから，加減算よりかなり遅いです．a/bのような除算にいたっては，減算の繰り返しを条件付きで
行うので，さらに時間がかかります．この演算速度の比較を書けば次のようになります．

　　　　加減算　≫　乗算　≫≫　除算

このため，割り算ができるだけ少なくなるような計算手順にします．例えば，

x = a/b/c

と書くより，

x = a/(b*c)

と書く方が速くなります．最近の計算機はかなり高速に計算することができるので，数回の計算だけな
らこのような書き換えはあまり意味がありませんが，大量に繰り返し処理をする時には価値があります．
べき乗算はもっと遅いので，2乗や 3乗程度のときは掛け算にする方が速くなります．例えば，

x = a**2 + b**3

と書くより，

x = a*a + b*b*b

と書く方が速くなります．ただし，指数が大きいときには意味がないし，プログラムもわかりにくくな
るので 3乗程度までで良いと思います．
実数のべき乗 (1.2乗とか，3.7乗など)は，対数関数と指数関数を使って計算するので時間がかかりま
す．このため，1

2 乗に関しては，組み込み関数 sqrtを利用する方が速いです．例えば，

–13–



y = x**0.5
z = y**1.5

と書くより，

y = sqrt(x)
z = y*sqrt(y)

と書く方が速くなります．

1.4.3 桁落ちに気を付ける
実数型数の有効数字は倍精度でも 15桁程度です．よって，値の接近した 2個の実数の引き算をすると
きは気を付けなければなりません．例えば，

　　　　 2000.06 − 2000.00 = 0.06

ですが，計算結果 0.06は元の 2000.06と比べると有効数字が 5桁も少なくなっています．これを “桁落
ち ”といいます．桁落ちするような引き算はできるだけ避けねばなりません．
例えば，次のように変数 x1に小さい正の数 h1を加えて x2を計算し，x2に小さい正の数 h2を加えて

x3を計算したとします．

real x1,x2,h1,h2
x1 = 1.0
h1 = 1e-7
h2 = 2e-7
x2 = x1 + h1
x3 = x2 + h2
print *,x3-x1,h1+h2

最後の print文の出力を見ればわかりますが，x3-x1の値が必要な時には，引き算で直接計算するより
も h1+h2で計算する方が精度が上がります．

2次方程式 ax2 + bx + c = 0 の解を求めるときにも注意が必要です．この方程式の解の一つは，

　　　　 x1 =
−b +

√
b2 − 4ac

2a

ですが，b > 0かつ b2 � |4ac|の時には，bと
√

b2 − 4acがほぼ等しくなるので，−b +
√

b2 − 4acの計
算をすると桁落ちする可能性があります．そこでこの解を計算する時は，分子の有理化をした公式を使
います．すなわち，分母分子に−b −

√
b2 − 4acを掛けると，

　　　　 x1 =

(
−b +

√
b2 − 4ac

)(
−b −

√
b2 − 4ac

)
2a
(
−b −

√
b2 − 4ac

) =
2c

−b −
√

b2 − 4ac

となりますが，最後の公式を使えば桁落ちする心配がありません．2解とも計算する時には，2解の積が
c/aであることを利用して，まず桁落ちしない方の解 x1を計算した後で，もう一方の解を c/(ax1)で計
算すると良いでしょう．

1.4.4 πの作り方
物理や数学の計算をするときには円周率 πを使うことが多いです．πの値を倍精度計算用に有効数字

16桁で表せば，

real pi
pi = 3.141592653589793

となります．これをこのまま使えば良いのですが，計算機によっては有効数字が異なるし，数値を覚え
るのも面倒です．そこで逆三角関数を使って計算で求める方法がよく使われています．例えば，

real pi
pi = acos(-1.0)
pi = 2.0*asin(1.0)
pi = 4.0*atan(1.0)

などのように書くことができます．関数計算には時間がかかりますが，πの値は変化しないので計算す
るのは 1回だけです．問題になるほどではありません．

1.4.5 コメント文の挿入
Fortranでは “ ! ”の後に続く文字は全て無視されます．すなわち何を書いても実行とは無関係です．
これをコメント文といいます．コメント文を機会あるごとにプログラム中に入れて，書かれている内容
を表示するように心がけましょう．さもなくば，プログラムが長くなるにつれて，自分でも何を書いた
のか忘れてしまいます．例えば，

! area of circles
s = pi*r*r

のように，１列目に “ ! ”を書けば，その行はコメント行になります．また，

s = pi*r*r ! 円の面積
v = 3*pi*r*r*r/4 ! 球の体積

のように，実行文の末尾に書き込むこともできるし，日本語を書くこともできます．ただし，日本語環
境によっては正しく認識できずに文字化けすることがあります．可般性を考慮するならローマ字つづり
でも良いから半角英数字だけでコメントを書いた方が無難です．

1.4.6 継続行と複文
計算式が非常に長くて，作成に使っているエディタウィンドウの横幅を越えると読みづらくなります．
また，コンパイラによっては 1行に書ける文字数に制限があるので，そのままではエラーになることも
あります．そこで，1行の文を途中で改行して，複数行に分割して書くことができます．このとき，次
の行に継続するという意志を示す印として，行末に “ & ”の文字を書きます．例えば，

print *,alpha,beta,gamma,delta,epsilon,zeta,eta,iota,kappa,lambda,mu

という 1行は，

print *,alpha,beta,gamma,delta,epsilon &
,zeta,eta,iota,kappa,lambda,mu

と分割して書くことができます．継続行は何行にわたってもかまいません．

print *,alpha,beta,gamma &
,delta,epsilon &
,zeta,eta,iota &
,kappa,lambda,mu

と書いても同じです．ただし，最後の行に “ & ”を付加するとエラーになるので注意して下さい．

長い文字列を書くときは，文字列の最後に “& ”を入れて継続させることができます．その際，次の行
の開始文字の前にも “ & ”を書いておきます．例えば，

print *,’ABCDEFG&
&hijklmn’

と書けば，ABCDEFGhijklmnと出力されます．
逆に，1行に複数の文を書くことも可能です．これを “複文 ”といいます．2個以上の文を 1行で書く
には，文と文を分離する記号として “ ; ”の文字を書きます．例えば，

x = 1; y = 2; z = 3

のように書くことができます．ただし，複文の使用はこのような短い代入文を書く場合に限定して下さ
い．プログラムは 1行で一つの完結した動作を表すのが基本です．1行に複数の文を書くと，動作の流
れが見えにくくなるので，多用しない方が良いと思います．

演 習 問 題 1

(1–1) 2次方程式の解
3個の実変数 a，b，cに適当な値を代入し，それから，

ax2 + bx + c = 0

の 2解を計算して出力するプログラムを作成せよ．さらに，その解を ax2 + bx + cに代入して 0に近い
値になることもプログラムして確かめよ．また，a，b，cの値が不適切だと，エラーが出ることも確認
せよ．(例えば，a = b = c = 1にしてみよ)

(1–2) ヘロンの公式
三角形の 3辺の長さを a，b，cとすると，その三角形の面積 Sは次式 (ヘロンの公式)で表される．

S =
√

p (p − a)(p − b)(p − c)

ただし，p =
a + b + c

2
である．

a，b，cに適当な数値を与え，この公式を使って三角形の面積を計算するプログラムを作成せよ．さ
らに，一辺 aの正三角形の面積を別の方法で計算するプログラムを作成し，ヘロンの公式から求めた面
積と一致するかどうか確かめよ．

(1–3) 面積，体積
1個の実変数 aと整変数 nに適当な数値を与え，これから，半径 aの円の面積，半径 aの球の体積，一
辺 aの正 n角形の面積を計算するプログラムを作成せよ．ただし，πの値はできるだけ正確な値 (有効
数字 15桁程度)を使うこと．

(1–4) ビオ・サバールの法則

QIP

r

A

α β

図 1.4 線分電流

右図のように線分導体PQに電流 I [A]が流れている時，そ
の線分から距離 r [m]の点 Aにできる磁場の磁束密度 B [T]
は次の公式で与えられる．

B =
µ0I

4πr
(cos α + cos β)

ここで，µ0は真空の透磁率 (4π×10−7 H/m)，α，βはそれぞ
れ点 P，点Qにおける三角形APQの内角である．
この公式を使って，I，r，α，βを入力して磁束密度Bを計
算するプログラムを作成せよ．ただし，角度は “度 ”の単位の数値で与えて，内部でラジアンに換算す
るようにせよ．
次に，一辺 a [m]の正三角形の導線に電流 I [A]が流れている時，中心にできる磁束密度B [T]を計算
するプログラムを作成せよ．
さらに，一辺 a [m]の正方形の導線に電流 I [A]が流れている時，中心にできる磁束密度B [T]を計算
するプログラムを作成せよ．

(1–5) 相対論的エネルギー
相対性理論によれば，質量m [kg]の物質は，真空中の光の速度を c (=2.99792458×108 [m/s])とすると，

mc2 [J]の静止エネルギーを持っている．この公式を使って電子と陽子の静止エネルギーを計算するプログ
ラムを作成せよ．ここで電子の質量は9.10938356×10−31 [kg]，陽子の質量は1.672621898×10−27 [kg]であ
る．なお，エネルギーの単位はエレクトロンボルト (eV)に換算して表示せよ．1eVは1.6021766208×10−19

[J]である．
また，物質が速度 v [m/s] で運動しているときの運動エネルギーは m(γ − 1)c2 である．ここで，

γ =
1√

1 − v2/c2
である．電子と陽子が光速の 1%で動いているときの運動エネルギーと，90%で動い

ているときの運動エネルギーをそれぞれ計算するプログラムを追加せよ．こちらも単位はエレクトロン
ボルトとする．

(1–6) 複素インピーダンス
インダクタンス L=0.04 [mH] ，抵抗 RL=10 [Ω] のコイルと，C=100 [µF] のキャパシタンス，およ
び抵抗R=100 [Ω] を並列に接続し，これに周波数 f=60 [Hz]の電圧E=100 [V] を加えたとき，流れる
複素電流 I，力率 cos θおよび有効電力 P を計算して出力するプログラムを複素変数を使って作成せよ．
この場合も πの値はできるだけ正確な値を使うこと．ここで，抵抗R，インダクタンス L，キャパシタ
ンス C，に対する複素インピーダンスは，それぞれ，

ZR = R, ZL = iωL, ZC =
1

iωC

である．ただし，ω = 2πf である．
また，それら 3つのインピーダンスを直列につないだときに流れる複素電流 I，力率 cos θおよび有効
電力 P を計算するプログラムも作成せよ．

(1–7) 回路計算
5個の抵抗R1，R2，R3，R4，R5がある時，この 5個を直列にした時の合成抵抗Rs，5個を並列にし
た時の合成抵抗Rp，およびR1とR2を並列にしたものとR3とR4を並列にしたものとR5とを直列に
した時の合成抵抗Rpsを計算するプログラムを作成せよ．ただし，5個の抵抗値は，要素数 5の 1次元
配列 R(5)で表すとする．

(1–8) 3次元ベクトルと電磁気学
3次元ベクトルA = (A1, A2, A3)を要素数 3の 1次元配列 a(3)で表すとする．まず電場ベクトルE，
磁場ベクトルBを配列で表して適当な数値を代入する．次に荷電粒子の速度ベクトル vを配列で表し
て適当な数値を代入し，これらの配列から，内積 v •E，および外積ベクトル v×Bを計算するプログラ
ムを作成せよ．外積ベクトルも配列にすること．

(1–9) 2行 2列の行列計算

2行 2列の行列，A =

(
a11 a12

a21 a22

)
を，2次元配列 a(2,2)で表すとする．2個の 2行 2列の行列A

と Bを表現する配列，a(2,2)と b(2,2)の各要素に適当な数値を代入して，Aと Bの足し算の行列
C = A + B，掛け算の行列M = AB，および，それぞれの行列式DA，DB を計算するプログラムを作
成せよ．

第2章 手順の繰り返しと条件分岐
第 1章で説明した基本的プログラミングは，単純な計算動作を並べたものであり，言ってみれば関数
電卓の操作をプログラムという書式で表しただけに過ぎません．コンピュータの能力を使いこなすには，
同じような動作を何度も繰り返したり，条件に応じて異なる動作をさせる手法の知識が必要です．本章
ではこれらの書き方について説明します．

2.1 手順の繰り返し — do 文
実行文は基本的に上から下へ順に実行されますが，それだけでは類似した手順を繰り返す時に，必要
な回数だけ同じ文を書かねばなりません．そこで，ある範囲の手順を必要な回数だけ繰り返し行わせる
手段として do文があります．do文を使う時の基本形は，

do 整数型変数 = 初期値, 終了値
.......
.......

enddo

です．最初の doの行が do文で，do文と最後の enddo文で範囲を指定した一連の実行文が繰り返し実
行されます．この範囲を “ doブロック ”といいます．また，プログラムの流れが循環するという意味で
“ doループ ”ともいいます．do文の整数型変数を “カウンタ変数 ”と呼びます．doブロックの繰り返し
は，カウンタ変数に “初期値 ”を代入することから開始し，doブロック内の手順を 1回実行するごとに
カウンタ変数を 1増加します．そして，カウンタ変数が “終了値 ”より大きくなった時点で繰り返しは
終了し，enddo文の次の文に実行が移ります．例えば，

do m = 1, 10
a(m) = m

enddo

と書けば，a(1)～a(10)までの配列要素に，1～10までの数値が順に代入されます．do文における，“カ
ウンタ変数＞終了値 ”の判定は doブロックの開始時にも行うため，初期値が終了値より大きい場合には
ブロック内部が一度も実行されずに enddo文の次に実行が移ります．
次の do文のように，整数値をもう一つ追加することで増加量を指定することもできます．

do 整数型変数 = 初期値, 終了値, 増分値
.......
.......

enddo

このときカウンタ変数は初期値から開始して，“増分値 ”ずつ増加しながら doブロック内の手順を繰り
返し，“カウンタ変数＞終了値 ”の時点で終了します．
例えば，mが 10以下の奇数の時のみ計算をしたい時は，

do m = 1, 10, 2
.......
.......

enddo

と書きます．この時の終了値は 10ですが，10は奇数ではないので計算しません．
増分値は負数を指定することもできます．負数の時にはカウンタ変数が減少していくので，“カウン
タ変数＜終了値”になった時点で終了します．例えば，100から順に下って 1まで繰り返す時には以下の
ように書きます．

do m = 100, 1, -1
.......
.......

enddo

増分値が負数の時には，“初期値＜終了値 ”ならば doブロック内部は一度も実行されません．
do文のカウンタ変数に増分値を加えるタイミングは，enddo文の実行時です．例えば，

do m = 1, 3
a(m) = m**2

enddo

という文は，

　　 m = 1
　　 ［m> 3の判定をする．満足しないので，doブロックを実行］
　　 a(m) = m**2
　　 m = m + 1
　　 ［m> 3の判定をする．満足しないので，doブロックを実行］
　　 a(m) = m**2
　　 m = m + 1
　　 ［m> 3の判定をする．満足しないので，doブロックを実行］
　　 a(m) = m**2
　　 m = m + 1
　　 ［m> 3の判定をする．満足するので，doブロックを終了］

という動作になります．この展開からわかるように，doブロックを終了した時点で，カウンタ変数には
終了値より大きい値が代入されています．上例では，doブロック終了後，m= 4です．doブロック終了
時の mの値を利用する時は，このことを考慮しなければなりません．
次のように，doブロックの中に別の doブロックを入れて多重にすることもできます．ただし，カウ
ンタ変数は異なるものを使わなければなりません．これは，doブロック内でカウンタ変数を変更するこ
とが禁止されているからです．

do k = 1, 100
a(k) = k**2
do m = 1, 10
b(m,k) = m*a(k)**3
c(m,k) = b(m,k) + m*k

enddo
d(k) = a(k) + c(10,k)

enddo

よく使うので覚えておくと便利なのが，合計を計算する do文です．例えば，n要素の 1次元配列，a(1),
a(2), ..., a(n)の中に数値データが入っている場合，これらの合計 sumを求めるには do文を使って
以下のように書きます．

sum = 0
do m = 1, n

sum = sum + a(m)
enddo

　
この文が合計を計算していることを理解するには，やはり以下のように手動展開してみると良いでしょう．

sum = 0
m = 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
......

ここで判定文は省略しました．このプログラムで重要なことは，do文の前で変数 sumに 0を代入してい
ることです．これがないと正しい結果が得られないことがあります．このようなプログラムならではの
書き方はパターンとして覚えておくと良いと思います．

2.2 条件分岐 — if 文
条件に応じて異なる手順を行わせることを “条件分岐 ”といい，if文を使って指定します．最も単純
に，一つの条件に応じて一つの文を実行するかしないかを決めるだけの時は単純 if文を使います．単純
if文は以下の形式です．

if (条件) 実行文
　

単純 if文はかっこ内の “条件 ”が満足されれば，その右の “実行文 ”を実行し，条件が満足されなけれ
ば何もしないで次の文に実行が移ります．例えば，

a = 5
if (i < 0) a = 10
b = a**2

と書くと，i< 0の場合には a= 10となり，それ以外は a= 5のままなので，それに応じて bに代入さ
れる値が異なります．
しかし単純 if文には実行文が一つしか書けないので，実行したい文が複数あるときには使えません．
また，条件に合った時の動作指定しかできないので，合わなかった時の動作を別に指定したい時には不
便です．そこで，通常は単純 if文ではなく，ブロック if文を使います．ブロック if文とは，if文の
実行文のところを thenにした文のことで，以下のようにブロック if文と endif文で一連の実行文の範
囲を指定します．

if (条件) then
.......
.......

endif

この指定された範囲を “ ifブロック ”といい，ブロック if文に書かれた条件を満足した時のみ，ifブ
ロック内の実行文が実行されます．例えば，

a = 5
b = 2
if (i < 0) then

a = 10
b = 6

endif
c = a*b

と書くと，i< 0の場合には a=10，b=6の代入文を実行してから c=a*bを計算しますが，それ以外，す
なわち i= 0の場合には ifブロック内を実行しないので，aも bも変化せず，a=5，b=2のままで c=a*b

を計算します．
さて，この例では，i< 0という条件を満足しないときには，あらかじめ a=5，b=2という代入をす
る必要がありません．このように “条件を満足しない場合 ”に別の動作をさせたいときには ifブロック
内に else文を挿入します．else文を挿入すると，ブロック if文で指定した条件を満足しない場合に，
else文から endif文までの実行文が実行されます．例えば上記のプログラムは，

if (i < 0) then
a = 10
b = 6

else
a = 5
b = 2

endif
c = a*b

と書くことができます．この ifブロックでは，i< 0の場合には a=10，b=6を実行，それ以外の場合に
は a=5，b=2を実行します．
また，条件を満足しない場合に，さらに別の条件を指定したいときには else if文を使います．else

if文も条件はかっこで指定し，その後に thenを書きます．

if (i < 0) then
a = 10
b = 6

else if (i < 5) then
a = 4
b = 7

else
a = 5
b = 2

endif
c = a*b

この場合，i< 0の場合には a=10，b=6を実行，0 5 i< 5の場合には a=4，b=7を実行，それ以外 (i= 5)
の場合は a=5，b=2を実行，となります．else if文による新たな条件は ifブロック内にいくつ入れて
もかまいません．その場合は，“その else if文より以前の条件を全て満足しない場合に，その条件を
満足すれば ”という意味になります．これに対し，else文は最後の一回しか使えません．
数値計算でよく使う代表的な比較条件の書き方を表 2.1に示します．

表 2.1 比較条件の書き方
比較条件記号 記号の意味 使用例

== 左辺と右辺が等しいとき x == 10

/= 左辺と右辺が等しくないとき x+10 /= y-5

> 左辺が右辺より大きいとき 2*x > 1000

>= 左辺が右辺以上のとき 3*x+1 >= a(10)**2

< 左辺が右辺より小さいとき sin(x+10) < 0.5

<= 左辺が右辺以下のとき tan(x)+5 <= log(y)

使用例のように，比較条件を指定する場合には両辺どちらにも計算式を書くことが可能です．
さらに表 2.2の論理演算記号を使えば，これらの条件を論理的につないだ条件や，否定した条件を与
えることもできます．

表 2.2 論理式の書き方

論理演算記号 演算の意味 使用例
“条件 1 ”.and.“条件 2 ” “条件 1 ”かつ “条件 2 ”のとき x > 10 .and. 2*x <= 50

“条件 1 ”.or.“条件 2 ” “条件 1 ”または “条件 2 ”のとき x > 0 .or. y > 0

.not.“条件 ” “条件 ”を満足しないとき .not.(x < 0 .and. y > 0)

例えば，

if (i > 0 .and. i <= 5) then
a = 10.0

else
a = 0.0

endif

と書くと，iが 0より大きく “かつ ” 5以下の時，すなわち，0 < i5 5のときに a= 10，それ以外は
a= 0となります．なお，横着してこのブロック if文を，

if (0 < i <= 5) then ! これはエラー

と書くことはできないので注意しましょう．
do文で繰り返し計算をする場合，“n回ごとに出力する”というように，一定間隔で動作を変更したい
ことがあります．この場合には，表 1.3に示した剰余を計算する組み込み関数 modを利用して条件分岐
をします．例えば，10回に 1回出力するなら，

do m = 1, 10000
......

if (mod(m,10) == 0) print *,m,x,y
......

enddo

のように書きます．プログラムというのは，基本的に全て計算であり，流れのコントロールも計算結果
を使って行わなければなりません．このようなケースで modを使うのは，パターンの一つとして覚えて
おくと良いでしょう．

2.3 無条件ジャンプ — goto 文，exit文，cycle文
プログラムというのは基本的に上から順に実行していくものです．do文を使えば，doブロックで指
定した範囲の実行文を所定の回数繰り返すことができますが，繰り返す範囲は固定されているし，繰り
返しを終了する条件はカウンタ変数の増加で決まります．
これに対し，より一般的にプログラムの流れを変えたい時，例えば途中で計算を中断してプログラム
の最初からやり直す，とか，最後の文に一気に移動して終了する，とかいう時には goto文を使います．
goto文を使えば，指定した行へ強制的に移動することができます．計算機的には，これを “ジャンプす
る ”といいます．
goto文とは，以下のように gotoの後に “文番号 ”と呼ばれる整数を指定した文です．

goto 文番号

これに対し，この goto文でジャンプしたい先の行には，以下のように実行文の前にスペースを 1個
以上空けて文番号を書きます．

文番号 実行文

goto文を使ったプログラム例を以下に示します．

cd = 10
goto 11 ! 文番号 11の行へジャンプ
cd = 50
ab = 20
ij = 1

11 ab = 1000 ! この行へジャンプ

最後の行で ab=1000の前に書かれた 11が文番号です．この例では，最初の cd=10の実行後，cd=50か
ら ij=1までの文は実行されず，直ちに ab=1000が実行されます．すなわち，cd=50，ab=20, ij=1の
文は書いていないのと等価です．このように有無をいわさずジャンプすることを “無条件ジャンプ ”と
いいます．
goto文でバックすることも可能です．例えば，次のように書けば，指定した文番号 22の行と goto文
の間の動作を繰り返し実行します．

cd = 50
22 ab = 200 ! この行へジャンプ
cd = cd + ab - ef
ef = 10
goto 22 ! 文番号 22の行へジャンプ
ij = 1

もっとも，この例ではいつまでたっても goto文の次の ij=1は実行されません．こういうのは “無限
ループ ”と呼ばれ，プログラムエラーの一つです．計算結果に応じて条件分岐し，goto文より下の行へ
ジャンプする別の goto文や，プログラム自体を終了させる stop文を挿入しなければ，プログラムは永
遠に終了しません．
文番号は行の指定に使うだけなので，重複さえしなければどの行にどんな数値を付けても良いのです
が，なるべく下に行くほど大きくなるように数値を選びます．また，文番号を特定の実行文に付けると，
変更する時に付け替えが面倒だと思う時には continue文を挿入します．continue文に動作は無く，文
番号で位置を指定するためだけに用います．例えば，上記のプログラムは，

cd = 50
22 continue ! この行は動作がない
ab = 200
cd = cd + ab - ef
ef = 10
goto 22
ij = 1

と等価です．
goto文は，多用するとプログラムの流れがわかりにくくなるし，無限ループに陥る可能性もあるので
使用はできるだけ避けるべきです．基本的な繰り返しや，条件に応じたジャンプは doブロックと ifブ
ロックでほとんどすべて書くことができます．
doブロックを使って繰り返し計算をする時，条件によって途中で繰り返しを終了したい場合がありま
す．この場合，原理的には goto文を使わなければなりませんが，goto文の使用を極力避けるという方
針から，exit文が用意されています．doブロック中で exit文を実行すると，その doブロックの外に
飛び出して enddo文の直後から実行を継続します．
例えば，要素数 nの配列 a(n)に代入されている値を条件付きで平均するため，

sum = 0
do m = 1, n

sum = sum + a(m)
if (sum > 100) exit

enddo
ave = sum/n

のように書いたプログラムは，次の goto文を使ったプログラムと等価です9．

sum = 0
do m = 1, n

sum = sum + a(m)
if (sum > 100) goto 10

enddo
10 ave = sum/n

なお，doブロックの中から外へのジャンプはできますが，外から中に入るジャンプは禁止されています．
doブロックの途中で実行を中断し，残りの部分をスキップしてカウンタ変数を進める時には cycle文
を使います．例えば，

do m = 1, n
sum = sum + a(m)
if (sum > 100) cycle
sum = sum*2

enddo

のように書いたプログラムは，次の goto文を使ったものと同じです．

do m = 1, n
sum = sum + a(m)
if (sum > 100) goto 10
sum = sum*2

10 enddo

すなわち，cycle文の実行は enddo文にジャンプするのと等価です．enddo文にジャンプすれば，カウ
ンタ変数 mを増加して，m> nかどうかをチェックした後，条件を満足しなければ再び doブロックを最初
から実行することになります．ただし，cycle文は doブロック内部の制御なので次のように ifブロッ
クを使って同じ動作をさせることができます．

do m = 1, n
sum = sum + a(m)
if (sum <= 100) then
sum = sum*2

endif
enddo

この方が，条件に応じた動作を明示している点で良いと思います．goto文を多用しない方がよい，とい
う意味合いと同じで，cycle文を使うとプログラムがわかりにくくなるのであまり使わない方が良いと
思います．
なお，doブロックが多重の場合，exit文や cycle文はその文を含む最も内側の doブロックに対して
の動作になります．より外側の doブロックの外に抜け出たいなどの場合には，do文にラベルを付けて
exit文や cycle文のジャンプ先を指定します．
ラベルを付けるには，do文と enddo文の 2箇所にラベル名を付加します．例えば，

9このプログラムを修正して，合計した要素 a(1)～a(m)の平均値を計算したい場合には，単に nの代わりに mで割るよう
に書き換えるだけでは不完全です．なぜだか考えてみましょう．

–25–



sum = 0
out: do m = 1, n

do l = 1, n
sum = sum + a(l,m)
if (sum > 100) exit out

enddo
enddo out
ave = sum/n

のように書くことができます．“out”がラベル名です．do文に付加するラベルは，ラベル名の最後にコ
ロン (:)を付けて，doの前に書きます．これに対し，enddo文に付加するラベルは，対応する do文と
同じラベル名を enddoの後ろにスペースを入れて書きます．コロンは不要です．このプログラムは，

sum = 0
do m = 1, n

do l = 1, n
sum = sum + a(l,m)
if (sum > 100) goto 10

enddo
enddo

10 ave = sum/n

と等価です．なお，複数の do文にラベルを付けるときは，ラベル名が重複しないようにして下さい．

2.4 プログラミング スマートテクニック (その 2)

本節では doループを使うときのテクニックや，プログラムを読みやすくするためのヒントをいくつか
紹介します．do文や if文がマスターできれば，本格的なプログラムが書けるようになります．大量の
データを処理する計算ができるし，色々な計算処理機能を複合して，条件に応じて使い分けることも可
能です．それに伴ってプログラムが長くなりますから，「文法的に間違いがなければ良い」という考えで
書くのではなく，読みやすいプログラムになるよう心がける必要があります．ここで，“読みやすいプ
ログラム ”というのは，チェックがしやすいプログラムのことです．読みやすく書いておけば，書き間
違いに気付きやすいし，実行時にエラーが出ても早期に誤りを発見することができます．以下のヒント
は，プログラミングの初心者が読むと「細かいことだから無視してもいいや」と感じるかもしれません
が，慣れればそれほど手間ではありません．面倒がらずに心がけましょう．

2.4.1 doループのテクニック
doブロック内部に同じ計算を繰り返す記述があるときには，あらかじめ計算をしておきます．例えば，

do i = 1, 10000
a(i) = b(i)*c*f**2
b(i) = sin(x)*a(i)

enddo

と書くと，繰り返しごとに c*f**2や sin(x)を計算することになりますが，これらは doブロック内部
では変化しないのですから，あらかじめ計算しておくと速くなります．例えば，

cf = c*f**2
sx = sin(x)
do i = 1, 10000

a(i) = b(i)*cf
b(i) = sx*a(i)

enddo

のように書き換えると速くなります．

また，doブロック内で何度も同じ割り算をするときには，逆数を掛けるように書き換えると速くなり
ます．これは，割り算が遅いからです．例えば，

do i = 1, 10000
a(i) = b(i)/c
x(i) = a(i)/10.0

enddo

と書くより，

d = 1.0/c
do i = 1, 10000

a(i) = b(i)*d
x(i) = a(i)*0.1

enddo

と書く方が速くなります．もっとも，プログラムがわかりにくくなるという欠点もあるので，割り算の
箇所が少なく，繰り返し回数がさほど多くない時にはそれほど神経質に変形する必要はないと思います．
配列を使った繰り返し計算をするときは，メモリをできるだけ連続的に読み書きする方が速くなりま
す．このため，多重 doブロックを使って 2次元以上の配列計算をするときは，左の要素から先に進める
ようにします．例えば，

real b(10,100)
integer m,n
do n = 1, 100

do m = 1, 10 ! 左の要素を先に進めると速い
b(m,n) = m*n

enddo
enddo

のように，2次元配列 b(m,n)に計算結果を代入するときは，左側の要素 mに関する doブロックを右側
の要素 nの doブロックの内側に持ってくる方が高速です．これは，内側の doブロックのカウンタ変数
が先に進むので，b(1,1),b(2,1),b(3,1)…のように，メモリの並んでいる順に格納していくからです．
これを，

real b(10,100)
integer m,n
do m = 1, 10

do n = 1, 100 ! 右の要素を先に進めると遅い
b(m,n) = m*n

enddo
enddo

のように書くと，b(1,1),b(1,2),b(1,3)…のように飛び飛びに格納していくので効率が悪くなり，ス
ピードダウンします．
大きな数に小さい数を加えると，小さい数が桁落ちして情報が失われる可能性があります．例えば，1
次元配列 a(i)の合計 sumを計算するプログラムは，

sum = 0
do i = 1, n

sum = sum + a(i)
enddo

で良いのですが，a(i)が全て正数で，nが非常に大きい場合には，合計 sumがだんだん大きくなってい
くので，後の方で加えた a(i)の情報が失われる可能性があります．倍精度実数を使うことを推奨して
いる理由の一つがこれです．この桁落ちを防ぐには，例えば，2個ずつ加えてそれらを別の配列に入れ，
その後それらを同様に 2個ずつ加えていって，...とするのが良いのですが，プログラムが複雑になっ

てしまいます．
ひとつの比較的簡単な解決法は，補助変数を使って，誤差を別に評価しておく方法です [2]．

sum = 0
res = 0
do i = 1, n

res = res + a(i)
tmp = sum
sum = sum + res
tmp = sum - tmp
res = res - tmp

enddo

この方法では，桁落ち分を補助変数 resに保存して別途加えるので，計算精度が上がります．倍精度計
算をしていれば必ずしもこの方法にする必要はありませんが，精度の良い合計計算が必要になった時の
ために覚えておくと良い方法です．

2.4.2 多項式を計算する手法
多項式を計算するときは，掛け算の回数ができるだけ少なくなるように考えます．一般的に良い計算
手法は horner法です．例えば，

　　　　 y = a0 + a1x + a2x
2 + a3x

3 + a4x
4

を，乗算を明示してプログラムにすると，

y = a0 + a1*x + a2*x*x + a3*x*x*x + a4*x*x*x*x

のようになり，10回の乗算が必要です．ところが，これを変形して，

y = a0 + (a1 + (a2 + (a3 + a4*x)*x)*x)*x

と書けば，4回の乗算で計算できます．これが horner法です．
一般的に，n次の多項式，

　　　　 y = a0 + a1x + a2x
2 + · · · + anxn

における係数 a0, a1, a2, ..., anが，下限が 0の 1次元配列，a(0),a(1),a(2),...,a(n)にそれぞれ代入
されている場合には，次のような do文を使って計算することができます．

y = a(n)
do i = n-1, 0, -1

y = a(i) + x*y
enddo

なお，x8 + 1は ((x2)2)2 + 1のように変形すれば 3回の乗算で計算できます．すなわち，多項式によっ
ては horner法より速く計算できる手順が存在することもあります．

2.4.3 do while文と無条件 do文
do文には，2.1節で述べた基本形の他に，条件でループの動作を続けるか終了するかを決める形式も
用意されています．これが do while文です．do while文は以下のような書式です．

do while (条件)
.......
.......

enddo

この場合，do while文の条件を満足しなくなるまで，その doブロック内部を繰り返し実行します．も
し条件を満足しなければ，doブロックは終了して enddo文の次に実行が移ります．例えば，

integer n
n = 100
do while (n > 0)

n = n/2
print *,n

enddo

と書けば，nを 2で割っていって，0になった時点でループが終了します．ここで，nを整数型にしてい
るところがポイントです．実数型だとループがいつ終了するかわかりません．
また，“doのみ”の do文もあります．この場合，doループを終了させる条件がないので，do文と enddo

文で指定した範囲をいつまでも繰り返します．例えば，

do
sum = sum + x**2
if (sum > 100) exit
x = 1.2*x + 0.5

enddo

と書くと，sumが 100を超えるまで計算を続けます．この無条件 do文を使うときは，ブロック内部に適
切な条件分岐で外に出る記述を入れておかないと，無限ループになって永遠に終了しないので注意して
下さい．

2.4.4 0.1を 10回足しても 1に等しくならない
実数型を使って条件分岐をするときに注意しなければならないことがあります．次のプログラムを考
えてみましょう．

real x
integer m
x = 0.0
do m = 1, 20

x = x + 0.1 ! 0.1を繰り返し加える
if (x == 1.0) exit

enddo
print *,’x = ’,x

この doループは，実数型変数 xに 0.1を繰り返し加えていきますが，ループの 10回目で if文の条件「x

が 1.0に等しい」を満足し，exit文によりループの外に出て，print文の出力は 1.0になると予想され
ます．しかし，実際にやってみるとわかりますが，print文の出力は 2.0です．つまり，if文の条件は
満足しない，すなわち，xが 1.0に等しくなることはないのです．なぜでしょうか．
これは，コンピュータが 2進数で計算しているためです．1.4.1節で述べたように，10進数で 123と
表現される数を 2の多項式で表せば，

　　　　 123 = 1 × 26 + 1 × 25 + 1 × 24 + 1 × 23 + 0 × 22 + 1 × 21 + 1 × 20

となり，これが 2進数で表したときに 1111011となる理由でした．コンピュータの数値は全て 2進数な
のですから，小数点以下の数値も 2進数で表さねばなりません．例えば，10進数の 0.123は

　　　　 0.123 = 1 × 10−1 + 2 × 10−2 + 3 × 10−2

のように，10の負べき乗を基本として表された数のことですが，これと同様に，2進数で小数点以下の
数 xを表すには，

　　　　 x = b1 × 2−1 + b2 × 2−2 + b3 × 2−2 + · · ·

となるような 0または 1の係数 b1, b2, b3, · · · を選んで，

　　　　 0. b1b2b3 · · ·

と並べます．コンピュータはこの係数 b1, b2, b3, · · · で小数点以下を表現しています．
さて，コンピュータのメモリで表現できる 2進数の桁 (bit数)には限りがあるので，小数はどこかで
打ち切る必要があります．ところが，2進数の場合，有限小数で表せるのは，2−nの和だけです．例え
ば，0.5 (=2−1)や 0.125 (=2−3)やその和である 0.625は有限の 2進小数で表すことができますが，2−n

の和で表せない数値は無限小数になってしまいます．例えば，10進数の 0.1は，

　 0.1 = 0×2−1 + 0×2−2 + 0×2−3 + 1×2−4 + 1×2−5 + 0×2−6 + 0×2−7 + 1×2−8 + 1×2−9 · · ·

となり，2進数で表せば，0.0001100110011...という循環小数になります．よって，コンピュータにおけ
る 0.1は近似値でしかなく，10回加えても 1に等しくなりません．
このように，実数計算をするときは 10進数と 2進数の違いを考慮してプログラムを書かないと，予想
外の結果になることがあります10．また，同じ少数点以下の数を繰り返し加えるときには，回数を掛け
算した方が精度が上がります．例えば，

real x
integer m
x = 0.0
do m = 1, 100

x = x + 0.1
.......

enddo

と書くよりも，

real x
integer m
do m = 1, 100

x = 0.1*m
.......

enddo

と書く方が，xの精度は上がります．

2.4.5 ブロックを明確にするために字下げする
これまでのプログラム例でも示していますが，doや ifなどのブロック中では字下げ (インデント)を
しましょう．これは，ブロックの範囲が一目でわかるからです．
例えば，ifブロック，

if (n < 0) then
x = 10.0

else if (n < 10) then
x = 1.0

else
x = 0.0

endif

10ちなみに，この理屈で言えば，0.1×10という掛け算も 1.0と等しくならないはずですが，手元のパソコンでやってみると
if文を満足しました．これは，四捨五入のように計算結果を丸めるためです．しかし，丸め方はハードウェアやコンパイラの
仕様に依存するので，それを期待してプログラムを書くのは避けた方が無難だと思います．

や doブロック，

do i = 1, n
b(i) = a(i)
c(i) = a(i)*sin(b(x))

enddo

のように，ブロック内部の文をスペースを入れてずらせておくとブロックの範囲が一目で判断できます．
通常，2～4個のスペース程度で良いと思います．
最近のエディタには，オートインデントといって，改行するとその前の行と先頭がそろうようにカー
ソルが移動する機能があるので，インデントをするのはそれほどの手間ではありません．

2.4.6 文字間や行間は適当に空ける
文中の文字間は適当に空けたほうが読みやすくなります．筆者は “=”と “+”の両側に必ずスペースを
入れることにしています．また，代入文が何行も続くときには “= ”を上下にそろえると読みやすくなり
ます．特に，同じパターンで少しずつ内容が違う文を並べるときには，パターンが並ぶようにスペース
を入れることをお勧めします．以下は，筆者のシミュレーションプログラムの一部です．

wm( 1,m1) = (1-dx)*(1-dy)*(1-dz)
wm( 2,m1) = dx *(1-dy)*(1-dz)
wm( 3,m1) = (1-dx)* dy *(1-dz)
wm( 4,m1) = dx * dy *(1-dz)
wm(11,m1) = (1-dx)*(1-dy)* dz
wm(12,m1) = dx *(1-dy)* dz
wm(13,m1) = (1-dx)* dy * dz
wm(14,m1) = dx * dy * dz

これをスペース抜きで書くと，以下のようになります．

wm(1,m1)=(1-dx)*(1-dy)*(1-dz)
wm(2,m1)=dx*(1-dy)*(1-dz)
wm(3,m1)=(1-dx)*dy*(1-dz)
wm(4,m1)=dx*dy*(1-dz)
wm(11,m1)=(1-dx)*(1-dy)*dz
wm(12,m1)=dx*(1-dy)*dz
wm(13,m1)=(1-dx)*dy*dz
wm(14,m1)=dx*dy*dz

この二つは全く同じことが書いてあるのですから全く同じ結果を出します．しかし，パターンをそろ
えた前者は単なる美的趣味でスペースを入れたのではありません．この例では，どこか１カ所，例えば
dxを dyに書き間違えただけでも正しい結果が得られません．しかも，dxと dyを書き間違えても文法
的なミスにはならず，出力結果を見ても間違いに気づかない可能性があります．よって，できるだけプ
ログラムを書いている段階でミスを取り除いておかねばなりません．
この時，前者のようにパターンをそろえておけば，横方向に一行一行チェックするだけでなく縦方向
にも比較しながらチェックすることができます．色々な角度からチェックすることで，ミスのないプロ
グラムにすることができるのです．
この他，文と文の間に，適当なコメント文や何も書いていない行を挿入すれば，プログラムの流れに
区切りができて読みやすくなります．これは文章を段落に分けるように，プログラムを区分けすると考
えればいいでしょう．

2.4.7 定数は意味のある名前をつけた変数に代入して使う
数学的に決まっている単純な定数 (2倍するとか，1を加えるとか，3乗するとか)の場合以外は，でき
るだけ定数の使用量を減らすほうがいいと思います．プログラムを書いている時には理解していても時

間が経ってから読んだ時に意味がわからなくなる可能性があるからです．
例えば，電子の質量me (= 9.10938356×10−31 kg)と光の速度 c (= 2.99792458×108 m/s)を使って
計算すればmec

2 = 8.1871056497×10−14 Jですが，これを単位とするエネルギー，energy/mec
2を計

算するプログラムを，

ene = energy/8.1871056497e-14

と書くと，後で 8.1871056497×10−14という数字がどこから来たのかわからなくなる可能性があります．
こういう時には多少面倒でも，

me = 9.10938356e-31 ! 電子の質量
cl = 2.99792458e8 ! 真空中の光速
mc2 = me*cl**2
ene = energy/mc2

としておけば，わかりやすいしプログラム内容の記録にもなります．
また，doブロックを 100回繰り返す，とか，10回ごとに出力する，とかいう制御数も，変数にしてお
けば数値の意味が明示されるし，変更もしやすくなります．例えば，

do i = 1, 100 ! 100が 1個
a(i) = b(i)**2

enddo
sum = 0
do i = 1, 100 ! 100が 1個

sum = sum + a(i)
enddo
ave = sum/100 ! 100が 1個

というプログラムにおいて，100という数は配列要素数に依存する共通の数値ですから，以下のように
変数にします．

imax = 100 ! 100を変更するときはここだけで OK
do i = 1, imax

a(i) = b(i)**2
enddo
sum = 0
do i = 1, imax

sum = sum + a(i)
enddo
ave = sum/imax

こうしておけば意味がはっきりするし，100を 200に変更したい時には変更点が１カ所ですみます．こ
のような変数化はプログラムが長くなるほど価値が増します．

演 習 問 題 2

(2–1) 繰り返し出力
i=1,2,3,...,nの整数に対し，i，i2，1/i，

√
iの値を並べて出力するプログラムを作成せよ．なお，整数

と実数の変換に注意すること．

(2–2) 統計計算
1次元配列，a(n)に，1,2,3,...のデータを順に n個代入し，

平　均 Ā=
1
n

n∑
i=1

Ai

標準偏差 σ=

√√√√ 1
n

n∑
i=1

(Ai − Ā)2

を計算するプログラムを作成せよ．
次に，a(n)に 1,3,5,...と奇数を順に n個入れて，同様に平均と標準偏差を計算するプログラムを追加
せよ．

(2–3) 定積分
等間隔 hごとに並んだ座標点 x1(= a)，x2，· · ·，xn(= b)に対して，関数値，f(x1)，f(x2)，· · ·，f(xn)
が与えられているとする．このとき f(x)の定積分を∫ b

a
f(x)dx＝ h

(
1
2
f(x1) + f(x2) + · · · + f(xn−1) +

1
2
f(xn)

)

と近似的に計算する方法を台形公式という．以下に与えた関数と積分範囲の定積分値を台形公式を使っ
て計算せよ．その際，分割数 nが大きくなるほど正確な値に近づくことも確かめよ．

　　　　　 (1) f(x) = sin(x)，a = 0，b = π

　　　　　 (2) f(x) = (x − 2)3，a = 1，b = 5

(2–4) 漸化式
次の漸化式で計算される数値を，それぞれ，配列 a(n)，b(n)に代入し，それぞれの平均と標準偏差
を計算せよ．

(1) ai+1 = 3ai + 2，　　　a1 = −1

(2) bi+1 =
2
3
bi − 4，　　　b1 = 3

(2–5) テーラー展開
テーラー展開の公式から指数関数や三角関数を計算するプログラムを作成せよ．ただし，第 0項から
第 n項までの指数関数と三角関数のテーラー展開の係数は以下の通りである．

ex = 1 +
x

1!
+

x2

2!
+

x3

3!
· · · + xn

n!
=

n∑
i=0

xi

i !

cosx = 1 − x2

2!
+

x4

4!
· · · + (−1)n x2n

(2n)!
=

n∑
i=0

(−1)i x2i

(2i)!

sinx = x − x3

3!
+

x5

5!
· · · + (−1)n x2n+1

(2n + 1)!
=

n∑
i=0

(−1)i x2i+1

(2i + 1)!

次数 nと変数値 xは複数個選んで計算し，組み込み関数で計算した値と比較して，nが大きくなるほ
ど正確な関数値に近づくこと，xが大きくなるほど近似が悪くなることなどを確かめよ．

(2–6) 2次方程式の解 (解の判別付き)

3個の実変数 a，b，cに適当な値を代入し，それから，

ax2 + bx + c = 0

の解を計算して出力するプログラムを作成せよ．まず，解の判別式，D = b2 − 4ac，を計算して，2実
解になった時，重解になった時，虚解になった時でそれぞれ正しい結果を出力するようにせよ．さらに，
その解を ax2 + bx + cに代入して 0に近い値になることを，判別式の値と数値型に応じて計算手順を変
えて確かめよ．

(2–7) 非線形方程式の解法：逐次代入法
x0 = 0として，次のように次々に代入していくと，xnは徐々に cosx = xの解に近づくことがわかっ
ている．

x1 = cos x0

x2 = cos x1

x3 = cos x2

...

xn+1 = cos xn

最初に，この代入を 100回繰り返し，その後で 101回目の値 x101と 100回目の値 x100の差を計算す
るプログラムを作成せよ．
それができたら，収束条件を |xn+1 − xn| < εとし，εを適当に小さい値に定めて (例えば，10−7)，収
束したら xnを出力して終了するプログラムにせよ．このとき，収束するまでの回数も出力せよ．
なお，この問題では配列を使わないこと．

(2–8) 非線形方程式の解法：Newton法
次の公式を繰り返し用いて方程式 f(x) = 0の解を近似的に求めるアルゴリズムをNewton法という．

xn+1 = xn − f(xn)
f ′(xn)

以下の関数 f(x)に対し，適当な初期値 x0 を与えて Newton法で解を求めるプログラムを作成せよ．
また収束条件を |xn+1 − xn| < εとし，εは適当に定めよ．このとき，収束するまでの回数も出力せよ．

　　　　　 (1) f(x) = x − cos(x)

　　　　　 (2) f(x) = x3 − 2

それぞれの問題に対し，得られた結果を f(x)に代入して，答えがどの程度 0に近いかを確認せよ．

(2–9) 常微分方程式 (初期値問題)の解法：サイクロトロン運動
一様な磁場中での荷電粒子の速度 v(t) = (u(t), v(t))は次の運動方程式を満足する．この運動方程式
をオイラー法で解け．ただし，サイクロトロン周波数 ωcは一定とする．

du

dt
= ωcv

dv

dt
= −ωcu

ここで 1回の時間ステップを∆tとして fn = f(n∆t)とした時に，オイラー法とは nステップ目の値
fnと n + 1ステップ目の値 fn+1を使って

df

dt
' fn+1 − fn

∆t
という近似で時間微分を評価する方法であ

る．例えば，x方向は
un+1 − un

∆t
= ωcvn

と近似できるから，これを変形して，un+1 = un +ωcvn∆tとなる．vn+1も同様に変形すれば，オイラー
法とは次の連立の漸化式を使って (un, vn)から (un+1, vn+1)を計算することである．

un+1 = un + ωcvn∆t

vn+1 = vn − ωcun∆t

1周期 2π
ωc
を適当な回数N で分割して∆tを決定し，初期条件 u0 = 1，v0 = 0から出発して，オイラー

法でN 回計算するようなプログラムを作成せよ．そして，その結果得られる 1周期後の速度，uN，vN

と u0，v0を比較せよ．また，10周期後の速度，u10N，v10N とも比較せよ．
次に，オイラー法を少し修正したシンプレクティック法，(vn+1の式に注意！ )

un+1 = un + ωcvn∆t

vn+1 = vn − ωcun+1∆t

のプログラムを作成し，オイラー法の結果と比較せよ．
なお，以上のプログラムは配列を使わないで作成すること．

(2–10) 常微分方程式 (境界値問題)の解法：1次元ポワソン方程式
電荷密度 ρ(x)が与えられているときに電位が満足する 1次元ポワソン方程式

d2V (x)
dx2

= −ρ(x)

を解くには，以下のようにする．
等間隔 hごとに並んだ x座標点 xi = hi, (i = 0～N)に対して，Vi = V (xi)，ρi = ρ(xi)として，上式
の左辺を差分で近似すれば，

Vi+1 − 2Vi + Vi−1

h2
= −ρi

となる．これを座標点に関して書き下せば，

V0 − 2V1 + V2 = −ρ1h
2

V1 − 2V2 + V3 = −ρ2h
2

V2 − 2V3 + V4 = −ρ3h
2

...

VN−2 − 2VN−1 + VN = −ρN−1h
2

になるが，両端の電位 V0 = V (x0)，VN = V (xN )が与えられていれば，これらは V1，V2，· · ·，VN−1に
対する連立 1次方程式になる．
一般に，連立 1次方程式，

aiVi−1 + biVi + ciVi+1 = di （i = 1, 2, . . . , N − 1）

を V0と VN を与えて解くときには，まず，G0 = 0，H0 = V0から出発して，

Gi = − ci

bi + aiGi−1
, Hi =

di − aiHi−1

bi + aiGi−1
（i = 1, 2, . . . , N − 1）

を計算し，次に，Viを以下の式で計算する (計算の方向に注意！ )．

Vi = GiVi+1 + Hi （i = N − 1, N − 2, . . . , 1）

以上の方法を用いて，次の 2種類の電荷密度と境界条件における電位を計算するプログラムを作成
せよ．

(1) ρ(x) = 0　　　　　　　境界条件は，V0 = 0，VN = 1

(2) ρ(x) = sin
(

2πx

Nh

)
　　 境界条件は，V0 = 0，VN = 0

第3章 サブルーチン
ここまで，Fortranの基本的な文法と高速化のテクニック，エラーの出にくいプログラムの書き方な
どを説明しました．ここまでの知識を使うだけで，原理的にはどんなプログラムを作ることも可能です．
プログラムとは計算機に仕事をさせるための手順ですから，それほどバラエティはありません．動作の
繰り返しを書くための do文や，条件分岐の if文が使いこなせれば，計算機の能力のほとんどを引き出
すことができます．
しかし長いプログラムを書くときは，メンテナンスのしやすさを考慮して書かなければなりません．
プログラムが短ければ全体を見ながらチェックや修正ができますが，長くなるとチェックに時間がかか
り，見落としによる修正ミスが起こる可能性も高くなります．見落としを防ぐには何度も見直す必要が
あり，チェック時間はさらに増大します．
このメンテナンスを容易にする手法の一つが “サブルーチン ”です．サブルーチンとはメインプログ
ラムと同じレベルの閉じたプログラムのことで，必要に応じて他のプログラムから呼び出してその機能
を使います．長いプログラムをサブルーチンに分割し，それらをビルディングブロックとして全体を構
成すれば，プログラム作成や修正の作業が楽になります．本章ではこのサブルーチンの作り方と使い方
を説明します．

3.1 サブルーチンの利用目的
“ルーチン”とは，動作開始点と終了点が定義されている閉じたプログラム手順を示す用語です．プロ
グラムを起動したときに最初に動作するのはメインプログラムですが，これを “メインルーチン”ともい
います．“サブルーチン”は，メインプログラムと同様に動作開始点と終了点を持っていますが，メイン
プログラムと異なり，単独で動作することはできません．必ず，他のルーチンに動作開始命令を記述し
て，必要に応じて実行させる必要があります．一つのプログラムにメインプログラムは一つですが，サ
ブルーチンは名前が異なれば複数存在してもかまいません．また，いくつかのルーチンごとにファイル
を作成して別々にコンパイルし，必要に応じて結合 (リンク)させることも可能です．
サブルーチンを利用する目的は主として３つあります．

(1) 複数の場所で同じ計算手順を使用するため
(2) 方程式の解法や行列式の計算などのような定型処理をするため
(3) 長いプログラムを分割してメンテナンスを容易にするため

(1)がサブルーチン本来の使い方です．ある一連の手順を複数の場所で使うとき，それぞれの場所に
同じプログラムを書くと，プログラムが長くなるし，修正の際に何カ所も同じ修正をしなければなりま
せん．このとき，その手順をサブルーチンに置き換えれば，プログラムは短くなるし，チェック作業も
短縮されます．

(2)は (1)を発展させたものです．複雑な数学公式や汎用性のある数値計算アルゴリズムをプログラム
中に直接記述するときは，プログラム中で宣言された変数や配列を使って書きます．このため，同じ計
算手順を別のプログラムで使いたくてもそのまま使えるとは限りません．そこで計算手順をサブルーチ
ンにして，計算に必要な数値を与える部分と計算結果の数値を受け取る部分だけが外部から見えるよう
にしておきます．そうすれば，受け渡し部を書き換えるだけで，複雑な計算手順を色々なプログラムか
ら利用することができます．このようにブラックボックス化したサブルーチンは，他の人に提供したり，
他の人からもらって利用することも可能で，そのような汎用性のあるサブルーチンを集めたものが，“ラ
イブラリ ”です．
さて，計算機シミュレーションのように多数の解析手順を複合したプログラムを書く場合には，(3)の
目的が重要になります．様々な物理過程の計算や，入出力に関する作業など，ある程度まとまった内容
ごとにサブルーチンにするのです．文章でいえば，章や節に分割するようなものです．このため，メイ

ンプログラムには，それぞれのサブルーチンを呼び出す文と，必要ならそれらを繰り返すための do文だ
けを書いておきます．シミュレーションプログラムは，サブルーチンという部品を使って組み立てるも
のだと考えればいいでしょう．
サブルーチンに分割しておけば，プログラムのメンテナンスが楽になります．これは，それぞれのルー
チンが独立しているので，プログラムを修正したときの影響や，エラーが発生する原因がルーチン内に
限定されるからです．

3.2 サブルーチンの宣言と呼び出し
サブルーチンは subroutine文で開始を宣言し，end subroutine文で終了します．すなわち，次の
ような構造にします．

subroutine subr1
implicit none
real a,b
integer i
.......
.......

end subroutine subr1

先頭の subroutine文で，subroutineの後に指定した文字の並び (この例では subr1)を “サブルーチ
ン名 ”といいます．また，最後の end subroutine文には subroutine文と同じサブルーチン名を指定
します．見てわかるように，メインプログラムと同じ構造です．サブルーチン内部のプログラムの書き
方も基本的にメインプログラムと同じで，subroutine文の次に implicit noneを書き，非実行文を上
方に集約して，その後に実行文を書きます．サブルーチンはそれ自体で閉じているので，実行文中で使
う変数や配列は，基本的にその内部で宣言しなければなりません．ただし，異なるルーチン間で共用可
能な変数を別途用意することは可能です．これについては，3.7節で説明します．
上の例では，subroutine文にサブルーチン名しかありませんが，これを “引数なしサブルーチン ”と
いいます．これに対して，サブルーチン名の後に，かっこで囲んだ変数リストを付加することができ，
これを “引数ありサブルーチン ”といいます．以下に引数ありサブルーチンの一例を示します．

subroutine subr2(x,m,y,n)
implicit none
real x,y,a,b,z(10)
integer m,n,i,k
.......
.......

end subroutine subr2

リスト中の変数を “引数 ”といいます．この例では，x,m,y,nが引数です．引数は，サブルーチンとそ
のサブルーチンを使うルーチンの間で数値を受け渡すために使います．引数もサブルーチン内部の変数
なので，必要な型に応じた宣言をしなければなりません．
サブルーチンを動作させてその機能を使うときは，call文を使ってそのサブルーチンを指名します．
このため，サブルーチンの動作を指定することを，“サブルーチンを呼び出す ”とか “コールする ”とい
います．call文は以下のような形式です．

call サブルーチン名 ! 引数なしサブルーチン用
call サブルーチン名 (数値または変数のリスト) ! 引数ありサブルーチン用

例えば，先ほど出てきた，引数なしと引数ありの二つのサブルーチンを使うときは，それぞれ，次の (1)
と (2)のようにコールします．

real z
integer m
call subr1 !........................ (1)
m = 21
call subr2(10.0,100,z,m*5+1) !..... (2)

1.2.1節で説明したように，計算式は計算結果の数値を動作命令に与えるので，call文の引数には，(2)
の一番右のように計算式を与えることもできます．
サブルーチンは単独では動作できないので，メインプログラムが別途必要です．例えば，次のような
セットを構成して初めて一つのプログラムが完成します．

program stest1
implicit none
real x,y
x = 5.0
y = 100.0
call subr(x,y,10) ! サブルーチンの呼び出し
print *,x,y

end program stest1

subroutine subr(x,y,n)
implicit none
real x,y
integer n
x = n
y = y*x

end subroutine subr

ここでは，メインプログラムを先に，サブルーチンを後に書きましたが，逆でもかまいません．サブ
ルーチンが複数存在する場合も，ルーチンを記述する順番は実行結果とは無関係です．サブルーチンの
中から別のサブルーチンをコールすることも可能です．
上記のプログラム実行の流れを図 3.1に示します．

program stest1

  implicit none

  real x,y

  x = 5.0

  y = 100.0

  call subr(x,y,10)

  print *,x,y

end program stest1

subroutine subr(x,y,n)

  implicit none

  real x,y

  integer n

  x = n

  y = y*x

end subroutine subr

プログラム領域
プログラム領域

メインプログラム
サブルーチン

実
行
の
流
れ

図 3.1 サブルーチンの呼び出しと実行の流れ

図のように，メインプログラムの call文でサブルーチンを指名すると，サブルーチンの一番最初の実
行文に動作が移り，サブルーチン内部の動作が完了すると，コールしたメインプログラムに戻って，そ
の call文の次の文から実行を継続します．
条件に応じて，途中でサブルーチンの処理を打ち切って戻るときには return文を用います．

subroutine subr(x,y,m,n)
implicit none
real x,y
integer m,n
.......

if (m < 0) return
.......

end subroutine subr

この例では，m< 0のときにサブルーチンの処理が終了し，コールしたルーチンに戻ります．ここでreturn

文の代わりに stop文を用いれば，プログラム自体が終了します．

3.3 ローカル変数と引数
サブルーチン内部で宣言した変数や配列は，メインプログラムや他のサブルーチンから独立していま
す．すなわち，同じ名前を使っても全く別の変数です．例えば，

program stest1
implicit none
real x,y
x = 10.0
y = 30.0
call subr1

end program stest1

subroutine subr1
implicit none
real x,y
print *,x,y

end subroutine subr1

と書いても，サブルーチン subr1中の print文によって出力される xと yはメインプログラムで代入し
た 10.0と 30.0ではなく，全く無関係な数字です．逆に言えば，他のルーチンでの宣言を気にせずに変数
や配列の名前を決めることができます．このルーチン内部でのみ有効な変数を “ローカル変数 ”といい
ます．
上記のプログラムを期待通り働かせるために，コール側ルーチンの数値をサブルーチンに伝えるのが
引数です．例えば，上記のプログラムを次の subr2のように書き直せば，サブルーチン中の print文で
出力される xと yは，メインプログラムと同じ 10.0と 30.0になります．

program stest2
implicit none
real x,y
x = 10.0
y = 30.0
call subr2(x,y)

end program stest2

subroutine subr2(x,y)
implicit none
real x,y
print *,x,y

end subroutine subr2

引数は，数値の受け渡しをするための窓口に過ぎないので，コール側の引数とサブルーチン側の引数
の変数名を同じにする必要はありません．次のサブルーチンに置きかえても全く同じ動作をします．

subroutine subr2(a,b)
implicit none
real a,b
print *,a,b

end subroutine subr2

外部からの影響を受けたり，外部に影響を与える，という性質を持つことを除けば，引数もローカル
変数です．すなわち，コールしたルーチンからその詳細は見えません．見えるのは，引数という窓口の
並びだけです．このため，引数ありサブルーチンを使うときに重要なポイントは，

(1) 引数の数
(2) 対応する引数の型

が call文と subroutine文とで一致していなければならないことです．例えば，

program stest3
implicit none
real z
integer n
z = 200.0
n = 21
call subr3(10.0,z**2,100,n*5+1)

end program stest3

subroutine subr3(x,y,m,n)
implicit none
real x,y
integer m,n
print *,x,y,m,n

end subroutine subr3

というプログラムでは，表 3.1のような対応になっています．

表 3.1 サブルーチンの引数対応
call文 subroutine文 数値型
10.0 x 実数
z**2 y 実数
100 m 整数
n*5+1 n 整数

ここで注意すべきなのは，第 1引数の xは実数型なので実定数 10.0を与え，第 3引数の mは整数型
なので整定数 100を与えていることです．もし，第 1引数に同じ意味だろうと思って 10という整定数
を与えると，実行時エラーになります．
サブルーチン内部の実行文で，引数に数値を代入すると，call文の対応する引数に与えた変数にその
数値が代入されます．例えば，

program stest4
implicit none
real x,y,p
x = 10.0
y = 30.0
call subr4(x+y,20.0,p)
print *,x,y,p

end program stest4

subroutine subr4(x,y,z)
implicit none
real x,y,z
z = x*y

end subroutine subr4

と書くと，サブルーチン subr4中の xはコール側の x+y，すなわち 40.0であり，サブルーチン中の yは
コール側の 20.0なので，サブルーチン中の zには x*yの計算結果である 800.0が代入されます．このと
き，zが引数なので，call文の対応する位置にある変数 pに 800.0が代入され，call文の次の print文
では x，y，pとして，10.0，30.0，800.0が出力されます．
このようにサブルーチンの引数に数値を代入することで，call文の引数変数に代入される数値を “戻
り値 ”といいます．戻り値を使えば，サブルーチンの動作で得られた結果をコール側で受け取ることが
できます．ただし，call文の引数には定数を与えたり計算式を書いてもよい，と述べましたが，戻り値
を指定する引数は別で，必ず変数か配列にしなければなりません．理由を次節で説明します．

3.4 間接アドレスを用いたルーチン間におけるデータの受け渡し
ルーチン間のつながりをもう少し詳しく考えてみましょう．各ルーチンは基本的にプログラム領域と
データ領域から構成されています．プログラム領域とは文字通りプログラム命令が記録されている領域
のことであり，データ領域とは変数や配列のために用意されたメモリ領域のことです．プログラム的に
は，変数や配列の宣言文がデータ領域の指定を意味しています．

program stest1

  implicit none

  real x,y

  x = 5.0

  y = 100.0

  call subr1

  print *,x,y

end program stest1

subroutine subr1

  implicit none

  real x,y

  x = 10.0

  y = 25.0

end subroutine subr1

プログラム領域

データ領域
x    5.0

y  100.0

プログラム領域

データ領域
x   10.0

y   25.0

メインプログラム サブルーチン

参
照

参
照

参
照
で
き
な
い

図 3.2 プログラム領域とデータ領域

変数や配列が各ルーチンごとに宣言されなければならないのは，データ領域が各ルーチンそれぞれに
付属しているからです．このため，図 3.2のように異なるルーチンで同じ名前の変数を宣言しても，そ
れらは別のデータ領域に所属しています．これがローカル変数です．ローカル変数は，そのルーチン内
部からしか参照することができません．
このため，あるルーチンで計算した数値を別のルーチンで使うには特別な手続きが必要になります．
これが引数です．引数を使えば，コール側のルーチンとサブルーチンの間でデータをやりとりすること
ができます．では，具体的にどうやってデータの受け渡しをしているのでしょう．
データをサブルーチンに渡す手段として，最も単純に考えられるのは，引数の数値を直接サブルーチ
ンの変数に代入することです．図 3.3を見て下さい．図のように，コール側の引数 xと yの内容 (この例

では 5.0と 100.0)が，サブルーチンの引数として宣言された変数，aと bに代入されています．この方式
は，引数の値を直接渡して呼び出す，という意味で “ call by value ”と呼ばれています．“ call by value ”
は，コール側からサブルーチン側への値の引渡しについては問題ありません．C言語はこの方式を採用
しています．

program stest2

  implicit none

  real x,y

  x = 5.0

  y = 100.0

  call subr2(x,y)

  print *,x,y

end program stest2

subroutine subr2(a,b)

  implicit none

  real a,b

  print *,a,b

  b = a+25.0

end subroutine subr2

プログラム領域

データ領域
x    5.0

y  100.0

プログラム領域

データ領域
a    5.0

b  100.0

メインプログラム サブルーチン

call by value では
数値を直接代入する

送
り
返
せ
な
い

こ
ち
ら
へ
代
入

図 3.3 C言語の “ call by value ”を使ったデータ渡しでは値を送り返せない

しかし，この方式では，サブルーチン側で作成したデータをコール側に戻すことはできません．サブ
ルーチンは，コール側から送られてきた “値 ”はわかるのですが，それ以上の情報がないので，コール
側のどこにデータを代入すればいいかわからないからです．図 3.3のように，サブルーチン中の代入文，
b=a+25.0，でサブルーチンの変数 bに 30.0という値を代入しても，サブルーチンのデータ領域の bに
代入されるだけで，メインプログラムのデータ領域には影響がありません．
この問題を解決するために，Fortranは “ call by value ”ではなく，“ call by reference ”という方式を
採用しています．“ call by reference ”とは，変数の内容ではなく，変数の存在場所 (メモリアドレス)を
サブルーチンに伝えて呼び出す方式です．
コンピュータのメモリには，“番地 ”と呼ばれる通し番号がついています．メモリアドレス (あるいは
単にアドレス)とはこの番地のことです．プログラムの実行中に，あるメモリのデータを読み書きする
ときは，そのメモリのアドレスを指定して行います．例えば，簡単な代入文，

a = 1500.0
b = a

を考えてみましょう．この結果，bのメモリに 1500.0という値が書き込まれますが，代入文 b=aにおい
て，a(アドレスを [103]とする)という変数からデータを取って来て bに代入するという流れは，

aのアドレス指定 → [103]番地のメモリ内のデータ (1500.0)の呼び出し
→ bに代入

となります．流れを図に描けば図 3.4のようになります．このような指定したアドレスのメモリに入っ
ているデータを読み書きする方式を “直接アドレス ”といいます．
これに対し，あるアドレス (AD1)のメモリに入っているのが別のメモリのアドレス (AD2)で，アド
レスAD1を指定して，アドレスAD2のメモリに入っているデータを読み書きする方式を “間接アドレ
ス ”といいます．Fortranでは，“ call by reference ”でサブルーチンを呼び出すので，引数にはコール側
ルーチンのメモリアドレスが入っています．このため，サブルーチン内で引数の値を読み書きするとき

101:

102:

103:

104:

105:

1500.0a:103 b

データ領域

....

図 3.4 直接アドレスによるメモリ参照

は間接アドレスを使います11．例えば，

program stest3
implicit none
real a
a = 1500.0
call subr(a)

end program stest3

subroutine subr(s)
implicit none
real b,s
b = s

end subroutine subr

というプログラムを考えてみましょう．サブルーチン subrの中の代入文 b=sによって，変数 bのメモ
リの値は 1500.0になりますが，引数変数 sには aの内容である 1500.0ではなく，メインプログラムの
変数 aのアドレス [103]が入っています．このため， b=sという代入文で，変数 s (メモリアドレスを
[106]とする)からデータを取ってきて変数 bに代入する流れは，

sのアドレス指定 → [106]番地にあるメモリアドレスデータ (103)の呼び出し
→ [103]番地にあるデータ (1500.0)の呼び出し
→ bに代入

となります．図示すれば図 3.5のようになります．

101:

102:

103:

104:

105:

106:

107:

1500.0

103

a:103 b

s:106

....

データ領域

図 3.5 間接アドレスによるメモリ参照

間接アドレスによる読み書きは直接アドレスよりも時間がかかります．よって，b=sの代入文におい
て，代入する値を保持している右辺の変数 sが間接アドレス指定の場合にはメリットがありません．し
かし，代入される左辺の変数 bが間接アドレス指定の場合には，この仕組みを利用することで，コール
側ルーチンの変数にサブルーチン側のデータを代入することができます．

11配列を実現するのにも間接アドレスが使われています．例えば，a(10) は a(1) から数えて 10 番目のメモリです．よっ
て，配列の先頭要素 a(1)のアドレスをメモリに記録し，それに 9 (=10 − 1)を加えて間接アドレスを用いれば，a(10)のメ
モリを読み書きすることになります．

program stest2

  implicit none

  real x,y

  x = 5.0

  call subr2(x,y)

  print *,x,y

end program stest2

subroutine subr2(a,b)

  implicit none

  real a,b

  print *,a,b

  b = a+25.0

end subroutine subr2

プログラム領域

データ領域
x    5.0

y   30.0

プログラム領域

データ領域
a  xのアドレス

b  yのアドレス

メインプログラム サブルーチン

call by reference では
変数 x,y のアドレスを送る

間接アドレスで送る

参
照

図 3.6 Fortranでは “ call by reference ”によりデータを送り返すことができる

例えば，図 3.6のようなプログラムを考えます．サブルーチンの引数変数 bに a+25.0の結果である
30.0を代入すると，bにはメインプログラムの変数 yのアドレスが入っているので，間接アドレス指定
により yに 30.0が代入されます．
戻り値を利用するときは，対応する引数にコール側変数のアドレスを与えなければなりません．これ
が変数または配列を指定しなければならない理由です．戻り値指定の引数に定数や計算式を与えると実
行時エラーになります．

3.5 配列を引数にする場合
配列を call文の引数にするときは，配列名を引数にすることも，配列要素を引数にすることも可能
です．1.3.2節で述べたように，配列名は配列の先頭要素アドレスを示すので，配列名を引数にすること
と，その配列の第 1要素を引数にすることは同じ意味になります．例えば，real a(10)と宣言した 1次
元配列に対して，

　　　　 call sub(a)　と　 call sub(a(1))

は同じ動作をします．
これを受けるサブルーチンでの宣言は，サブルーチン内部でそのデータをどう使うかで決めます．一
例を示します．

subroutine sub(x)
implicit none
real x ! 単一変数宣言
x = 10.0

end subroutine sub

この例では引数 xが単一変数として扱われています．このため，call文で指定した a(1)だけがサブルー
チン内部で利用でき，他の要素は使えません．これは，call sub(a)のように，配列名を与えた場合で
も同じです．
これに対し，

subroutine sub(x)
implicit none
real x(10) ! 配列宣言
integer i
do i = 1, 10
x(i) = i

enddo
end subroutine sub

のように引数 xを配列宣言すれば，あたかもコール側ルーチンの aという配列がサブルーチン内部の配
列 xになったかのように取り扱うことができます．ここで “あたかも ”という言葉を使ったのは，サブ
ルーチンでの宣言文の書き方によってはそうならないことがあるからです．例えば，サブルーチンでの
配列宣言 x(10)の要素数 10はコール側と同じ数にする必要はありません．x(100)のように大きくても，
x(1)のように小さくてもかまわないのです．これは，引数 xが配列の先頭要素アドレスを保持している
1個の変数にすぎないからです．サブルーチンでの引数配列の宣言は，配列の形状 (次元や下限値)を明
示するためのダミーにすぎません．
このため，コール側ルーチンで宣言した要素数とサブルーチンで宣言した要素数を一致させる必要は
ありません12．与えられた配列をどう宣言するかは，サブルーチン内部の計算の都合に合わせて，サブ
ルーチン側で決めることができます．よって，汎用性のあるサブルーチンにするには，配列のアドレス
以外に，配列の要素数も引数にしてサブルーチンに伝える必要があります．
配列の形状さえわかればいいのですから，サブルーチンで引数配列を宣言するときは，数字の代わり
に “ * ”を書くことができます．例として，要素数 nの 1次元配列 aのデータを，同じ長さの 1次元配列
bにコピーするプログラムを考えてみましょう．最も単純な答えは，以下のようなものです．

subroutine copy(a,b,n)
implicit none
real a(*),b(*)
integer n,i
do i = 1, n
b(i) = a(i)

enddo
end subroutine copy

このように，サブルーチンの引数配列 aや bの宣言を “ * ”にすることで，要素数が不定の 1次元配列で
あることを明示しています13．

“*”による宣言は 2次元以上の配列でも使うことができますが，制限があります．例えば，a(*,*)と
いう形の宣言はできません．なぜなら，2番目の要素を進めるのは 1番目の要素が最大値，すなわち宣
言した上限数に達したときなので，1番目の要素が不定では情報不足で進められないからです．このた
め，“ * ”が使えるのは，一番右側の要素数のみです14．例えば，real a(3,*)と宣言すると，第 1要素
数が 3の 2次元配列として計算されます．
しかし，これではどんな 2次元配列でも使えるサブルーチンを作ることはできません．そこで，“整
合配列 ”と呼ばれる機能が用意されています．整合配列とは subroutine文の引数の中にある整数型変
数を利用して宣言した形の引数配列のことです．例えば，以下のように宣言することができます．

12要素数だけではなく，次元さえ一致させる必要はありません．コール側が 2次元配列で，サブルーチンでの宣言が 1次元
配列でも動作は可能です．例えば，上記のサブルーチン copyは，2次元配列でも使用できます．これは，1.3.2節で述べたよ
うに 2次元配列もメモリ上では 1次元的に並んでいるからです．ただし，要素数を与える引数 nには全要素数を指定する必要
があります．例えば，a(10,10)と宣言された配列ならば，nに 100 (=10×10)を与えます．

13なお，“ * ”だけ書くと配列の下限は 1です．もし下限を変更したいときには下限を別途指定します．例えば，配列 aの下
限を 0にする場合は，real a(0:*)と宣言します．

141.3.2節の配列の順番を与える公式が一番右側の要素数に依存しないことを思い出して下さい．

subroutine copy2d(a,b,m,n)
implicit none
real a(m,n),b(m,n)
integer m,n,i,j
do j = 1, n
do i = 1, m

b(i,j) = a(i,j)
enddo

enddo
end subroutine copy2d

この例では，mと nが引数の中にあるので，これらを使って引数の配列 aと bを宣言することができる
のです．a(m,n)のような単純な宣言だけではなく，a(0:2*m,n-1)のような，下限指定や計算式を使っ
た宣言も可能です．
このサブルーチンの使用例を次に示します．サブルーチンの引数 a，bにどんな要素数の 2次元配列を
与えても，コール側ルーチンにおける配列宣言と同じ要素数を m，nに与えれば，正しく動作します．

program stest1
implicit none
real a(10,20),b(10,20),c(100,200),d(100,200)
.......

call copy2d(a,b,10,20)
call copy2d(c,d,100,200)

end program stest1

なお，配列宣言の際には，コロンを付加して下限を指定することができますが，これをサブルーチン
に引き継ぐにはサブルーチン側でも同じ下限指定をする必要があります．例えば，

program stest2
implicit none
real a(-1:100)
call sub(a,a,100)

end program stest2

subroutine sub(x,y,n)
implicit none
real x(-1:n),y(*)
integer n
x(10) = 10.0
y(10) = 10.0

end subroutine sub

と書いた場合，メインプログラムと同じ下限指定をした配列 xでは x(10)がメインプログラムの a(10)

に対応するので，a(10)に 10.0が代入されます．しかし，下限指定をしていない配列 yでは下限が 1で
すから，a(-1)から数えて 10番目の a(8)に 10.0が代入されます．任意の下限を持つ配列に対して動作
するプログラムにするには，下限の数値も引数変数にした整合配列を使います．

3.6 関数副プログラム
sin(x)のように，引数を使って内部で計算した結果を計算式中で使うことができる “関数 ”を自作す
ることもできます．これを “関数副プログラム ”といいます．関数副プログラムは，サブルーチンとほ
とんど同じ構造をしていますが，以下のような違いがあります．

(1) subroutineの代わりに functionを書く
(2) 関数名を変数として宣言し，それに計算結果 (関数値)を代入しなければならない

これ以外は，引数ありサブルーチンと同じです．引数に関する注意もサブルーチンと同じで，戻り値
を与える引数を含めることもできます．関数副プログラムとは，引数以外に戻り値を 1個持つサブルー
チンの変種だといえます．この戻り値が関数値になります．
関数副プログラムの一例を示します．

function square(x)
implicit none
real square,x ! 関数名の型宣言
square = x*x ! 関数名に値を代入する

end function square

このように，function文で開始し，end function文で終了します．また，関数名 squareを実数型
宣言し，引数 xの 2乗を代入して終了しています．すなわち，この例は引数 xに実数型の数値を与える
と，x2を関数値として与える関数になります．
関数副プログラムを使うことは，“関数名”という変数を使うことであると考えます．このため，作成
した関数を使用するルーチンでも関数名を型宣言する必要があります．例えば，上例の squareを使う
には，

program ftest1
implicit none
real x,y,square ! 使用する関数名を型宣言する
x = 5.2
y = 3.0*square(x+1.0) + 50.5
print *,x,y

end program ftest1

のように，squareという関数名を関数副プログラムでの宣言と同じ型で宣言しておかなければなりませ
ん．関数副プログラム中で宣言した関数名の型と，その関数を使用するルーチンで宣言した関数名の型
が異なる場合には実行時エラーになります．

3.7 モジュールを使ったグローバル変数の利用
ローカル変数のおかげで各ルーチンの独立性は保たれますが，引数でしかルーチン間のデータ受け渡
しができないのは不便です．引数は，かっこを使った並びで指定するので，受け渡す変数が多いと書く
のが大変です．しかも並びが重要なので，順番を１カ所間違えただけでもエラーが発生します．そもそ
も引数による受け渡しはアドレス情報を経由したものなので，コール側とサブルーチン側で完全に同じ
変数であるという保証はありません．

3.1節で，計算機シミュレーションでは計算内容に応じてサブルーチンを作成すると述べましたが，こ
の用途においては，ルーチン間で共用する変数が多数必要です．しかし，共用変数を全て引数にするの
は効率が悪く，エラーも発生しやすくなります．
そこで，ルーチン内部でのみ意味を持つ “ローカル変数 ”に対して，“グローバル変数 ”が用意されて
います．グローバル変数とは，どのルーチンから参照しても共通した値を保持している変数のことで，
Fortranでは，モジュールと use文の組み合わせで利用可能です15．
モジュールとは，変数やサブルーチンなどを集めて一つのパッケージにしたもので，module文で開始
して end module文で終了します．サブルーチンと同じ構造をしていることからわかるように，モジュー
ルの記述は，メインプログラムやサブルーチンと同レベルです．例えば，

15モジュールや use文は Fortran90から採用された比較的新しい仕様です．それ以前では common文でグローバル変数を実
現していました．しかし，common文には不便な点が多いので本書では説明を省略します．

module data1
integer nmin,nmax
real tinitial,amatrix(20,30)

end module data1

のように書きます．先頭の module文で，moduleの後に指定した文字の並び (この例では data1)を “モ
ジュール名”といいます．また，最後の end module文には module文と同じモジュール名を指定します．
モジュールはサブルーチンや関数副プログラムと異なり，プログラムに記述しただけではその内容を
利用することができません．利用するルーチンの先頭に，use文を使ってモジュール名を指定してその
利用を宣言する必要があります．use文は以下のような形式です．

use モジュール名

use文は implicit文よりも前に書かなければなりません．モジュールを使ったプログラムの一例を
以下に示します．

module global
real xaxis,yaxis

end module global

program stest4
use global
implicit none
xaxis = 5.0
yaxis = 100.0
call subr4
print *,xaxis,yaxis

end program stest4

subroutine subr4
use global
implicit none
print *, xaxis,yaxis
yaxis = 25.0

end subroutine subr4

モジュールの中で宣言された変数や配列は，メインプログラムやサブルーチンとは独立して存在したデー
タ領域にあり，use文で利用を宣言すれば，どのルーチンからでも参照することができます．このプロ
グラムのメモリイメージを図示すれば，図 3.7のようになります．
モジュールは，名前が異なれば，一つのプログラムに複数作成することも可能であり，一つのルーチン
が複数のモジュールを利用することも可能です．また，モジュールの中に use文を書いて別のモジュー
ルの変数を利用することもできます16．ただし，モジュールは use文で利用を宣言するより前で (プロ
グラムの上方で)定義されている必要があります．このため，通常はこの例のように全てのルーチンよ
り前に記述しておきます．
モジュールで宣言されている変数は，use文を書くだけで利用できるという利便性がありますが，ルー
チン内で明示的に宣言されていないので，不用意に使ってしまう可能性があります．そこで，次のよう
な only句を使って，ルーチン内で必要な変数だけに宣言を限定することができます．

use モジュール名, only : 変数 1，変数 2，...

配列の場合には配列名だけを変数の位置に記述します．
例えば，前の例で，

16一つの応用例として parameter変数の利用があります．3.8.2節を参照して下さい．

program stest4

  use global

  implicit none

  xaxis = 5.0

  yaxis = 100.0

  call subr4

  print *,xaxis,yaxis

end program stest4

subroutine subr4

  use global

  implicit none

  print *,xaxis,yaxis

  yaxis = 25.0

end subroutine subr4

プログラム領域

データ領域
xaxis     5.0

yaxis    25.0

プログラム領域

メインプログラム サブルーチン

module global

参
照

参
照

図 3.7 モジュールで宣言されているグローバル変数の参照

use global, only : yaxsis

のように書くと，yaxsis以外の変数 (xaxsis)は，このルーチンでは宣言されていないのと同等です．
宣言されていない変数は，ローカル変数として別途宣言することも可能です．
グローバル変数は，多用するとルーチンの独立性が失われてしまいます．基本的にはローカル変数で
プログラムを作り，ルーチン間で共用する必要がある変数のみグローバルにしてください．また，グロー
バル変数は広い範囲で存在する可能性があるので，意味のある長めの名前を付けます．これは，1.2.3節
で述べた “大域的に有効な変数名の付け方 ”です．

3.8 プログラミング スマートテクニック (その 3)

サブルーチンを利用するときは，変数の使い分けが一つのポイントです．これまでローカル変数とグ
ローバル変数の使い分けについて説明しましたが，ローカル変数にも 2種類あり，これを使い分ければ
サブルーチンの動作をより細かく制御することができます．ここでは，ローカル変数の使い分け方と，
プログラムのメンテナンスをさらに容易にしたり汎用性を持たせるための文法などを紹介します．

3.8.1 拡張宣言文とそれを用いたローカル変数の使い分け
これまで変数や配列の宣言に使っていた宣言文は，

型指定 変数 1, 変数 2，...

という形式でした．ここで，“型指定”には “integer”とか，“real”のような数値型を記述し，“変数”
には変数名か，配列名とその要素数を記述します．この宣言文は，データ領域におけるメモリの確保と
いう意味もあります．
さて，Fortran90から宣言文の記述形式が拡張されました17．拡張宣言文を使うと，メモリを確保する
と同時に初期値を代入したり，メモリの特性を指定することができます．拡張宣言文は以下の形式です．

17本書では，拡張された宣言文のことを “拡張宣言文 ”と呼んで，旧来の型指定だけの宣言文と区別します．

型指定［,属性 1，属性 2，...］:: 変数 1［=数値 1］,変数 2［=数値 2］，...

ここで，角かっこ，“［ ”と “］ ”は，この中が省略可能であるという意味で使っているだけなので，角
かっこ自体は書かないで下さい．拡張宣言文を使用するときは，型指定部と宣言する変数や配列の間に
コロン 2個 “ :: ”を書く必要があります．また，“属性 ”には宣言する変数の特性を指定するための予約
語を記述します．属性も数値も省略して，単に型指定と変数の間に “ :: ”を書いただけの宣言文は，書
かずに宣言した旧来の書式と同じ意味になります．
属性を書かずに，単に数値を代入した形で宣言すると，その変数に指定した数値を代入して，プログ
ラムの動作が開始することを意味します．例えば，

integer :: imax=10, jmax=100
real :: xx=1.0, yy=2.0

のように宣言すると，それぞれの変数に指定された値を代入した状態で動作が開始します．ただし，こ
の宣言文中での数値代入は一度だけです．サブルーチン中で数値代入した変数を宣言しても，コールす
るたびに数値が代入されるのではありません．このため，その変数を実行文で変更すると，その変更し
た結果が残ります．例えば，

program stest1
implicit none
call subr1
call subr1
call subr1

end program stest1

subroutine subr1
implicit none
integer :: n=1
print *,n ! コールするたびに nは増加する
n = n + 1

end subroutine subr1

というプログラムでは，メインプログラムでサブルーチン subr1が 3回コールされていますが，サブルー
チン中の print文の出力は，1回目が 1，2回目が 2，3回目が 3，となります．宣言文における代入は，
プログラム開始時の初期値だと考えて下さい．
さて，上記のサブルーチンの動作を考えるとき，一つ注意しなければならないことがあります．3.4節
で，ローカル変数は各ルーチンのデータ領域に所属しているという説明をしましたが，もう少し詳しく
言うと，サブルーチンのデータ領域には 2種類あります．一つは，サブルーチンがコールされた時点で
一時的に生成されるメモリ領域で，もう一つは，サブルーチンの呼び出しに関係なく常に確保されたメ
モリ領域です．本書では，前者を “一時メモリ領域 ”と呼び，後者を “固定メモリ領域 ”と呼ぶことにし
ます．旧来の宣言文で宣言したローカル変数は，一時メモリ領域に所属します18．一時メモリ領域は，サ
ブルーチンを呼び出すたびに場所や内容が変わる可能性があるので，同じサブルーチンを 2回コールし
た場合，1回目のコールでローカル変数に代入した値が，2回目にコールした時点でそのまま残っている
とは限りません．
例えば，

18これに対し，メインプログラムの変数とモジュール内のグローバル変数は，宣言文の記述にかかわらず，全て固定メモリ
領域に所属します．

program stest2
implicit none
call subr2(1)
call subr2(2)

end program stest2

subroutine subr2(n)
implicit none
integer n
real x,y
if (n == 2) print *,x,y 　　! この出力値は不定
x = 10.0
y = 100.0

end subroutine subr2

のようなプログラムを書いたとします．1回目の call文，call subr2(1)で，ローカル変数 xと yに，
それぞれ 10.0と 100.0という値が代入されますが，2回目の call文，call subr2(2)を実行したとき
に，print文で出力した xと yが 10.0と 100.0になる保証はないのです19．ローカル変数に代入した値
を保証するには，固定メモリ領域に所属させなければなりません．拡張宣言文を使って初期値を代入し
たローカル変数は，一時メモリ領域ではなく，固定メモリ領域に所属します．最初のサブルーチン例，
subr1の変数 nが，コールするごとに 1ずつ増加した値を出力するのは，初期値 1を代入して宣言する
ことで固定メモリ領域に所属させたためです．
数値を代入せずに，ローカル変数を固定メモリ領域に所属させるには，変数に save属性を指定しま
す．例えば，上記のサブルーチン subr2を以下のように書き換えれば，2回目のコールで 10.0と 100.0
が出力されます．

subroutine subr2(n)
implicit none
integer n
real, save :: x,y ! save属性を指定
if (n == 2) print *,x,y
x = 10.0
y = 100.0

end subroutine subr2

変数の初期値が未定のときや，ローカル配列を固定メモリ領域に所属させたいときには save属性を使っ
て下さい．また，初期値を代入する場合でも，固定メモリ領域に所属させることを積極的に利用するので
あれば，次のように save属性を付加して，固定メモリ領域にあることを明示した方が良いと思います．

integer, save :: n=1

拡張宣言文の属性にはsaveの他にも色々ありますが，ここでは配列を宣言するときに便利なdimension

属性を紹介しましょう．例えば，10×10の 2次元実数型配列を 5個用意するとき，通常の宣言では，

real a(10,10), b(10,10), c(10,10), d(10,10), e(10,10)

のように書きますが，dimension属性を使えば，これを次のように書くことができます．

real, dimension(10,10) :: a, b, c, d, e

すなわち，dimension属性に配列の形状 (次元や要素数)を書いておけば，宣言する変数の位置には，配
列名を記述するだけになります．属性は複数付加することができるので，dimension属性に save属性
を加えて宣言することも可能です．

19コンパイラによっては，全てのローカル変数を固定メモリ領域に所属させるものがあり，このプログラムでもうまくいく
場合があります．しかし，文法的には保証されていないのですから，当てにするべきではありません．

3.8.2 parameter変数の利用
配列を宣言するときに便利なのが，parameter属性を指定した変数，parameter変数です．parameter

変数には必ず値を代入しなければなりません．例えば，

integer, parameter :: imax=10, jmax=200

のように書くと，imaxと jmaxが parameter変数になります．
さて，配列を宣言するとき，その要素数は整数定数で指定しなければなりません．このため，配列の
長さを変更するときには，宣言した要素数に関連した数値を全て変更する必要があり，手間がかかるだ
けでなく，見落としたり，書き間違える可能性があります．例えば，

real a(100),b(100) ! 100が 2カ所
integer i
do i = 1, 100 ! 100が 1カ所

a(i) = i*i
b(i) = a(i)**2

enddo

というプログラムにおいて，配列要素数の 100は，宣言文だけでなく，do文の終了値にも入っています．
このため，配列要素数を変更するときは 3カ所変更しなければなりません．このとき，配列宣言の要素
数を parameter変数に代入する形で用意しておくと，配列要素数をその parameter変数で置きかえる
ことができます．例えば，上のプログラムは次のように書き換えることができます．

integer, parameter :: imax=100
real a(imax),b(imax)
integer i
do i = 1, imax

a(i) = i*i
b(i) = a(i)**2

enddo

このプログラムならば，imaxに代入する数値を変更するだけで，配列要素数の変更は完了です．
parameter変数を含んだ計算式を別の parameter変数への代入値にしたり，配列宣言の要素数に使う
こともできます．これらは，その計算式の結果で宣言したことに相当します．例えば，

integer, parameter :: imax=100, imax2=imax**2
real a(imax-1),b(0:imax*2),c(-imax2:imax2)

という宣言は，以下の宣言文と同等です．

integer, parameter :: imax=100, imax2=10000
real a(99),b(0:200),c(-10000:10000)

ただし，他の parameter変数に代入するときは，代入される変数 (この例では imax2)が，代入する計算
式に使う変数 (この例では imax)より後で宣言されなければなりません．
実は，parameter属性を指定した変数は，“変数＝数値”の形式で変数名と数値の対応関係を示している
だけで，メモリとしての実体はありません．この対応関係を使った “変数”から “数値”への置き換えは，
コンパイルの段階で行われるので，非実行文である宣言文でも使えるのです．逆に言えば，parameter

変数に実行文を使って代入することはできません．例えば，

integer, parameter :: imax=100
real a(imax),b(imax)
imax = 5 ! これはエラー

と書いた場合，“ imax=5 ”はコンパイルエラーです．これは，プログラムに “ 100=5 ”と書いたことに等
しいのですから当然だと言えます．

このように，parameter変数を使えば配列要素数の変更が楽になりますが，それでもルーチンごとの
ローカルな宣言では，変更が必要になったときに使用している全てのルーチンで parameter変数の設定
値を変更しなければなりません．そこで，

module global_param
integer, parameter :: imax=300, jmax=200

end module global_param

のように，parameter変数を記述したモジュールを用意して，共有するルーチンで use宣言をするのが
良いと思います．use宣言はモジュールの中でも使うことができるので，以下のように記述することが
可能です．

module global_param
integer, parameter :: imax=300, jmax=200

end module global_param

module mod_array
use global_param ! モジュール内でも use宣言可能
real abc(imax,jmax),cd(jmax*2-1) ! グローバル配列宣言

end module mod_array

program mtest1
use mod_array ! use global_paramは不要
implicit none
integer km(imax) ! ローカル配列宣言
.......

ここで，あるモジュールの中で別のモジュールを use宣言した場合，前者のモジュールを use宣言し
たルーチンでは，後者のモジュールも合わせて use宣言したことになります．上記のメインプログラム
mtest1ではモジュール mod_arrayしか use宣言をしていませんが，mod_arrayの中で use宣言をして
いるモジュール global_paramも同時に宣言したことになり，その中にある変数 imaxもメインプログ
ラムで利用可能になります．
なお，次のようにメインプログラムに global_paramの use宣言を加えてその使用を明示してもエ
ラーではありません．

program mtest1
use global_param ! 書いてもエラーではない
use mod_array
implicit none
integer km(imax)
.......

これは，同じモジュールの use宣言を重複して書いてもエラーにはならないからです．
parameter変数は，数学定数や物理定数を用意するのにも便利です．筆者は次のような定数の入った
モジュールを作成して，必要なサブルーチンから利用するようにしています．

module constants
real, parameter :: pi=3.141592653589793, pi2=pi*2
real, parameter :: clight=2.99792458e8, hplanck=6.626070040e-34
real, parameter :: emass=9.10938356e-31, pmass=1.672621898e-27
real, parameter :: echarge=1.6021766208e-19, emc2ev=emass*clight**2/echarge

end module constants

parameter変数は不用意に書き換えられることがないので，このような定数の定義にはうってつけで
す．また，コンパイル時に数値の置き換えをするのですから，代入という実行動作が不要になり，わず
かですが計算時間の短縮になります．

3.8.3 include文
プログラムが完成して，後は問題に応じて parameter変数を変更するだけになると，parameter変数
の変更のためだけにプログラムファイルを修正するのは煩わしいし，変更するときに誤ってプログラム
の内容を書き換えてしまうリスクもあります．プログラムを変更すると，プログラムファイルの日付が
変更されるので，いつの時点で完成したプログラムかもわからなくなります．これらは，特に他人にプ
ログラムを提供するときに問題になります．
そこで，parameter変数の部分だけをプログラム本体から分離して保存し，必要に応じて結合すると便
利です．結合する一つの手段に include文の利用があります．include文とは，以下のように include

の後に，ファイル名を文字列で記述した文です．

include ’ファイル名’

include文を書くと，コンパイラはその位置に “ファイル名 ”で指定したファイルの内容を挿入し
たプログラムを生成して，それをコンパイルします．例えば，3.8.2節のモジュール global_paramを
“ global.inc ”というファイルに書き込んで保存しておけば，次のように記述することができます．

include ’global.inc’ ! ここに global.incの内容が挿入される
module mod_array

use global_param
real abc(imax,jmax),cd(jmax*2-1)

end module mod_array

program mtest1
use mod_array
implicit none
integer km(imax)
.......

include文が指定するファイルの内容は，それを include文の位置に挿入したときに，プログラム全
体が正しく動作すればいいので，モジュールやサブルーチンのような完結したプログラムである必要は
ありません．また，同じファイルを一つのプログラムの複数の場所で include指定することもできます．
例えば，

integer, parameter :: imax=300, jmax=200

だけをファイルに書いて，サブルーチンごとに include指定することも可能です．

3.8.4 サブルーチンや関数副プログラムを引数にする方法
本章で説明したサブルーチンや関数副プログラムは，変数や配列などの数値を引数に持つものでした．
これに対し，サブルーチンや関数副プログラムの名前を引数に持つサブルーチンや関数副プログラムを
作ることも可能です20．例えば，方程式 f(x) = 0を解くサブルーチンにおいて，関数 f(x)を計算する
関数副プログラムの名前も引数に加えておけば，コールするときにユーザーが関数を指定することがで
きる，より汎用的なサブルーチンになります．
サブルーチン名を引数にしたサブルーチンを作るのは簡単で，引数並びの中にサブルーチン名を入れ
るだけです．例えば，

20ここではサブルーチンと関数副プログラムを合わせて “外部副プログラム ”と呼びます．

subroutine subrout(subr,xmin,xmax,n)
implicit none
real xmin,xmax,dx,y
integer n,i
dx = (xmax-xmin)/n
do i = 0, n

call subr(dx*i+xmin,y) ! 引数 subrを使った call文
print *,i,y

enddo
end subroutine subrout

のように書くことができます．この例では，subrが引数としてのサブルーチン名で，これをコールする
文を書くことができます．関数副プログラムを引数にする場合も同様ですが，関数名の型宣言は必要で
す．例えば，次のように書きます．

subroutine funcout(func,xmin,xmax,n)
implicit none
real func,xmin,xmax,x,dx,y ! 引数 funcが関数なので型宣言をする
integer n,i
dx = (xmax-xmin)/n
do i = 0, n

x = dx*i + xmin
y = func(x)**3 ! 引数 funcを関数として使う
print *,x,y

enddo
end subroutine funcout

このように，外部副プログラムを引数に持つサブルーチンを作るのは簡単ですが，このサブルーチン
を使用するときは，「引数が変数ではなく，外部副プログラムである」という宣言が必要です．この宣言
は external文で行います．例えば，

subroutine sub(x,y)
implicit none
real y,x
y = 2*sin(x) + cos(x**2)

end subroutine sub

function fun(x)
implicit none
real fun,x
fun = sin(x)**3

end function fun

というサブルーチンや関数副プログラムを引数にして，上記のサブルーチン (subroutや funcout)を
コールするときは，

program test_func
implicit none
external fun,sub ! external文を使って宣言する
call subrout(sub,0.0,3.0,10)
call funcout(fun,0.0,3.0,10)

end program test_func

のように書きます．external文による宣言がないと，subや funという単語が “変数 ”と見なされて，
型宣言されていないというエラーになります21．external文は非実行文なので，変数の型宣言と同様
に，全ての実行文より前に書く必要があります．

21ちなみに，サブルーチン funcoutに与えている関数名 funは型宣言をしていませんが，これはエラーになりません．ただ
し，このプログラム内で funを関数として計算に使う場合には型宣言をする必要があります．

–56–



さて，なぜこの程度で外部副プログラムを引数にできるのか，しくみを簡単に説明しましょう．3.4節
で述べたように，Fortranでは引数に変数を与えるとき，変数の内容を与えるのではなく，そのアドレ
スを与えます．変数にアドレスがあるのと同様に，プログラムにもアドレスが付いています．これは，
プログラムが保存されているメモリ番地のことです．goto文での無条件ジャンプや if文での分岐，サ
ブルーチンコールなど，必要に応じて実行の流れを変えるときには，プログラムのアドレスを指定して
ジャンプします．プログラムも変数も，場所は違いますが，コンピュータのメモリ上にあるという意味
では同じです．よって，引数に変数を与えるのも，サブルーチン名を与えるのも，メモリアドレスを与
えるという意味では同じであり，特に書式を変更する必要はありません．
しかし，プログラム名が指定するアドレスと変数や配列名が指定するアドレスはメモリ領域が異なる
ので区別しなければなりません．外部副プログラムを引数として持つサブルーチンの内部では，その副
プログラムを使用している記述があるはずなので，それを判定材料として引数がプログラムアドレスか
変数アドレスかを区別しています．一方，このサブルーチンを使う call文側では，外部副プログラム
の名前しか引数に書けないので，それだけで変数か外部副プログラム名かを判定することはできません．
この区別のために用意されているのが external文です．external文を使って宣言された名前は，サブ
ルーチン名または関数副プログラム名と判定されてコンパイルされます．なお，external文による宣
言を忘れてコンパイルエラーになったとき，エラーを防ぐために subや funを realや integerなどで
型宣言すると，エラーは出ず，コンパイルもリンクも成功します．しかし，実行しても正常な動作はし
ないので注意が必要です．
なお，関数を引数に持つサブルーチンに sinや expのような組み込み関数名を与えてコールするとき
は，external文の代わりに intrinsic文で宣言します．例えば，

program test_sin
implicit none
intrinsic sin
call funcout(sin,0.0,3.0,10)

end program test_sin

のように書きます．これは，Fortranにおける組み込み関数が宣言をしなくても使える特別なものだから
です．しかし，それなら宣言せずとも使えるようにしてくれればいいと思うのですが，intrinsic宣言を
しておかないと「sinという変数が宣言されていない」というエラーになります．このあたりは Fortran
コンパイラの仕様の問題だと思います．

3.8.5 interface文
Fortranは古くからある言語なので，文法的にゆるい記述がいくつか残っています22．implicit none

を付けなければ暗黙の変数宣言と見なされ，宣言しなくても変数が自由に使えるというのはその一つで
す．これ以外に，サブルーチンを呼び出すための call文の記述において引数のチェックをしない，と
いう問題があります．このため，call文で引数の数を間違えたり引数の数値型を間違えたりしてもコン
パイルエラーにはならず，実行時に正常な動作をしないという形で不具合が現れます．
この欠点を補うため，最近の Fortranでは，interface文を使ってプログラム中に外部副プログラ
ムの引数の形状を宣言しておき，外部副プログラムを使用する際の引数チェックをコンパイラにさせ
ることができます．interface文は，interfaceと end interfaceではさんだブロックで，その中に
subroutine文または function文，その引数宣言 (関数副プログラムの場合は関数値も)，および最後の
end文だけを取り出して入れておきます．例えば，3.8.4節のサブルーチン subroutや関数 funを使用
する場合，以下のように書くことができます．

22キーボードやモニターがなかった時代，プログラムを書くのは時間がかかる作業でした．このため，記述量を減らすため
に暗黙の宣言などがあったのだと思います．C言語は比較的新しい言語なので，変数宣言は必ず行わなければならないし，関
数の引数チェック機能も含まれています．

program test_interface
implicit none
external sub
interface

subroutine subrout(subr,xmin,xmax,n)
external subr ! 外部副プログラムは external文で宣言する
real xmin,xmax
integer n

end subroutine subrout
function fun(x)

real fun,x ! 関数の場合は関数値も宣言する
end function fun

end interface
call subrout(sub,0.0,3.0,10)

! call subrout(0.0,3.0,10) ! これを入れるとエラーになる
print *,fun(10.0)

! print *,fun(10) ! これを入れるとエラーになる
end program test_interface

この例のように，interface文中には複数のサブルーチンや関数副プログラムの宣言を書くことができ
ます．また，interface文中に関数副プログラムの宣言を入れておくと，関数名の型宣言が入っている
ので，その関数名の型宣言を別途する必要はなくなります．なお，外部副プログラムを引数にする場合
には，この例のように external文で宣言しておきます．
このプログラム例で，二つ目の subroutの call文を含めてコンパイルするとエラーになります．な
ぜなら，二つ目の call文は引数の数が足らず，一番左の引数が外部副プログラムではなく実数になっ
ているからです．また，二つ目の関数 funを使用した print文を含めてもエラーです．これは，xが実
数指定なのに整定数 10を与えているからです．
interface文で引数の型を宣言しておくと，引数変数に値を代入した形でサブルーチンをコールする
ことができます．この場合，引数を記述する順番は自由です．例えば，

program test_interface
implicit none
external sub
integer m
interface

subroutine subrout(subr,xmin,xmax,n)
external subr
real xmin,xmax
integer n

end subroutine subrout
end interface
m = 5
call subrout(n=m**2,subr=sub,xmax=3.0,xmin=0.0)

end program test_interface

と書くことができます．
ただし，3.5節において，「配列名を引数にすることと，その配列の第 1要素を引数にすることは同じ意
味になる」と言いましたが，interface文を使う場合は区別する必要があります．なぜなら，interface
文は引数の型だけでなく形状のチェックも行うからです．このため，interface文中の宣言が 1変数で
あれば 1変数か定数を，1次元配列で宣言したならば 1次元配列名を，というように引数の形状と一致し
た変数や配列を call文の引数に与えなければなりません．さもなくば，コンパイルエラーになります．
interface文を使うと，より安全なプログラムになります．大きなプログラムを作る時や，引数が多い
サブルーチンを利用するときは，interface文の利用をお勧めします．必要な部分だけをコピー・ペー
ストするだけなので，それほど手間はかかりません．

演 習 問 題 3

(3–1) 2次方程式の解 (サブルーチン利用)

3個の実変数 a，b，cを引数とし，それから，

ax2 + bx + c = 0

の 2解 x1，x2，および判別式D =
√

b2 − 4acを計算して戻り値にするサブルーチンを作成せよ．ただ
し，2実解になる場合は，x1，x2をそれぞれ戻り値変数 x1と x2に代入し，2虚解になる場合は，実部
を x1に，虚部を x2に代入するようにせよ．
次にそれを使って問題 (2–6)の解答をサブルーチンを使う形に修正せよ．

(3–2) ヘロンの公式 (関数副プログラム利用)

三角形の三辺の長さを a，b，cとすると，その三角形の面積 Sは次式（ヘロンの公式）で表される．

S =
√

p (p − a)(p − b)(p − c)

ただし，p =
a + b + c

2
である．

a，b，cの数値を引数とし，この公式を使って三角形の面積を計算して関数値にする関数副プログラ
ムを作成せよ．
次にそれを使って問題 (1–2)の解答を関数副プログラムを使う形に修正せよ．

(3–3) 回路計算 (サブルーチン利用)

N 個の抵抗R1, R2, R3, · · · , RN に対し，このN 個を直列にした時の合成抵抗RsとN 個を並列にし
た時の合成抵抗Rpを同時に計算して戻り値として返すサブルーチンを作成せよ．ただし，N 個の抵抗
値は，要素数 nの 1次元配列 R(n)で表すとする．
次に，それを使って問題 (1–7)の解答をサブルーチンを使う形に修正せよ．

(3–4) 3次元ベクトルと電磁気学 (サブルーチン利用)

3次元ベクトルA = (A1, A2, A3)を要素数 3の 1次元配列 a(3)で表すとする．電場ベクトルE，磁
場ベクトルBおよび荷電粒子の速度ベクトル vを配列で表して，これらの配列を引数として与えると内
積 v •E，および外積ベクトル v × Bを計算して戻り値とするサブルーチンを作成せよ．なお戻り値の
外積ベクトルも配列とする．これを使って問題 (1–8)の解答をサブルーチンを使う形に修正せよ．

(3–5) 統計計算 (サブルーチン利用)

要素数 nの 1次元配列，a(n)を引数とし，

平　均 Ā=
1
n

n∑
i=1

Ai

標準偏差 σ=

√√√√ 1
n

n∑
i=1

(Ai − Ā)2

を計算して戻り値とするサブルーチンを作成し，問題 (2–2)の解答をサブルーチンを使う形に修正せよ．

(3–6) 非線形方程式の解法：Newton法 (サブルーチン利用)

問題 (2–8)の Newton法を用いて方程式の解を計算するプログラムをサブルーチンを使う形に修正せ
よ．まず，変数値 xを与えると関数値 f(x)とその微分値 f ′(x)を返すようなサブルーチンを作成する．
それを使ってNewton法で解を求めるプログラムにすればよい．

(3–7) 行列計算
n行 n列の行列 aij を 2次元配列 a(n,n)の要素 a(i,j)で表すとする．2個の行列 aij と bij から行列
の和，cij，および行列の積，dij を計算するサブルーチンを作成せよ．この時，整合配列を用いて与える
配列の次元が自由に変えられるようにすること．
ここで，行列の和と積の定義は以下の通りである．

cij = aij + bij

dij =
n∑

k=1

aikbkj

(3–8) 行列式
u(3)，v(3)，w(3)という 3つの 1次元配列データから行列式

Det =

∣∣∣∣∣∣∣
u(1) v(1) w(1)
u(2) v(2) w(2)
u(3) v(3) w(3)

∣∣∣∣∣∣∣
を計算する関数副プログラムを作成せよ．

(3–9) 3元連立方程式の解法
3元 1次の連立方程式，

a11x1 + a12x2 + a13x3 = b1

a21x1 + a22x2 + a23x3 = b2

a31x1 + a32x2 + a33x3 = b3

を解くプログラムのサブルーチンを作成せよ．
クラメールの公式によれば，答えは以下のようになる．

x1 =

∣∣∣∣∣∣∣
b1 a12 a13

b2 a22 a23

b3 a32 a33

∣∣∣∣∣∣∣
D

, x2 =

∣∣∣∣∣∣∣
a11 b1 a13

a21 b2 a23

a31 b3 a33

∣∣∣∣∣∣∣
D

, x3 =

∣∣∣∣∣∣∣
a11 a12 b1

a21 a22 b2

a31 a32 b3

∣∣∣∣∣∣∣
D

ただし，

D =

∣∣∣∣∣∣∣
a11 a12 a13

a21 a22 a23

a31 a32 a33

∣∣∣∣∣∣∣
である．
プログラムは，3行 3列の行列 aij を 2次元配列 a(3,3)の要素 a(i,j)で表し，biを b(i)という 1次
元配列で表し，xiを x(i)という 1次元配列で表して，サブルーチンの引数は，aと bと xとする．行列
式の計算には，問題 (3–8)で作成した関数副プログラムを利用すればよい．

(3–10) モンテカルロ法
世の中の動きは，運動方程式のような微分方程式を用いて確実に予測できるとは限らない．例えば，
さいころを投げて次にどの数字が出るかで進み方を決めるすごろくのように，次の瞬間に何が起こるか
わからない事象を元にして未来の様子を計算する方法をモンテカルロ法という．
計算機でさいころの代わりをするものは “乱数 ”と呼ばれている．Fortranには，サブルーチン

random_numberが用意されているので，これを用いれば乱数を使った計算ができる．例えば，1個の乱
数を使うときには

call random_number(x)

と書けば，変数 xに 0 5 x < 1の範囲の実数が代入される．このサブルーチンの戻り値 xは予測がつか
ないので，同じ呼び出しを繰り返しても、それ以前に代入された値と等しい数値が代入される確率は非
常に小さい．
乱数を 1回で複数用意したいときには配列を使う．例えば，配列 a(10)に 10個の乱数を代入すると
きは次のように書けばよい．

call random_number(a(1:10))

モンテカルロ法を使って円周率 π を計算してみよう．2個の乱数 xと y を発生させて 2次元座標点
(x, y)を作る．この乱数の座標点をN 個作り，その中で原点から半径 1の円より内側にある点の数N1

を数えて p = 4N1

N
を計算する．座標点数N が増えると，この pが πに近づくことを確かめよ．

第4章 データ出力の詳細とデータ入力
プログラムは計算をさせるだけでは意味がありません．結果を出力して初めて完結します．これまで
は，最も単純な print文でとりあえず画面に表示する方法を使ってきましたが，本章では好みに応じて
数値の出力形式を変える方法や，ファイルに保存する方法について説明します．さらにデータを入力す
ることでプログラムを変更せずに動作を変える方法についても説明します．

4.1 データ出力先の指定
これまでもたびたび登場しましたが，最も単純な出力命令は，print文です．

print form，データ 1，データ 2,　...

print文で出力すると，画面 (正確には標準出力)にデータが表示されます．formは出力形式を指定す
るためのものですが，とりあえず “*”を書いておけば，データの数値型に応じた標準形式で表示します．
例えば，

print *，x, y(i), x**2+5, ’ abc = ’, 10

のように，データの位置には，変数や配列を与えても良いし，計算式や定数を与えることも可能です．
また，文字列を適当に入れて，データの意味を並記することもできます．
画面ではなく，ファイルに出力するときは write文を用います．print文との違いは，出力先を指定
する整数値 ndと出力形式指定の formをかっこで記述することです．

write(nd, form) データ 1，データ 2,　...

ndを装置番号といいます．装置番号は，出力ファイルを識別する数字で，任意に選ぶことができます．
また，変数や整数値を結果とする計算式を与えることもできるので，条件に応じて出力先を変更するこ
とも可能です．ndに適当な整数を与えて write文を実行すると，“ fort.nd ”という名のファイルに出力
されます23．例えば，

write(20,*) x,y,z

と書けば，x,y,zの値が “ fort.20”という名前のファイルに出力されます．この例のように，formに “*”
を与えれば，print文と同じく，データの数値型に応じた標準形式で出力します．
なお，原則として nd = 10にして下さい．これは，Fortranコンパイラの仕様によって，1桁の数値は
予約されている可能性があるからです．例えば，nd = 6にすれば print文と同じ標準出力の指定になる
ので，write(6,*)と書けば，ファイルへの出力はなく，画面に表示されます．

4.2 配列の出力，do型出力
出力文において，データの位置に配列名を書けば，全要素が並んで出力されます．例えば，

real a(4)
do i = 1, 4

a(i) = i
enddo
print *,a

23“ fort ”の部分はコンパイラに依存しますが，多くは fortのようです．ファイル名は 4.6節で説明する open文を使えば変
更することができます．また，コンパイラによっては，open文を使わなくても装置番号に対応する環境変数の指定で，任意の
名前を持つファイルに変更することが可能です．

というプログラムを実行すると，

1.00000000000000 2.00000000000000 3.00000000000000 4.00000000000000

のように，配列要素が先頭から順に並んで出力されます．このとき，配列要素が多いと適当に改行を入
れて出力されます．よって，この単純な出力は配列要素が少ないとき以外はあまりお勧めできません．
例えば，2次元配列を，

real a(3,3)
do j = 1, 3

do i = 1, 3
a(i,j) = i*j

enddo
enddo
print *,a

のように出力すると，a(1,1),a(2,1),a(3,1),a(1,2),...,a(3,3) という並びで 9個の要素が出力さ
れます．そこで，次のように do文を使って出力する方がいいでしょう．

do j = 1, 3
do i = 1, 3
print *,a(i,j)

enddo
enddo

しかし，print文や write文は，1文ごとに改行をするので，複数の print文を使って，横に続けて
書くことはできません24．よって，この例の出力では全部で 9行必要です．行数を減らすために，1行に
行列の 1行を出力するには，次のように書かなければなりません．

do i = 1, 3
print *,a(i,1),a(i,2),a(i,3)

enddo

しかし，この例のように列の数がわかっているときは良いのですが，列数が多いときや変数で指定し
たいときには不便です．そこで do型出力が用意されています．do型出力とは，

　　　　 (データ 1,データ 2,...,整数型変数 = 初期値, 終了値)

のような，かっこで囲んだ形式です．これをデータの位置に記述すれば，do文のように，整数型変数が
初期値から終了値まで 1ずつ増えていき，その変数で指定されるデータが横に並んで出力されます．do

型出力を複数並べたり，通常のデータと並べて書くこともできます．例えば，

print *,(a(i,j),j=1,3)

は，さきほどの “ print *,a(i,1),a(i,2),a(i,3) ”と書くことと同等です．また，

print *,5*x,(i,a(i),b(i),i=1,3)

は，“ print *,5*x,1,a(1),b(1),2,a(2),b(2),3,a(3),b(3) ”と書くことと同等です．
do文と同様に，3番目の数として増分値を付加することもできます．

　　　　 (データ 1,データ 2,...,整数型変数 = 初期値, 終了値, 増分値)

例えば，

24formatを利用すれば続けて書くことは可能です．4.3節を参照して下さい．

print *,(a(i),i=10,1,-2)

は，“ print *,a(10),a(8),a(6),a(4),a(2) ”と書くことと同等です．
do型出力は多重にすることもできます．例えば，

print *,((a(i,j),j=1,3),i=1,3)

は，“ print *,a(1,1),a(1,2),a(1,3),a(2,1),a(2,2),a(2,3),a(3,1),a(3,2),a(3,3) ”と書くこ
とと同等です．この例のように，多重のときには内側の整数型変数 (この例では j)が先に進みます．
なお，do型出力の整数型変数は，do文のカウンタ変数に相当します．このため，do型出力の入って
いる出力文を doブロックの中に入れるときには，do文のカウンタ変数とも重複しないようにしなけれ
ばなりません．

4.3 formatによる出力形式の指定
標準出力形式 (“*”指定)で実数を画面に出力すると，有効数字 15桁の数字を使って表示されます．こ
のため，あまり多くのデータを横に並べることはできないし，結果の確認だけなら，それほど有効数字
は必要ありません．また，標準形式の出力には 1行の出力文字数に制限があるため，出力数が制限を超
えると自動的に改行されてしまいます．このため，同じ形式の出力を繰り返しても，行ごとに小数点の
位置が違うこともあり，大量に出力するのには向きません．
これらの問題は，出力形式 (format)を指定することで解決することができます．これまで “*”を書い
ていた formの位置に formatを指定すれば，小数点以下の桁数を小さくしたり，必要に応じて数字と数
字の間にスペースを空けたり，改行を入れたりすることができます．また，自動改行されることがない
ので，幅広く出力することも可能です．

formatの指定方法は 2通りあります．まず，format文による指定です．一例を示します．

real x,y
integer n
x = 1.5
y = 0.03
n = 100
print 600,x,y,n ! 600は format文の文番号

600 format(’ x = ’,f10.5,’ y = ’,es12.5,’ n = ’,i10)

最後の文が format文です．format文は，サブルーチンの引数のように，出力形式の指定をリストに
してかっこで囲み，先頭に文番号を付けます．この文番号を print文や write文の formに指定すれば，
その formatにしたがってデータが整形されて出力されます．この例では，600が form指定です．文番
号は重複できないので，ルーチン内では format文ごとに異なる数字をつけなければなりません25．
複数の print文や write文が同じ format文を指定するのは可能です．例えば，次のように共用する
ことができます．

real x,y,u,v
integer n,k
.......

print 600,x,y,n
write(20,600) u,v,k
600 format(’ x = ’,f10.5,’ y = ’,es12.5,’ n = ’,i10)

format文は，書式を示すだけで，コンピュータの動作はありません．このため，記述の位置には無関係
で，非実行文より後でさえあれば，指定する print文や write文より前に書くことも可能です．

25文番号の付け方に決まったルールはありませんが，“ format用である”という意味をどこかに入れておいた方が良いでしょ
う．この例で 600を使っているのは，“ 6 ”が昔の出力装置番号だったころの慣習です．

–64–



formatを指定するもう一つの方法は，文字列を使って formの位置に直接 formatの内容を記述するこ
とです．例えば，上記の formatを print文や write文に埋め込んで，

print "(’ x = ’,f10.5,’ y = ’,es12.5,’ n = ’,i10)",x,y,n
write(20,"(’ x = ’,f10.5,’ y = ’,es12.5,’ n = ’,i10)") u,v,k

のように書くことができます．このとき，“format”の文字は不要ですが，両端のかっこは必要です．な
お，Fortranでの文字列は「’」で囲むのが基本ですが，「"」も使えるので，format内部に「’」が入っ
ているときには「"」で囲みます．

formatを出力文中に書き込む方法は，文番号を必要としない点は良いのですが，出力文が長くなるし，
同じ formatを何度も使うときには不便です．そこで，文字列変数を利用して書く方法があります．文字
列変数の利用方法については 5.3節で説明します．
出力形式の指定方法を上記の formatを例にして説明しましょう．まず，format中の文字列はそのま
ま出力されるので，必要に応じて適宜挿入します．この例では，’ x = ’や，’ y = ’は，スペースも
含めてそのまま出力されます．
次に，出力文中のデータ値の並びに対し，それぞれの出力形式を指定する “編集記述子”を選んで，前
から順に記述します．この例では，f10.5，es12.5，i10，が編集記述子で，print文の並びに対し，

print 600, x , y , n
↓ ↓ ↓

600 format(’ x = ’,f10.5,’ y = ’,es12.5,’ n = ’,i10)

という対応で出力形式を指定しています．文字列と編集記述子で指定したデータ値は，その並び順に出
力されます．よって，この print文を実行したときの出力は以下のようになります．

x = 1.50000 y = 3.00000e-02 n = 100

出力形式を指定する編集記述子の主要なものを表 4.1に示します．データの数値型に応じた編集記述
子を使用しないと正しい値が出力されないので注意して下さい．なお，表 4.1で斜体文字 (w, m, d)は整
定数で指定します．また，編集指定の文字 (Fや ESなど)を指定数 (wなど)と区別するために大文字で
書きましたが，小文字でも同じ意味です．例えば，i10は整数型値を幅 10文字で出力することを意味し，
f10.5は実数型値を幅 10文字，小数点以下 5桁で出力することを意味しています．このため，

real x,y
integer m,n
x = 1.5
y = 0.03
m = 100
n = 10
print "(f10.5,f10.5,i10,i10.5)",x,y,m,n

というプログラムの出力は，

1.50000 0.03000 100 00010
+----+----+----+----+----+----+----+----+

となります．2行目の目盛りは位置を確認するために書いたものですが，10文字の中に右寄りで出力さ
れているのがわかります．なお，出力文字数が指定の幅 wを越えると，“ ***** ”のように “ * ”が w個
出力されます．

表 4.1 主要な編集記述子
編集指定 数値の型 編集の意味

Iw 整数 幅 wで整数を出力する

Iw.m 整数 幅 wで整数を出力する
出力整数の桁がmより小さい時には，先頭に 0を補う (w = m)

Fw.d 実数 幅 wで実数を固定小数点形式で出力する
dは小数点以下の桁数 (w = d + 3)

Ew.d 実数
幅 wで実数を浮動小数点形式で出力する
dは小数点以下の桁数 (w = d + 8)
仮数部の 1桁目は 0になる

Gw.d 実数
幅 wで実数を固定小数点形式または浮動小数点形式で出力する
どちらになるかは，実数の指数部の大きさで決まる
dは小数点以下の桁数

ESw.d 実数 幅 wで実数を浮動小数点形式で出力する (dは E編集と同じ)
0以外の数値を出力すると，仮数部の 1桁目は 1から 9になる

ENw.d 実数
幅 wで実数を浮動小数点形式で出力する (dは E編集と同じ)
0以外の数値を出力すると，仮数部の整数は 1以上 1000未満と
なり，指数部は 3で割り切れる数になる

A 文字列 文字列をそれ自身の長さの幅で出力する
Aw 文字列 幅 wで文字列を出力する

E編集を使って実数を浮動小数点形式で出力すると，小数点の前が 0になります．例えば，

real x,y
x = 1.5
y = 3.14e10
print "(e15.5,e15.5)",x,y

の出力結果は，

0.15000e+01 0.31400e+11

となります．これでは感覚的にわかりにくいし，表示字数が 1個無駄になります．そこで，ES編集や
EN編集を使う方が良いでしょう．例えば，

real x
x = 3.14e10
print "(es15.5,en15.5)",x,x

の出力結果は，

3.14000e+10 31.40000e+09

となります．
各編集記述子と出力の数値は 1対 1対応にしなければならないので，配列を出力するときには出力
要素数と同数の編集記述子を書かなければなりません．このとき，同じ編集記述子を繰り返すならば，
編集記述子の前に整数 rを付加して，“ r回反復する ”という指定ができます．例えば，“ 3f10.5 ”は
“ f10.5,f10.5,f10.5 ”と書くことと同等です．
さらに，実数，整数，実数，整数のような繰り返しのときには，かっこで囲んで反復指定をすること

ができます．例えば，次のように書くことができます．

real x,y
integer m,n
x = 1.5
y = 0.03
m = 5
n = 100
print "(2(f10.5,’ ’,i7))",x,m,y,n

この print文の formatは，“ f10.5,’ ’,i7,f10.5,’ ’,i7 ”と書くことと同等です．
複素数は，“実部，虚部 ”という実数のペアであり，計算機内部的には要素数 2の 1次元配列と同型で
す．このため，複素数を出力するときは，複素数 1個あたり，実数の編集記述子を 2個並べる必要があ
ります．例えば，以下のように書きます．

complex c
c = (1.0,-2.0)
print "(4f8.3)",c,c**2

この出力結果は，

3.000 -2.000 5.000 -12.000

となります．しかし，これでは複素数という感じが出ないので，少し工夫して，

complex c
c = (1.0,-2.0)
print "(2(f8.3,’ + (’,f8.3,’)i ’))",c,c**2

などとしてみれば良いでしょう．この出力結果は，

3.000 + ( -2.000)i 5.000 + ( -12.000)i

となります．
なお，format中の編集記述子の数よりも出力文のデータ値の方が多い場合には，編集記述子の数だけ
出力した後で改行し，同じ formatを再度使って残りのデータ値を出力します．文字列が入っていれば，
文字列も再度出力されます．
逆に，format中の編集記述子の数よりも出力文のデータ値の方が少ない場合には，指定したデータ値
を出力した段階で終了し，残りの編集記述子は無視されます．無視された記述子以降は文字列等が入っ
ていても全て無視されます．そこで，配列を出力するときなどは，反復指定に大きめの数値を与えてお
くことができます．例えば，

real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10(f10.2,’cm ’))",(a(m),m=1,4) ! 反復指定は 10

のように，反復指定を 10回にしておいても，出力結果は，

2.25cm 30.20cm 400.70cm 5000.60cm

となります．
文字列のように出力文中のデータ値との対応がない編集記述子もあります．代表的なものを表 4.2に
示します．ここで，rは整定数で指定します．例えば，2次元配列 a(3,3)の配列要素を 3行 3列の行列

のように 1行あたり 3個ずつ出力するときは，スラッシュ (/)編集を使って，

real a(3,3)
.......

print "(3(3f12.5/))",((a(i,j),j=1,3),i=1,3)

と書くことができます．

表 4.2 出力文中のデータ値との対応がない編集記述子
編集指定 編集の意味

/ 改行する
r/ r回改行する
rX r個スペースを挿入する
$ print文や write文終了時の改行を抑制する

:
出力文中の数値の出力が終わった時点で format中
の以後の出力を打ち切る

X編集は出力中に適当な数のスペースを入れるときに使います．これは「’ ’」のように，スペー
スの文字列を与えるのと同等です．
ドル ($)編集は，改行コードを出力しない指定です．通常，print文や write文で出力すると，出力
後に改行をするため，複数の print文や write文を使って 1行に出力することはできません．これは，
指定した出力文字に加えて改行コードを出力しているためです．ドル ($)編集を使えば改行コードを出
力しないので，複数の print文や write文を使って 1行に続けて文字を出力することができます．例え
ば，doループ中に入れた print文で，計算結果を横に並べて出力することもできます．
最後のコロン (:)編集はわかりにくいので，例を使って説明します．先ほど反復指定に大きめの数値
を与えておくことができる例を示しましたが，このとき数字の前に文字列を付けようとすると問題が起
こります．例えば数値の前に＄記号を付けたくて，

real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10(’ $’,f10.2))",(a(m),m=1,4) ! 反復指定は 10

と書くと，出力結果は，

$ 2.25 $ 30.20 $ 400.70 $ 5000.60 $

となります．すなわち，最後に余分な 5個目の “ $ ”が出力されるのです．これは，数値と編集記述子が
対応しなくなった時点で出力が終了するのですが，5個目の “ $ ”の時点ではまだ終了するかどうか未定
だからです．これを防ぐために使うのがコロン (:)編集です．上記の formatを修正して，

print "(10(’ $’,f10.2:))",(a(m),m=1,4) ! f10.2の後にコロンを挿入

のように f10.2の後に “: ”を入れておけば，対応しなくなった時点からさかのぼって，“: ”の場所で出
力が打ち切られます．これならば，

$ 2.25 $ 30.20 $ 400.70 $ 5000.60

のように，“ $ ”は 4個しか出力されません．

4.4 データの入力方法
プログラムが実行しているとき，そのプログラム中の指定した変数に外部からデータを代入すること
を “入力する”といいます．計算条件を設定するための変数にデータを入力できるようにしておけば，プ
ログラムの実行を開始してから条件を設定して，それに応じた計算をさせることができます．これによ
り，プログラムはその動作だけを利用する “ブラックボックス ”になり，コンパイルしたプログラムを
“アプリケーション ”として他の人に提供することも可能です．

4.4.1 入力文の一般型
データ入力には read文を用います．

read(nd,*) 変数 1，変数 2,　...

ndは装置番号で，ndに適当な整数を与えると “ fort.nd ”という名のファイルから入力します26．装置
番号に関する条件や注意事項は出力の場合と同じで，原則として nd = 10にして下さい．出力ファイル
と違うのは fort.ndという名のファイルが存在していなければエラーになるので，あらかじめ用意してお
かなければならないことです．なお，“*”の位置には，write文のように formatによる書式指定ができ
ますが，あまり使うことがないので説明は省略します．
例えば，fort.30という名前のファイルに，

5.2 1.5 3

と書いて保存しておき，プログラム中に，

read(30,*) x,y,z

と書けば，この read文実行後，x= 5.2，y= 1.5，z= 3.0となって実行が継続します．入力ファイルに
改行が入っていても，read文の変数入力が完了するまで読み込みを続けるので，fort.30の入力数値は次
のように 3行に分けて書くこともできます．

5.2
1.5
3

read文実行時に，ファイルが存在しなかったり，データが足らない場合には，エラーになってプログ
ラムが強制終了します．これに対し，read文が要求する数値よりもファイルに書かれている数値の方が
多い場合は，read文に記述されている全ての変数に数値が代入された時点で入力が終了します．この
後，別の read文で再び入力を実行すると，最後に読み込んだ行の次の行から入力を再開します．例え
ば，fort.20という名前のファイルに，

5.2 1.5
10 20

と書いて保存しておき，プログラム中に，

read(20,*) x,y
read(20,*) m,n

と書けば，この 2回の read文実行後，x= 5.2，y= 1.5，m= 10，n= 20となります．read文の処理は
行単位で行われるので，最後に読み込んだ行に余分な数値が書かれている場合は無視されます．例えば，
上記の fort.20を書き換えて，

26出力ファイルと同様，“ fort ”の部分はコンパイラに依存します．ファイル名は 4.6節で説明する open文を使えば変更す
ることができます．また，コンパイラによっては，open文を使わなくても装置番号に対応する環境変数の指定で，任意の名前
を持つファイルに変更することが可能です．

5.2 1.5 30 40
10 20

のように 1行目に数字を余分に書いても，2回目の read文の結果は変わりません．
read文の変数の位置には，配列名や，do型出力と同型の do型入力を書くこともできます．これらは，
出力と入力という方向が異なりますが，入力要素数や繰り返しの意味は同じです．
ファイルではなく，キーボード (正確には標準入力)を使って入力したいときには，以下のように書き
ます27．

read *，変数 1，変数 2,　...

この文を実行すると，プログラムの実行が一時停止し，キーボードからの数値入力を待つ状態になり
ます．そこで，適切な数値をキーボードから入力すると，その数値を所定の変数に代入した後，実行が
再開します．このため，read文のタイミングを考慮して数値を入力しなければ，いつまでたっても停止
したままです．そこで，read文の前に入力を促すような文字を出力する print文を入れることをお勧
めします．例えば，

print *,’Input X and Y :’
read *,x,y

と書いておけば，入力のタイミングもわかるし，どの変数へ入力するための数値を要求されているかも
わかります．

4.4.2 入力時のエラー処理
ファイルからデータを入力するとき，要求したファイルが存在しなかったり，書き込まれたデータ数
より多くのデータを入力しようとすれば，実行時エラーになってプログラムは強制終了します．これを
防ぐため，read文中にエラー処理指定を入れることができます．

read(nd,*,err=num) 変数 1，変数 2,　...

ここで，numには文番号を与えます．この read文を実行したとき，入力エラーが起こると numで指定
した文番号の行へジャンプします．例えば，

do k = 1, 100
read(10,*,err=999) x,y,z
.......

enddo
999 x = 100

と書けば，エラーが起こると文番号 999の行にジャンプして，その行から実行を継続します．
もし “ファイルの終了 ”，すなわち，データを入力するときに，それ以上入っていなかった，という
場合を検知するだけなら，err=numの代わりに end=numと書くこともできます．両方入れてもかまい
ません．入力データ数が不明のときには，errか end指定を入れておき，データ終了時点で次の処理に
進むようなプログラムにしておくと良いでしょう．

4.4.3 ネームリストを用いた入力
便利なデータ入力手段として，“ネームリスト ”を用いる方法があります．例えば，

read(10,*) x,y,n

という入力文では，入力ファイル fort.10を，
27標準入力の装置番号は 5です．よって，“ read *, ”と書く代わりに，“ read(5,*) ”と書くこともできます．

10.0 1.e10 100

のように作成しますが，作成するためには入力変数の対応を常に覚えておかなければなりません．必要
なデータを全部書き込まなければならないし，順番を間違えることもできません．
これに対し，ネームリスト入力では，入力データを “変数＝データ値 ”という代入形で記述するので，
どの変数に代入するかを入力ファイルの中で明示することができます．
ネームリスト入力を使うときは，まず入力する可能性のある変数や配列名を namelist文で登録しま
す．namelist文は次のような形式です．

namelist /ネームリスト名/ 変数 1，変数 2,　...

namelist文は非実行文なので，全ての実行文より前に書かなければなりません．また，変数や配列名
の登録だけなので，型宣言は別途必要です．例えば，

real x,y,a(10)
integer n
namelist /option/ x,y,n,a

と書きます．ローカル変数だけではなく，use文で指定されたモジュール中で宣言されているグローバ
ル変数も登録可能です．
namelist文を使ってネームリストに登録された変数に対し，入力文は，

read(nd,ネームリスト名)

だけです．これをネームリスト入力文といいます．ネームリスト入力文は変数を指定しません．必要な
らば，

read(nd,ネームリスト名,err=num)

のように，文番号 numを使った err=numを追加してエラー発生時の処理をすることも可能です．例え
ば，ネームリスト名が optionならば，

read(15,option,err=999)

などのように書きます．
ネームリスト入力文に対する入力ファイルは次の形式で用意します．変数の順番は namelist文の登
録順とは無関係なので，自由に並べることができます．

&ネームリスト名
変数 1 = データ 1, 変数 2 = データ 2, ...

/

“&ネームリスト名”から “/”までがネームリスト入力文 1回で入力されるデータです．例えば上例のよ
うにネームリスト名が optionのときには，

&option
x=10.0, y=1.e10, n=100

/

のようにファイルに書いておきます．入力を開始すると，ネームリスト入力終了の記号 “/ ”を読み込む
まで入力を続けるので，次のように 1行ずつ書くこともできます．

&option
x=10.0
y=1.e10
n=100

/

ネームリスト入力にはもう一つ利点があります．それは，必ずしも登録された変数全部を入力ファイ
ルに記述する必要がないことです28．記述しなかった変数には，ネームリスト入力文の実行前までに代
入されていた値がそのまま残ります．このため，あらかじめ全ての登録変数にデフォルト値を代入して
おけば，変更したい変数だけ入力ファイルに記述することができます．例えば，

real x,y,a(10)
integer n,i
namelist /option/ x,y,n,a
x = 100.0
y = 100.e10
n = 0
do i = 1, 10

a(i) = i
enddo
read(10,option)

のようにプログラムを書いたとします．入力ファイルとして fort.10という名のファイルに，

&option x=10.0, a(3)=5.0 /

と書き込んでおけば，read文実行後，xは変更されますが，yや nはそのままです．配列 aの場合には，
代入された要素 a(3)のみが変更されます．
なお，ネームリストに登録された変数の内容は，次のネームリスト出力文で出力することもできます．

write(nd,ネームリスト名)

この場合，全登録変数が “変数＝データ値 ”という形で出力されます．もっとも，ネームリスト出力文
を実行すると，登録された全変数のデータが標準形式で出力されるので，変数が多いと煩雑です．取り
あえず値を確認したいとき以外は，あまり使わない方が良いと思います．

4.5 書式なし入出力文によるバイナリ形式の利用
これまで，print文・write文による出力や read文による入力には，文字を使って表現した数値を用
いていました．これは我々人間の都合によるものです．計算機内部の数値は 2進数で表現されていて，
“ 10 ”とか，“ 1.000e20 ”とかの文字ではありません．しかし，2進数の並びでは我々が理解できないた
め，出力するときは “ 2進数→ 10進数文字表現 ”というデータ変換を行って画面に表示したりファイル
に保存し，入力するときは，“ 10進数文字表現→ 2進数 ”というデータ変換を行って所定の変数に数値
を代入するのです．この文字で表現した形式を “テキスト形式 ”といいます．
しかし，2進→ 10進変換には時間がかかる上に，文字に変換することでデータ量が増えます．これは，
文字データが 1文字あたり 1byte必要だからです．例えば，倍精度実数の 2進数表現は 8byteですが，実
数を有効数字 15桁で出力するには 15文字 (15byte)以上必要です．そこで，大量のデータを精度を落と
さずに保存するときは，内部表現のままで保存するのが有効です．この内部表現を “バイナリ形式 ”と
いいます．Fortranでバイナリ形式の入出力を行うときは，write文や read文において formを省略し
た書式を用います．これを “書式なし入出力文 ”といいます．これに対し，これまで説明してきた form
を指定する write文や read文は “書式付き入出力文 ”です．

28逆に，入力ファイルに同じ変数を複数回記述することもできます．この場合は最後に代入した値が有効になります．

書式なし write文は，次のように装置番号だけをかっこ内に指定した write文です．

write(nd) データ 1，データ 2,　...

また，書式なし read文は，装置番号だけをかっこ内に指定した read文です．

read(nd) 変数 1，変数 2,　...

書式なし read文は，書式付き read文と同様に，文番号 numを使って，

read(nd,err=num) 変数 1，変数 2,　...

のように，err=numや end=numを追加したエラー発生や終了時の処理をすることもできます．
書式なし write文で作成したバイナリ形式ファイルから，書式なし read文を使ってデータを読み込
むときにはいくつか注意が必要です．まず，write文で出力したときのデータと同じ数値型の変数を同
じ順番で read文に並べる必要があります．例えば，

real x,y
integer n
x = 10.0
y = 100.0
n = 10
write(20) x,n,y

というプログラムで作成された fort.20というファイルから数値を入力するには，

real x,y
integer n
read(20) x,n,y

のように書かなければなりません．この場合，xも nも同じ 10だからと思って，

read(20) n,x,y

と入力すると，出力と数値型の異なる nと xには，正しい値が代入されません．
また，テキスト出力のような “改行 ”はありませんが，write文 1回ごとに印 (ヘッダ)が付くので，

write文 1回の出力に対し，read文 1回で入力しなければなりません29．ただし，write文 1回で書き
込まれたデータ数よりも read文 1回で入力するデータが少ないのは問題ありません．このとき入力し
なかったデータは読み飛ばしたことに相当します．例えば，上記の fort.20のデータを，

read(20) x,n
read(20) y

のように 2行の read文に分けて入力しようとすると，1行目の read文実行時における xと nには正常
な値が代入されますが，2行目の read文実行時に「入力データがない」というエラーで強制終了します．
逆に，上記の write文を修正して，

write(20) x,n
write(20) y

という 2行で出力したファイルから，

29このように，書式なし write文を使って保存したバイナリ形式ファイルには，データだけでなく Fortran独自のヘッダが
付加されています．このため，C言語など，他のプログラミング言語で作成したプログラムから読み込む場合や，データ解析
ソフトを使って解析するときにはうまく入力できない場合があります．そのような利用が目的でデータを保存するときは，テ
キスト形式で保存した方が良いでしょう．

read(20) x,n,y

のように，read文 1回で読み込むこともできません．
書式なし write文は，1回で出力できるデータ数に特に制限はありません．このため，大きな配列を

1回で出力することも可能です．例えば，

real p(10000),q(10000),r(10000)
write(30) p,q,r

のように，1回の write文で複数の配列の全要素を出力することができます．
もっとも，出力用のプログラムと入力用のプログラムのメンテナンスを考えれば，次のように出力す
る配列要素数をあらかじめ出力しておいた方が良いと思います．

real p(10000),q(10000),r(10000)
write(30) 10000
write(30) p,q,r

これを入力するときは，出力数を考慮して

real p(10000),q(10000),r(10000)
integer i,imax
read(30) imax
read(30) (p(i),i=1,imax),(q(i),i=1,imax),(r(i),i=1,imax)

のように書きます．こうしておけば，出力用のプログラムを修正して要素数を減らしても，入力用のプ
ログラムの変更は不要です．

4.6 ファイルのオープンとクローズ
write文や read文を使ってファイルから入出力をする場合，何も指定がなければ，装置番号 ndを付
加した “ fort.nd ”という名のファイルを使用します．これに対し，任意の名前を持つファイルを使いた
いときには，open文を使って入出力文の実行前にファイル名を指定しておきます．これを “ファイルを
オープンする ”といいます．open文は以下の形式です．

open(nd,file=name［, form=format］［, status=stat］［, err=num］)

［ ］は，その内容が省略可能という意味です．それぞれの記述 (制御指定子)の意味を表 4.3に示します．
例えば，装置番号 10のテキスト形式ファイルを “ text.out ”という名にするときは，

open(10,file=’text.out’)

と書きます．また，装置番号 30のバイナリ形式ファイルを “ binary.dat ”という名にするときは，

open(30,file=’binary.dat’,form=’unformatted’)

と書きます．ただし，既存のファイルを指定して write文で書き込むと，それまで書き込まれていたデー
タが上書きされて消えるので注意が必要です．上書きを防ぎたいときには，

open(30,file=’binary.dat’,form=’unformatted’,status=’new’,err=999)

のように，status=’new’と errを指定します．statusに’new’を指定すると，ファイルが存在しなけ
れば新しく作成し，存在すればエラーになります．そこで，エラーの処理を errで指定した文番号 999

の行に用意しておけば上書きを防ぐことができます．

表 4.3 open文の制御指定子の意味

指定子 指定情報 指定子の意味と注意

nd 装置番号 整数を与える (整数型変数や整数式を与えることも可能)
オープンした後，read文や write文の装置番号として使う

name ファイル名 文字列で指定する (文字変数も可能)
ファイル名は大文字・小文字を正しく指定する必要がある

format ファイル形式
文字列で指定する
省略するとテキスト形式の入出力が仮定される
バイナリ形式の入出力を使うときは「’unformatted’」を指定する

stat ファイル情報

文字列で指定する
既存のファイルを使うときは「’old’」を，存在しないファイルを使
うときは「’new’」を指定する
条件に合わないとエラーが発生する

num 文番号 エラーのときにジャンプする行の文番号を指定する
省略すると，エラーが起きたときにはプログラムが強制終了する

statusの指定を省略して open文を実行したとき，指定したファイルが存在しなければ，その名前の
ファイルが新たに作成されます．このとき，そのファイルが read文の入力用であれば，データが入って
いないので read文実行時にエラーになりますが，プログラム終了後も作成された空のファイルが残っ
てしまいます．そこで，ファイルが入力用のときは，status=’old’を指定して，その存在をチェック
した方が良いでしょう．例えば，バイナリ形式ファイル “binary.inp”を入力ファイルとして装置番号 20
に指定するには，

open(20,file=’binary.inp’,form=’unformatted’,status=’old’,err=999)

のように書きます．この open文では，オープンした時点でファイルが存在しなければ，err=999で指
定した文番号 999の行へジャンプします．
ファイルへ出力する場合，write文実行時の出力命令のタイミングと実際にディスクに書き込まれる
タイミングは必ずしも一致していません．これは入出力ハードウェアを効率よく運用するために，一時
記憶領域への読み書きが介在するためです．このため，確実に書き込みを完了させたいときにはファイ
ルをクローズします．クローズは，次の close文で行います．

close(nd)

ndはクローズするファイルの装置番号です．例えば，装置番号 30のファイルをクローズするときは，

close(30)

のように書きます．close文を実行すると，その時点までに装置番号 ndに出力した全てのデータがディ
スクに書き込まれます．もっとも，プログラムが正常に終了すれば，全てのファイルが自動的にクロー
ズされるので，通常は close文を書かなくても問題ありません．
ファイルをクローズすると，装置番号 ndと open文で指定したファイル名の関係は途切れます．この
ため，同じ装置番号に新たに別のファイルを指定してオープンすることも可能です．また，同じ名前の
ファイルを再度オープンすることもできます．ただし，再オープンしてもそれ以前の読み書きの継続に
はならず，ファイルの先頭に戻って読み書きをします．すなわち，“巻き戻し ”をすることになります．
ファイルを巻き戻すと，read文による入力は一からやり直すことになり，write文による出力はファイ
ルの先頭から書き直します．このため，巻き戻してから write文で出力すると，それ以前に書き込んだ

データは消去されます．
なお，巻き戻すのが目的であれば，rewind文を使うことで，クローズせずともファイルを巻き戻すこ
とができます．rewind文とは，次のような文です．

rewind(nd)

ndは再度最初から読み書きするファイルの装置番号です．巻き戻しを利用すれば，まず write文を
使ってファイルにデータを出力しておき，次にそれを巻き戻して，read文を使ってそのデータを入力し
て使うことも可能です．

演 習 問 題 4

(4–1) 連立常微分方程式：2次のRunge-Kutta法
N + 1個の質量mのおもりが一直線に並んでいて，隣り合ったおもりはバネ定数 kのバネでつながれ
ているとする．このおもりの位置を左側から，z0，z1，· · ·，zN とすると，i番目のおもりの運動方程式
はその速度を viとして，

dzi

dt
= vi

m
dvi

dt
= −k(zi − zi−1) − k(zi − zi+1)

である．このおもりの運動を 2次のRunge-Kutta法で解析せよ．

ここで，2次のRunge-Kutta法とは変数ベクトル xに対する連立常微分方程式 dx

dt
= f(t, x)に対し，

時刻 tnの値 xnから出発して時刻 tn+1 = tn + ∆tの値 xn+1を

k1 = f(tn,xn)

k2 = f(tn + ∆t, xn + ∆t k1)

xn+1 = xn +
∆t

2
(k1 + k2)

で計算する方法である．∆tは適当に決める．プログラムは

(1) 時間を進めるメインプログラム
(2) Runge-Kutta法の 1ステップを計算するサブルーチン
(3) 運動方程式の右辺を計算するサブルーチン

の 3個から構成するものとする．さらに，ziや viは引数で与え，mや kはグローバル変数としてモジュー
ルと use文を使って共有するようにせよ．なお，初期条件は以下の通りとする．

zi = hi + g sin(kpi) (i = 0 · · ·N)

vi = 0

ここで，kpは適当な整数 pを使って，kp = 2πp
N
で与える．また，端にあるおもり（z0と zN）は動か

ないとする．（つまり，計算しないで初期条件のままとする）

(4–2) 連立常微分方程式：4次のRunge-Kutta法
問題 (4–1)を修正して 4次のRunge-Kutta法で解析せよ．ここで，4次のRunge-Kutta法とは微分方
程式 dx

dt
= f(t, x)に対し，時刻 tnの値 xnから出発して時刻 tn+1 = tn + ∆tの値 xn+1を

k1 = f(tn, xn)

k2 = f(tn +
∆t

2
, xn +

∆t

2
k1)

k3 = f(tn +
∆t

2
, xn +

∆t

2
k2)

k4 = f(tn + ∆t, xn + ∆tk3)

xn+1 = xn +
∆t

6
(k1 + 2(k2 + k3) + k4)

で計算する方法である．初期条件や，境界条件は問題 (4–1)と同じにし，問題 (4–1)の結果と比較せよ．

(4–3) 楕円型偏微分方程式
電荷密度 ρ(x, y)が与えられているときに電位の満足する 2次元ポワソン方程式

∂2V (x, y)
∂x2

+
∂2V (x, y)

∂y2
= −ρ(x, y)

を解くには次のようにする．
等間隔 hごとに並んだ x座標点 xi = hi, (i = 0～M)と y 座標点 yj = hj, (j = 0～N)に対して，

Vi,j = V (xi, yj)，ρi,j = ρ(xi, yj)として，上の偏微分方程式を差分化すれば次式になる．

Vi−1,j − 2Vi,j + Vi+1,j

h2
+

Vi,j−1 − 2Vi,j + Vi,j+1

h2
= −ρi,j

これは Vi,j に関する連立 1次方程式である．これを解くには，以下のような漸化式に変形して反復法
を用いる．

V
(k+1)′

i,j =
1
4

(
V

(k+1)
i−1,j + V

(k+1)
i,j−1 + V

(k)
i+1,j + V

(k)
i,j+1 + ρi,jh

2
)

V
(k+1)
i,j = V

(k)
i,j + ω(V (k+1)′

i,j − V
(k)
i,j )

これを適当な初期条件 V
(0)
i,j から出発して，k回目の反復値 V

(k)
i,j と k + 1回目の反復値 V

(k+1)
i,j の差が十

分小さくなれば，元の連立方程式の近似解になると考えられる．
なお，第 1式の V

(k+1)′

i,j をそのまま次の反復値 V
(k+1)
i,j として用いる手法を，ガウス・ザイデル法という．

しかし，ガウス・ザイデル法は収束が遅いので，第 2式を使って解を改良する．これを，SOR(Successive
Over Relaxation，逐次過緩和)という．ωは加速係数であり，ω = 1のときがガウス・ザイデル法，ω > 1
が SORである．ただし，ω = 2では発散するので，1 < ω < 2の範囲で最も収束が速くなる数値を選ぶ．
この反復法を使って，次の電荷密度における電位を計算するプログラムを作成せよ．

ρ(x, y) = sin
(

2πx

Mh

)
sin
(

4πy

Nh

)
ここで，初期条件は V

(0)
i,j = 0，境界条件は V0,j = Vi,0 = VM,j = Vi,N = 0 (i = 0～M, j = 0～N)で

よい．また，収束の判定は
max

i,j
|V (k+1)

i,j − V
(k)
i,j | < ε

で行うこと．ただし，εは適当な小さい値とする (例えば，10−12)．
プログラムは，

(1) 反復を行い，収束条件を確認するメインプログラム
(2) 初期値を代入するサブルーチン
(3) ガウス・ザイデル法+SORを 1ステップ行うサブルーチン

の 3個から構成するものとする．

(4–4) 放物型偏微分方程式：陽解差分法
次の温度 T (x, t)に関する熱伝導方程式を，陽解差分法 (Explicit法)を用いて解析せよ．

∂T

∂t
= κ

∂2T

∂x2

このような時間発展の偏微分方程式を解くには，まず等間隔 hごとに並んだ x座標点 xi = hi, (i = 0
～L)に対して初期の温度分布を与え，時間間隔∆tごとに時間を進めた温度分布を計算する．陽解差分
法とは，nステップ目の温度分布を Tn

i = T (xi, ∆t n)と表したときに，熱伝導方程式を，

Tn+1
i − Tn

i

∆t
= κ

Tn
i+1 − 2Tn

i + Tn
i−1

h2

と近似する方法である．この方程式を Tn+1
i に関して解けば，

Tn+1
i = Tn

i +
κ∆t

h2

(
Tn

i+1 − 2Tn
i + Tn

i−1

)
（i = 1, 2, . . . , L − 1）

となるので，これを使って Tn
i から Tn+1

i を繰り返し計算していけばよい．
ただし，境界条件として，Tn

0 = 0，Tn
L = 0とする．また，初期条件は， T 0

i = i (i < L/2)

T 0
i = L − i (i > L/2)

とする．プログラムは

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 陽解法で 1ステップ進めるサブルーチン

の 3個から構成するものとする．温度 Tn
i と，h，∆t，κなどのパラメータはグローバル変数にしてモ

ジュールと use文で共有する．また，メモリの節約のため，Tn
i を配列 T1(i)で表し，Tn+1

i を配列 T2(i)

で表して，T1と T2のデータをうまく交換するように考える．
完成したプログラムを使って，κ∆t

h2 = 0.2にすると安定に計算ができるのに，κ∆t

h2 > 0.5にすると，

不安定になることなども確かめよ．

(4–5) 放物型偏微分方程式：陰解差分法
温度 T (x, t)に関する熱伝導方程式を，陰解差分法 (Implicit法)を用いて解析せよ．
ただし，陰解差分法とは，問題 (4–4)の偏微分方程式を

Tn+1
i − Tn

i

∆t
=

1
2

[
κ

Tn+1
i+1 − 2Tn+1

i + Tn+1
i−1

h2
+ κ

Tn
i+1 − 2Tn

i + Tn
i−1

h2

]

と近似する方法である．この方程式を変形すれば，

aiT
n+1
i−1 + biT

n+1
i + ciT

n+1
i+1 = di （i = 1, 2, . . . , L − 1）

の形になるので，問題 (2–10)で示した連立 1次方程式の解法を使って Tn+1
i が計算できる．

ここで，境界条件や初期条件は陽解法と同じでよい．プログラムは

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 陰解法で 1ステップ進めるサブルーチン

の 3個から構成するものとする．温度 Tn
i と，h，∆t，κなどのパラメータはグローバル変数にしてモ

ジュールと use文で共有する．また，メモリの節約のため，Tn
i を配列 T1(i)で表し，Tn+1

i を配列 T2(i)

で表して，T1と T2のデータをうまく交換するように考える．
完成したプログラムでは，κ∆t

h2 > 0.5にしても不安定にならないことなどを確かめよ．

(4–6) 分子動力学
2個の距離 r離れた分子間の位置エネルギーを表す近似式の一つに次式のようなレナード・ジョーン
ズポテンシャルがある [3]．

U(r) = 4ε

{(σ

r

)12
−
(σ

r

)6
}

この式は，分子が近づくと斥力（反発力）が働き，遠ざかると引力が働くことを示している．レナー
ド・ジョーンズポテンシャルの勾配を計算することで分子間に働く力が計算できる．すなわち，2個の
分子 iと jの位置ベクトルを ri，rj とすると，分子 jが分子 iに及ぼす力は，

F ij = −∇iU(|ri − rj |) =
4ε

σ

{
12
(

σ

rij

)13

− 6
(

σ

rij

)7
}

ri − rj

rij

となる．ここで,rij = |ri − rj |である．
今，分子がN 個あるとすると，分子 iに加わる力は i以外の分子から受ける力の合力なので，

F i =
N∑

j=1

i6=j

F ij =
N∑

j=1

i 6=j

4ε

σ

{
12
(

σ

rij

)13

− 6
(

σ

rij

)7
}

ri − rj

rij

となる．これを用いてN 個の分子の運動方程式

mi
d2ri

dt2
= F i =

N∑
j=1

i6=j

4ε

σ

{
12
(

σ

rij

)13

− 6
(

σ

rij

)7
}

ri − rj

rij
(i = 1...N)

を解くプログラムを作成せよ．ここで，分子数が増えると時間がかかるので，効率よく時間積分をする
ためにベルレ法を用いる [3]．ベルレ法とは次のような 2段階の計算で位置ベクトルと速度ベクトルを更
新する方法である．
まず，現時刻 (nステップ後)の位置ベクトル rn

i を用いて計算した各分子に加わる力F n
i と現時刻の速

度ベクトル vn
i を使って次式のように位置ベクトルを更新する．

rn+1
i = rn

i + ∆tvn
i +

∆t2

2mi
F n

i

ここで，∆tは 1ステップの時間間隔である．
次に，この更新された位置ベクトル rn+1

i を用いて新しい力F n+1
i を計算し，次式のように速度ベクト

ルを更新する．
vn+1

i = vn
i +

∆t

mi

F n+1
i + F n

i

2

このとき，新しい力 F n+1
i は次ステップの位置ベクトルや速度ベクトルの更新に利用できることに注意

すること．
なお，力の計算に r−2k

ij が入っているため，2分子の距離がかなり接近すると，力が大きくなりすぎて，
安定に計算することができなくなる．そこで，r−2

ij を次式のように修正するとよい．

1
r2
ij

→ 1
(xi − xj)2 + (yi − yj)2 + (zi − zj)2 + ε2

ここで，εは最も接近する可能性のある距離で，σに比べて十分小さい数にする．
プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) ベルレ法第 1段階で位置ベクトルを更新するサブルーチン
(4) ベルレ法第 2段階で速度ベクトルを更新するサブルーチン

の 4個から構成するものとする．
初期条件としては，まず適当な間隔離れた 2個の分子でテストする．これにより正常に動作すること
が確認されたら，正三角形の頂点に配置した 3個の分子や正四面体の頂点に配置した 4個の分子などの
動きを計算せよ．

(4–7) 1次元移流方程式
次の関数 f(x, t)に関する 1次元移流方程式を，CIP法を用いて解析せよ [4]．

∂f

∂t
+ u

∂f

∂x
= 0

ここで，CIP法とは，間隔 hの x座標点 xi = hi, (i = 0～L)で定義された，nステップ目の関数値
fn

i = f(xi, n∆t)と，その x方向微分 g(x, t) =
∂f

∂x
の値 gn

i = g(xi, n∆t)が与えられているとき，時間間
隔∆t後の関数値と微分値を，

fn+1
i = aiX

3 + biX
2 + gn

i X + fn
i

gn+1
i = 3aiX

2 + 2biX + gn
i

で近似する方法である．ここで，X = −u∆tである．また，ai，biは次式で与えられる．

ai =
gn
i + gn

i+1

h2
+

2(fn
i − fn

i+1)
h3

bi =
3(fn

i+1 − fn
i )

h2
−

2gn
i + gn

i+1

h

初期条件は，グリッド数 Lと，その中間の値，i1，i2を適当に決めて， f0
i = 1 (i1 < i < i2)

f0
i = 0 (それ以外)

とせよ．なお，h = 1，u = −1とする．
まず，nステップ目の座標点データ fn

i と gn
i を 1次元配列 f1(i)と g1(i)で表し，これらと適当な値

の u，h，∆tを用いて，n + 1ステップ目の座標点データ fn+1
i と gn+1

i を表す 1次元配列 f2(i)と g2(i)

を計算するサブルーチンを作成する．そして，これらを適宜交換しながら繰り返すことで，fn
i や gn

i が

時間発展するプログラムを作成すればよい．なお，iの最大値 Lでは ai，biを計算することができない
ので，常に fn

L = 0，gn
L = 0で良い．

プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) CIP法で関数値と微分値を更新するサブルーチン

の 3個から構成するものとする．

(4–8) 粒子密度 – 分布関数
2次元領域，0 5 x 5 Xm，0 5 y 5 Ym，の内部にN 個の粒子があるとき，粒子の密度分布は以下の
ような手順で計算することができる．
まず，適当な整数 LとM を使って，2次元領域を hx = Xm/Lごとに並んだ x座標点 x0，x1，· · ·，

xL(= Xm)と，hy = Ym/M ごとに並んだ y座標点 y0，y1，· · ·，yM (= Ym)で指定したグリッドを導入す
る．次に密度配列 Rho(0:L,0:M)に初期値として 0を代入しておいて，k番目の粒子座標 (x(k),y(k))

から以下のようにして各グリッドに所属する粒子数をカウントする．

i = x(k)/hx ! hx = hx

j = y(k)/hy ! hy = hy

Rho(i,j) = Rho(i,j) + 1

ただし，iと jは整数型変数である．すなわち，x(k)/hxと y(k)/hyを切り捨てることで粒子が所属す
るグリッド番号を計算する．
問題 (3–10)で紹介した乱数発生用のサブルーチンを使って，一様乱数を 2個ずつ発生させ，これらに

Xmと Ymを掛けることで 2次元領域中にランダムで一様に分布した粒子の座標点をN 個作成し，その
密度の平均と標準偏差を計算せよ．そして，N が大きくなると，標準偏差が小さくなることを確かめよ．

(4–9) ブラウン運動 – 拡散
ある点から粒子を速度 vで出発させたところ，時刻∆tごとに周りの粒子に衝突して方向を変えると
する．この時の粒子の進む様子を計算してみよう．
問題 (4–8)と同じ 2次元領域で，粒子の初期位置は，全て (Xm

2 ,
Ym

2 )，すなわち，2次元領域の中央と
する．計算は

(1) 1ステップごとに乱数Rを発生させ，θ = 2πRで，方向角 θを計算する
(2) 1ステップ後の位置を (xs+1, ys+1) = (xs + v∆t cos θ, ys + v∆t sin θ)で計算する
(3) 粒子が 0 5 x 5 Xm，0 5 y 5 Ymの領域から外に出たら計算をストップする

のように進めればよい．
この過程によってN 個の粒子のNT ステップ後の位置を計算し，その密度を計算せよ．また，ステッ
プNT を増やすと，密度がどの様に変化するかを調べよ．
プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 粒子の座標を更新するサブルーチン
(4) 粒子の位置から密度を計算するサブルーチン

の 4個から構成するものとする．

第5章 文字列の活用
これまで print文や format文に記述して，数値の意味などを表示するのに使ってきた文字列ですが，
この他にも様々な用途があります．また，固定した文字の並びだけではなく，文字列変数を使って条件
に応じてその内容を変更したり，文字列と文字列を連結したり，文字列の一部を取り出したりするなど
の “文字列演算 ”をすることもできます．本章では文字列をより積極的に活用する手法を紹介します．

5.1 文字列定数と文字列変数
Fortranにおける文字列とは 2個の「’」または「"」で囲んだ文字の並びのことです．例えば，

’abc’　’TaguchitT.’ "(123.5678+X^2)" "漢字も書けます"

等が文字列です．文字列にはスペースや記号を入れることもできます30．「’」と「"」は同等なので，どち
らを使うかは自由です．例えば，通常は「’」で囲み，文字列の中に「’」を使いたいときは，「"Teacher’s"」
のように全体を「"」で囲む，というように決めておけば良いでしょう．
コンピュータの “文字 ”は文字コードという整数値と対応していて，文字列は文字コードを表す整数
の配列で実現されています．このため，整数定数の「1」と文字列の「’1’」は全く異なるものです．文
字コードは半角英数字なら 1文字あたり 1byteの整数，全角の漢字やひらがななら 1文字あたり 2byte
の整数です．半角英数字のコードは，現在ほとんど全てのコンピュータでASCIIコードが使われていま
す31．このため，半角英数字だけでプログラムを書けば移植性の問題はありません．しかし，全角文字
はOSや利用環境によって文字コードが違う可能性があるので，プログラム中に日本語を記述すると，別
のコンピュータに移したときに文字化けすることがあります32．Fortranにおいて日本語が使えるのは
コメント文とテキスト表示くらいなので，それほど重要ではありません．以下，文字列中の “文字 ”は
1byteのASCIIコードであると仮定して説明をします．
「’」または「"」で囲んだ文字列は，文字列 “定数 ”ですが，文字列を代入して保存する文字列 “変
数 ”を作ることもできます．文字列変数を作るには character宣言を使います．character宣言は，以
下のような書式です．

character 文字列変数 1*文字数 1，文字列変数 2*文字数 2，...

文字数とは，文字列変数に代入可能な最大の文字数です．指定できる最小の文字数は 1です．また，拡
張宣言文にして属性などを付加することも可能です．例えば，

character c1*10,c2*20,cc*1
character chr(20)*30,chs(100,200)*50

と宣言すると，文字列変数 c1には 10個まで，c2には 20個までの文字を代入することができます．こ
れに対し ccには 1文字しか入りません33．また chrや chsのように文字列の配列を宣言することも可
能です．chrは 30文字まで入る 1次元配列で，chsは 50文字まで入る 2次元配列です．
同じ文字数の文字列変数を複数個宣言するときは，以下の 2種類の書式で宣言することもできます．

character(文字数) 文字列変数 1，文字列変数 2，...
character(len=文字数) 文字列変数 1，文字列変数 2，...

30本章では半角スペース記号を明記するときに “ t”を使います．
31ASCIIコード表は付録 Fにあります．
32パソコンで使われているのは Shift JISコードが多く，Linuxなどの UNIX系 OSで使われているのは EUCコードが多

いようです．最近は，コード体系をより統一化した Unicodeを使う OSが増えてきました．
33文法的には “ *文字数 ”を省略して変数名だけで宣言すると，文字数が 1の文字列になります．しかし，プログラムがわか

りにくくなるので，常に “ *1 ”を付加して文字数 1を陽に指定した方が良いでしょう．

ここで，“文字列変数 ”には，変数名か，配列名とその要素数を記述します．例えば，文字数 10の文
字列変数や文字列配列を宣言するときは，

character(10) cs1,ds1,cas1(100)
character(len=10) cs2,ds2,cas2(100)

などと書きます．2行目の書式は，入力に手間がかかりますが「文字数」という意味を明示していると
ころが利点です．文字数は，次のように parameter変数 (3.8.2節)で指定することもできます．

integer, parameter :: kp = 10
character(kp) ch2
character(len=kp) ch3

文字列変数に文字列を代入するには，数値の代入と同じで，イコール「=」を使います．例えば，上記
の 10文字の文字列変数 c1に文字列定数を代入するには，

c1 = ’abc’

のように書きます．このとき，代入される文字 ’abc’ は 3文字なので，10文字の変数 c1の先頭から
順に 1文字ずつ代入され，残りの領域は半角スペースが代入されます．スペースも文字ですから，c1の
文字数は，代入した文字列の文字数にかかわらず，10のままです．図に描けば，以下のようなイメージ
です．

c1 a b c t t t t t t t

逆に，代入する文字列の文字数の方が多い場合には，その文字列の先頭から代入できる最大文字数ま
でが代入され，残りは切り捨てられます．拡張宣言文を使えば，次のようにあらかじめ文字列定数を代
入した文字列変数を用意することも可能です．

character :: chr*10=’abcde’

この場合，文字列変数の文字数が 10文字なのに代入しているのは 5文字ですから，後の 5文字はスペー
スが代入されています．

5.2 部分文字列と文字列演算
文字列変数に代入された文字列は部分的に取り出すことができます．これを “部分文字列 ”といいま
す．部分文字列は，文字列変数の先頭から数えて何番目から何番目という範囲を「:」を使って指定しま
す．例えば，c1という文字列変数の n1番目から n2番目の文字を取り出すには，

c1(n1:n2)

と指定します．この文字列は，n2-n1+1文字の文字列として扱われます．例えば，c1=’abcdefg’なら
ば，c1(3:5)は，’cde’です．n1と n2を等しくすることで 1文字を取り出すこともできます．例えば，

c1 = ’abcdefg’
do i = 1, 7

print *,c1(i:i)
enddo

とすれば，1文字ずつ縦に出力されます．
文字列配列の部分文字列を取り出す場合には，要素指定を先に，部分文字列指定を後に書きます．例
えば，1次元の文字列配列 chrに対し，k番目の要素の部分文字列は，

chr(k)(n1:n2)

のように指定します．かっこが連続するので，順番に気を付けて下さい．
文字列は連結することもできます．文字列の連結には演算子「//」を用います．例えば，

c1 = ’abc’//’xyz’

と書くと，c1には’abcxyz’という文字列が代入されます．文字列の連結は，文字列定数と文字列変数，
文字列変数と文字列変数という組み合わせでも可能です．ただし，文字列変数に代入された文字列を連
結するときには注意が必要です．例えば，

character c1*10,c2*20
c1 = ’abc’
c2 = c1//’xyz’

と書いた場合，c2に’abcxyz’という文字列が代入されると思ったら間違いです．正しくは

’abctttttttxyz’

が代入されます．これは，上記のように 10文字の文字変数 c1に 3文字の’abc’を代入しても，c1の文
字数は変わらないからです．末尾の不要なスペースを削除するには，部分文字列を使う必要があります．

character c1*10,c2*20
c1 = ’abc’
c2 = c1(1:3)//’xyz’

このプログラムならば，c2に’abcxyz’が代入されます．
しかし，この方法は変数に代入されている文字数が不明のときには使えません．そこで，関数 trimが
用意されています．trimは末尾のスペースを除去した文字列を返す組み込み関数です．例えば，上記の
例は，

character c1*10,c2*20
c1 = ’abc’
c2 = trim(c1)//’xyz’

と書くことができます．この結果も c2には’abcxyz’が代入されます．
文字列をサブルーチンの引数にするときは，「(*)」を指定して宣言します．例えば，

subroutine csubr1(chr)
implicit none
character(*) chr ! 文字数は不要
.......

の chrのように宣言します．Fortranの文字列には，文字コードの並びだけではなく，文字数の情報も
含まれています．このため，(*)指定のように文字数が明記されていなくても，コール側で指定した文
字数の文字列として使用することができます．例えば，

program ctest1
implicit none
call csubr1(’abcde’)

end program ctest1

subroutine csubr1(chr)
implicit none
character(*) chr
print *,chr,len(chr) ! 文字列 ’abced’と文字数 5が出力される

end subroutine csubr1

のようなプログラムにおいて，サブルーチン csubr1中の print文の出力は，文字列 ’abcde’と文字数
5になります．ここで，関数 lenは文字列の文字数を取得する関数です．文字列に関する組み込み関数

は，5.5節で説明します．
文字列には大小関係があり，これを利用して if文で条件分岐をすることもできます．2個の文字列を
比較するときは，以下の手順で行います．

(1) 文字数の長さが異なるときは，短い方の文字列の末尾にスペースを追加して文字数を等しくする
(2) 2個の文字列を先頭から 1文字ずつ比較していって，全て同じならば，“等しい (==) ”
(3) 異なる文字があれば，最初に異なる文字のASCIIコードを比較して，コード値の大小が文字列
の “大 (>) ”または “小 (<) ”

付録 Fの表からわかるように，ASCIIコードは，“スペース”＜ “数字”＜ “英大文字”＜ “英小文字”
の順で大きくなり，個々の文字は次のような順序になっています．

t < ’0’ < ’1’ <...< ’9’ < ’A’ < ’B’ <...< ’Z’ < ’a’ < ’b’ <...< ’z’

例えば，文字列の比較を使って次のようなプログラムを書くことができます．

character c1*10,c2*20
c1 = ’abcde’ ! ３番目が ’c’
c2 = ’abdce’ ! ３番目が ’d’
if (c1 < c2) print *,’c1 < c2’
if (c1 == c2) print *,’c1 == c2’
if (c1 > c2) print *,’c1 > c2’

この結果は，“ c1 < c2 ”です．なぜなら，3番目の文字が初めて異なり，c1は ’c’，c2は ’d’ です
が，’c’ < ’d’だからです．なお，c1は 10文字，c2は 20文字ですが，後ろにスペースを補うので，こ
の例の結果には無関係です．

5.3 出力における文字列の利用
最もよく文字列を使う場面は，print文や write文の中に入れて表示の補助に使うことでしょう．例
えば，

real spd,pres
spd = 10.0
pres = 960.0
print *,’ Wind Speed = ’,spd,’ Pressure = ’,pres

と書けば，出力が

Wind Speed = 10.0000000000000 Pressure = 960.000000000000

のようになって，どの数値がどの変数の値なのかが一目でわかります．
ここで，文字列はその長さに合わせて出力されています．これは，print文が文字列に入っている文
字数の情報に合わせて出力するからです．この機能は，formatで文字列の出力形式を指定するときにも
使うことができます．文字列の出力指定には，A編集 (表 4.1)を用いますが，A編集は，「a」だけ書く
と，文字列の文字数が出力幅になります．例えば上記の print文を，

print 600,’ Wind Speed = ’,spd,’ m/s’
print 600,’ Pressure = ’,pres,’ hPa’

600 format(a,f8.2,a)

と書けば，次のような文字幅に合わせた出力結果が得られます．

Wind Speed = 10.00 m/s
Pressure = 960.00 hPa

4.3節の formatの説明で紹介しましたが，出力形式指定を print文や write文の中に埋め込む書式
も，文字列の利用方法の一つです．例えば，

print 600,x
write(10,600) x

600 format(’ x = ’,es12.5)

というプログラムは次のように書き換えることができます．

print "(’ x = ’,es12.5)",x
write(10,"(’ x = ’,es12.5)") x

すなわち，format文のかっこ以下を，両端のかっこを含めて文字列にして print文や write文の書式
指定の位置 (form)に書き込みます．このとき，文字列変数を使えば，複数の場所で同じ formatを使う
ときに便利です．

character :: form*20="(’ x = ’,es12.5)"
print form,x
write(10,form) x

文字列変数を利用すれば，出力内容に応じて実行時に formatを変更することも可能です．例えば，

real x
character form*20
if (abs(x) >= 1.e5) then

form = "(’x = ’,es12.5)"
else
form = "(’x = ’,f10.5)"

endif
print form,x

とすれば，xの絶対値が 105以上のときには es12.5編集で，さもなくば f10.5編集で出力されます．
また，部分文字列を利用して変更することも考えられます．例えば，

real x
character form*20
form = "(’x = ’,f10.5)"
if (x >= 100.0) form(10:13)=’12.3’ ! 10.5から 12.3に変更
print form,x

のように書けば，xが 100.0以上のときは f12.3編集で，さもなくば f10.5編集で出力されます．

5.4 数値・文字列変換
ここまでは write文や read文の書式指定に文字列を使用していましたが，装置番号の位置に文字列
を記述することもできます．これは “文字列 ”から “数値 ”への変換，またはその逆変換を行うときに使
用します．5.1節で述べたように，「123」という数値と「’123’」という文字列は計算機内部の表現が異
なりますが，処理の過程で 123という数値からそれに相当する文字を作ったり，逆に’123’という文字
列を数値として計算に利用したい場合があります．そもそも，書式付き write文は “計算機内部の 2進
数”を “文字で表現された 10進数”に変換して出力する動作であり，書式付き read文はファイルなどか
ら “文字で表現された 10進数 ”を入力して “計算機内部の 2進数 ”に変換する動作です．装置番号の位
置に文字列を利用すると，その変換機能だけを使うことができるのです．

“数値 ”→ “文字列 ”変換をするには write文を用います．例えば，

real x
character ch*20
x = 123.5
write(ch,"(f10.5)") x

と書けば，文字列変数 chに’ 123.50000’という文字列が代入されます．ただし，文字列に変換すると
きは，出力文字数以上の長さを持つ文字列変数を用意しておく必要があります．
open文 (4.6節)を使ってファイルをオープンするとき，ファイル名は文字列で指定する必要がありま
す．そこで，file01.dat，file02.dat，...，file10.datという通し番号の付いた 10個のファイルが存在して，
それぞれを装置番号 11～20としてオープンするときは，次のように書きます．

character name*20
integer m
do m = 1, 10

write(name,"(’file’,i2.2,’.dat’)") m
open(10+i,file=name)

enddo

ここで，整数 mの出力編集を i2.2にしているのは，スペースを ’0’で埋めて，整数の出力を 01，02，...
のようにするためです．
逆に，“文字 ”→ “数値 ”変換をするには read文を用います．例えば，

real x
character ch*20
ch = ’12345.0’
read(ch,*) x

と書けば，実数型変数 xに 12345.0という “数値 ”が代入されます．

5.5 文字列に関する組み込み関数
文字列に関する組み込み関数は，trimや lenの他にも色々あります．代表的なものを表 5.1に示しま
す．本節ではこれらの関数の中で比較的良く使うものの使用方法について説明します．

表 5.1 文字列に関する組み込み関数
文字列関数 引数の型 関数値の型 関数の意味
trim(c) 文字列 文字列 末尾の空白を削除
len(c) 文字列 整数 文字列の文字数

len_trim(c) 文字列 整数 末尾の空白を削除した文字列の文字数
adjustl(c) 文字列 文字列 文字列を左にそろえた文字列
adjustr(c) 文字列 文字列 文字列を右にそろえた文字列

index(c,cp) 2個の文字列 整数 文字列 c中で文字列 cpが含まれていれば，
その開始位置を返す　無ければ 0を返す

repeat(c,n) 文字列と整数 文字列 文字列 cを n個連結した文字列

char(n) 整数 1文字の文字列 ASCIIコード値を与えると，
それに対応する半角英数文字を返す

ichar(c) 1文字の文字列 整数 半角英数 1文字を与えると，
それに対応するASCIIコード値を返す

まず，5.2節にも出てきた関数 lenは，文字列の文字数を取得するための関数ですが，使用の際は注

意が必要です．例えば，

character c1*10
integer m,n
m = len(’abc’) ! m = 3
c1 = ’abc’
n = len(c1) ! n = 10

とすると，mの値は 3ですが，nの値は 10になります．これは，文字列変数の文字数が宣言時の文字数
だからです．末尾のスペースを除去した文字列の長さを取得するには，関数 trimを使って除去してから
関数 lenで調べるか，関数 len_trimを使います．例えば，上のプログラムで，最後の nへの代入文を

n=len(trim(c1)) または n=len_trim(c1)

で置き換えれば，n= 3になります．
最後の charと icharの使用例を示します．この二つの関数を組み合わせれば，read文や write文を
使わなくても，文字と数値の変換をすることができます．例えば，2桁の整数を 3桁の 8進数で表示す
るプログラムは以下のようになります．

integer zero,num
character c1*10
num = 12
c1 = ’000’
zero = ichar(’0’) ! ’0’の文字コードを整数値に変換
c1(3:3) = char(zero+mod(num,8)) ! numの 1の位の文字
c1(2:2) = char(zero+mod(num/8,8)) ! numの 8の位の文字
c1(1:1) = char(zero+mod(num/64,8)) ! numの 64の位の文字
print *,’decimal(’,num,’) = octal( ’,trim(c1),’ )’

ASCIIコードでは，’0’,’1’,’2’,...,’8’,’9’の順で値が 1ずつ増加しています．そこで，先頭の’0’

のコード値に 1桁の数字を加えれば，その数字のコード値が得られます．同様に，大文字の英数字や，
小文字の英数字も 1ずつ増加して並んでいるので，先頭の文字である’A’や’a’のコード値に，その文
字の順番の差の数字を加えて逆変換すれば，対応する文字を得ることができます．整数の 16進数表示を
するプログラムを考えてみて下さい．

第6章 配列計算式
第 5章で説明した文字列の処理は，それ以前に説明した数値の計算と少し毛色が違います．数値計算
が 1個の数値と 1個の数値の間の演算を基本としているのに対し，文字列の処理では，“文字列 ”という
文字コードの集合をひとまとめにして取り扱い，文字列と文字列の連結や文字列変数への代入を一つの
式で実行することができます．文字列とは文字コードの配列なのですから，これを数値計算的に処理す
るならば，本来は 1文字ずつ処理しなければなりません．文字列の演算が可能なのは，Fortranコンパ
イラが，“文字列”という数値集合の処理を，内部で 1文字ずつ処理するプログラムに変換してくれるか
らです．

Fortranでは，この概念を一般の配列に拡張した “配列演算 ”が可能です．配列演算とは，数値の集合
を配列名で代表させ，配列要素と配列要素の演算を，配列名と配列名の演算の形で記述するものです．配
列演算を使えば，do文を使わずに配列を使った計算を記述することができます．これを “配列計算式 ”
と呼びます．配列計算式を使う利点は，単にプログラムが短くなるだけではありません．計算が最適に
なるように，繰り返しの順序やメモリ配置をコンパイラが決めてくれるので，場合によっては，do文を
使って書くより高速に計算させることができます．
本章では，配列計算式の使い方について説明します．また，プログラム実行中に任意の要素数を持つ
配列を用意する “動的割り付け ”についても説明します．

6.1 基本的な配列計算式
配列演算は，配列の全要素に対して，全て同じパターンの演算をするのが基本です．このため，1行
の配列計算式で用いる配列は，次元と各次元の要素数が全て等しい “同型の配列 ”でなければなりませ
ん．その点に注意すれば，配列の全ての要素に同じ定数を代入するときや，2個の配列の対応する要素
間で全て同じ四則演算をするときに，あたかも配列名が一つの変数であるかのような記述をすることが
できます．例えば，次のような記述が可能です．

real x(10,10),y(10,10),z(10,10)
x = 1.0
y = 2.0
z = x + 3.5*x*y**2

このプログラムを実行すると，配列 xの全ての要素は 1.0になり，配列 yの全ての要素は 2.0になり，配
列 zの全ての要素は 15.0 (=1.0 + 3.5×1.0×2.02)になります．このプログラムは do文を使った以下の
プログラムと同じ結果になります34．

real x(10,10),y(10,10),z(10,10)
integer i,j
do j = 1, 10

do i = 1, 10
x(i,j) = 1.0
y(i,j) = 2.0
z(i,j) = x(i,j) + 3.5*x(i,j)*y(i,j)**2

enddo
enddo

配列計算式中の定数は，全要素に対して共通の値として計算します．
次のような，組み込み関数を使った配列計算式も可能です．

34ただし，ループの処理はコンパイラが自動的に生成するので，実際にこの通りの順序で計算するかどうかはわかりません．
あくまでも，このプログラムと同じ結果になるという意味だと考えてください．

–90–



real a(10,10),b(10,10),c(10,10)
.......

a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))

この配列計算式を do文で表すと，次のようになります．

do j = 1, 10
do i = 1, 10
a(i,j) = sqrt(b(i,j)*sin(c(i,j))/(3.2*c(i,j) + 1.5e-3))

enddo
enddo

配列計算式においては，式中に含まれる全ての配列が同型でなければならないので，異なる次元や要
素数の配列が混じっているとエラーになります．ただし，下限は異なってもかまいません．例えば，

real a(3,3),r(-1:1,0:2)
.......

a = 3.14*r**2

のように計算することができます．ただし，この配列計算式を do文で表せば，

do j = 1, 3
do i = 1, 3
a(i,j) = 3.14*r(i-2,j-1)**2

enddo
enddo

のように，対応する要素の位置がずれるので注意して下さい．
このように，配列計算式を使うとプログラムがシンプルになります．逆に，単一変数の計算式と配列
計算式との区別がなくなるので，両者が入り交じるとプログラムがわかりにくくなり，エラーが見つけ
にくくなるという欠点もあります．
なお，サブルーチンの引数配列を使って配列計算をするときには注意が必要です．例えば，

subroutine subr(a,b)
implicit none
real a(*),b(*)
a = b**2 ! これはエラーになる

のようなプログラムでは，配列計算式はエラーになります．なぜなら，“ * ”を入れて配列宣言をした場
合は，要素数が不定だからです．これに対し，

subroutine subr(a,b,n)
implicit none
real a(n),b(n)
integer n
a = b**2 ! この場合は OK

のように，整合配列を使うと，配列計算が可能になります．

6.2 部分配列
配列名だけを使った配列計算式では全ての要素について計算をしますが，常に全要素の計算が必要と
は限りません．そこで，配列の一部を取り出した配列，“部分配列”を指定することができます．部分配
列は “ : ” (コロン)を使って要素番号の範囲を指定します．例えば，1次元配列 aに対して，

a(n1:n2)

と書けば，a(n1)～a(n2)という n2-n1+1個の要素から構成された 1次元配列として配列計算式に使う
ことができます．また，2次元配列 bに対して，

b(n11:n12,n21:n22)

と書けば，b(n11,n21)～b(n12,n21)～b(n11,n22)～b(n12,n22)という (n12-n11+1)×(n22-n21+1)の
2次元配列として配列計算式に使うことができます．また，ある次元の要素番号を固定して，

b(n,n21:n22)

と書けば，b(n,n21)～b(n,n22)という n22-n21+1個の要素から構成された 1次元配列として配列計算
式に使うことができます．
逆に，要素番号に “ : ”だけを記述すると，「その次元の全要素」という意味になります．例えば，

b(:,n21:n22)

と書けば，第 1次元は全要素で，第 2次元が n21～n22の範囲の要素から構成された 2次元配列として
配列計算式に使うことができます．このため，全ての要素番号を “ : ”にすると，配列全体を代表するこ
とになります．そこで，全配列要素を使った配列計算式を書くときにも “: ”を使って配列であることを
明示することができます．例えば，6.1節の 2次元配列計算式，

a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))

は，次のように書くことができます．

a(:,:) = sqrt(b(:,:)*sin(c(:,:))/(3.2*c(:,:) + 1.5e-3))

この方が，単一変数の計算式と区別できるので良いと思います．本書でもこれ以降はこの書式を使って
全要素の配列計算式を記述します．
要素範囲の指定に “ :整数値 ”を追加して，do文のような増分値を指定することもできます．例えば，

1次元配列 aに対して，

a(n1:n2:n3)

と書けば，a(n1)，a(n1+n3)，a(n1+2*n3)，...という要素からなる 1次元配列として配列計算式に使
うことができます．この場合，終了要素は n2とは限らないので注意して下さい．例えば，次のように書
くことができます．

real a(10),b(5)
.......

b(1:5) = a(1:10:2)

この結果は，b(1)=a(1)，b(2)=a(3)，b(3)=a(5)，b(4)=a(7)，b(5)=a(9)という 5個の代入文と等価
になります．増分値は負数を与えることも可能なので，次のように逆向きに代入させることも可能です．

real a(10),c(10)
.......

c(1:10) = a(10:1:-1)

この結果は，c(1)=a(10)，c(2)=a(9)，...，c(10)=a(1)という 10個の代入文と等価になります．
配列計算式で部分配列を使うときは，その部分配列がその他の配列と同型であれば良いので，宣言文
での次元や要素数が一致している必要はありません．例えば 1次元配列なら，

real a(10),b(5)
.......

a(3:5) = b(1:3)**2

のように書くことができます．また，2次元配列の部分配列を使って，

real a(10,10),b(5,4),c(10)
.......

a(3:5,6:8) = b(1:3,1:3)**2

などと書くことができます．この配列計算式を do文で表せば，

do j = 6, 8
do i = 3, 5
a(i,j) = b(i-2,j-5)**2

enddo
enddo

となります．ある次元の要素番号を固定した 2次元配列は 1次元配列として扱えるので，2次元配列と
1次元配列が混在した計算も可能です．例えば，

real a(10,10),c(10)
.......

c(3:8) = a(3,3:8)**2

と書くことができます．この配列計算式を do文で表せば，

do j = 3, 8
c(j) = a(3,j)**2

enddo

となります．
なお，部分配列を使った配列計算式で，左辺と右辺に同じ配列を使う場合には，単純に do文で置き換
えた動作と結果が異なる場合があるので注意が必要です．例えば，

real a(10)
do i = 1, 10

a(i) = i
enddo
a(2:10) = a(1:9) ! この配列計算式は単純な do文で置き換えられない
print *,(a(i),i=2,10)

というプログラムを考えます．この中の配列計算式 a(2:10)=a(1:9)を do文にすると，

do i = 1, 9
a(i+1) = a(i)

enddo

と置き換えられそうですが，実際にやってみると print文の出力結果は異なります．配列計算式の場合
は 1から 9までの異なる数字が出力されるのに対し，この doループで置き換えると全て 1が出力され
ます．
これは，配列計算式が doループのように 1要素ずつ計算と代入を繰り返すのではなく，まず右辺の配
列要素計算を全て行って結果を補助配列に保存しておき，その後で保存した補助配列から左辺の配列へ
の代入を行うからです．このため，配列代入機能を利用すると，次のように配列要素の順番を逆転させ
るのも簡単です．

a(1:10) = a(10:1:-1)

–93–



これと同じ動作をするプログラムを 1回の doループで書くのは難しいです．この代入機能は，配列計算
式を使う利点の一つです35．
部分配列は要素が等間隔に並んでいますが，整数の 1次元配列を使って要素指定をすれば，任意の順
番で指定した配列を作ることができます．例えば，次のような指定ができます．

real a(10)
integer m(3),i
do i = 1, 10

a(i) = i
enddo
m(1) = 7; m(2) = 2; m(3) = 5
print *,a(m)

最後の出力結果は，7.0 2.0 5.0となります．すなわち，a(m)とは，(a(m(1)), a(m(2)) a(m(3)))

という要素数 3の配列のことです．すなわち，整数配列で要素指定をすると，要素指定配列の要素数を
持つ配列になります．a(m(2:3))のように部分配列を使って要素指定をしたり，6.4節の配列構成子を
使って要素指定をすることもできます．

6.3 where 文による条件分岐
配列演算は便利ですが，全ての要素について同じ形の計算をするので，要素の条件に応じて異なる処
理をさせることができません．そこで，配列要素の条件に応じて動作を分岐させるための where文が用
意されています．where文は，以下のような形式です．

where (配列条件) 配列計算式

これは，“配列条件 ”に合った要素に対してのみ，“配列計算式 ”を実行するという意味です．例えば，

real a(10,10),b(10,10)
.......

where (a(:,:) > 0) b(:,:) = a(:,:)**2

のように書きます．この where文の部分を do文で表せば，次のようになります．

do j = 1, 10
do i = 1, 10
if (a(i,j) > 0) b(i,j) = a(i,j)**2

enddo
enddo

where文はブロック構文にすることもできます．ブロック where文では，次のように else where文
を付加して，条件に一致しない場合の記述をすることもできます．

where (配列条件 1)
配列計算式 1
.......

else where (配列条件 2)
配列計算式 2
.......

else where
配列計算式 0
.......

endwhere

この場合，配列条件 1を満足する要素は配列計算式 1のブロックを実行し，配列条件 1を満足せず，配列

35ただし，配列要素数が非常に大きくて，コンピュータのメモリ利用可能限界に近くなると，補助配列の生成のおかげでメ
モリオーバーになる可能性もあります．

条件 2を満足する要素は配列計算式 2のブロックを実行し，．．．という具合に続いて，全ての条件を満足
しない要素は配列計算式 0のブロックを実行するという動作になります．中間の else whereに関する
ブロックは省略可能です．if文と違って，“then ”が不要なこと，“全ての条件以外 ”を表す文が “else
where”であること，ブロックの最後に “endwhere”文を付加すること，などに注意して下さい．例えば，

real a(10,10),b(10,10)
.......

where (3.0*b(:,:) > 0)
a(:,:) = b(:,:)**2

else where
a(:,:) = b(:,:)**3

endwhere

のように書くことができます．このブロック where文を do文で表せば，次のようになります．

do j = 1, 10
do i = 1, 10
if (3.0*b(i,j) > 0) then

a(i,j) = b(i,j)**2
else
a(i,j) = b(i,j)**3

endif
enddo

enddo

whereブロック中で使用する配列は，全て同型でなければなりません．このため，whereブロックの
中に，単一変数の計算式や，次元や要素数の異なる配列計算式を入れることはできません．

6.4 配列構成子
配列演算を使えば，do文を使わなくても配列要素の計算ができますが，これまでの書式では配列要素
ごとに異なる定数を代入することはできません．そこで，1次元配列だけですが，定数や変数などを並
べて明示的に配列要素を与える配列構成子が用意されています．n個の要素からなる 1次元の配列構成
子は以下の形式です．

(/数値 1,数値 2,...,数値 n/)

この時，n個の数値は全て同じ型の数値型でなければなりません．例えば，

real a(5)
a(:) = (/1.0,2.0,3.0,4.0,5.0/)

と書けば，右辺は実数型の 1次元配列構成子であり，この配列計算式は，a(1)=1.0，a(2)=2.0，a(3)=3.0，
a(4)=4.0，a(5)=5.0という 5個の代入文を実行したことに相当します．
数値の位置には変数や計算式を書くこともできます．例えば，

real a(5),b
b = 2.5
a(:) = (/b,2*b,3*b,4*b,5*b/)

と書けば，この配列計算式は，a(1)=b，a(2)=2*b，a(3)=3*b，a(4)=4*b，a(5)=5*bという 5個の代
入文を実行したことに相当します．
部分配列で指定することも可能です．例えば，

real a(5),c(5)
a(:) = (/1.0,2.0,3.0,4.0,5.0/)
c(:) = (/a(3:5),a(1:2)/)

と書けば，この配列計算式は，c(1)=a(3)，c(2)=a(4)，c(3)=a(5)，c(4)=a(1)，c(5)=a(2)という 5
個の代入文を実行したことに相当します．
この配列構成子は，数値が全て定数であれば，次のように拡張宣言文の初期値代入に使うこともでき
ます．

real :: a(5)=(/1.0,2.0,3.0,4.0,5.0/)

しかし，配列要素が多いと，数値を並べて配列構成子を記述するのに手間がかかります．そこで，4.2節
で説明した do型並びを使って数値の設定をすることができます．do型並びには，増分値なしと増分値
付きの 2種類があります．増分値を省略した時の増分値は 1です．

(データ 1,データ 2,...,整数型変数=初期値,終了値) ! 増分値なし
(データ 1,データ 2,...,整数型変数=初期値,終了値,増分値) ! 増分値付き

この形式を配列構成子中に記述すると，まず整数型変数 (カウンタ変数)を初期値にして，データ 1，デー
タ 2，...と並べ，次に，カウンタ変数に増分値を加えて，再度，データ 1，データ 2，...と並べ，カ
ウンタ変数が終了値より大きくなるまでくり返して得られる数値を並べたことに相当します．
例えば，

(/(n, n=1,5)/)

という do型並びを使った配列構成子は，整数型の配列構成子，

(/1,2,3,4,5/)

と同じです．また，

(/(n, n**2, n=1,5)/)

という do型並びを使った配列構成子は，整数型の配列構成子，

(/1,1,2,4,3,9,4,16,5,25/)

と同じです．増分値を指定して，

(/(n, n**3, n=1,8,2)/)

という do型並びを使った配列構成子は，整数型の配列構成子，

(/1,1,3,27,5,125,7,343/)

と同じです．
複数の do型並びを並べたり，通常の数値と混在させることも可能です．例えば，

(/(real(n), n=0,2),1.5,1.8,(real(n), n=3,5)/)

という実数型の配列構成子は，

(/0.0,1.0,2.0,1.5,1.8,3.0,4.0,5.0/)

と同じです．この時，型変換関数の real(n)を単に nと書くとエラーになるので注意して下さい．これ
は，nが整数型なので，nだけを書くと一つの配列構成子の中に実数型と整数型が混在するからです．

do型並びは多重にすることも可能です．例えば，

(/((i+j,i=1,3),j=1,4)/)

という do型並びを使った整数型の配列構成子は，

(/2,3,4,3,4,5,4,5,6,5,6,7/)

と同じです．多重にした場合，カウンタ変数は内側が先に進みます．また，do型並びを doブロックの
中に入れる時には，カウンタ変数が do文のカウンタ変数とも重複しないようにしなければなりません．
本節最初の例は，次のように do型並びを使って書くことができます．

real a(5)
a(:) = (/(n, n=1,5)/)

ここで，右辺は整数型の配列で，左辺は実数型の配列ですが，この場合は全要素に対して整数型から実
数型への変換が行われて代入されるのでエラーにはなりません．なお，do型並びは拡張宣言文での初期
値代入に使うこともできますが，カウンタ変数はその宣言文より前に宣言されている必要があります．
配列構成子は 1次元配列しか用意されていません．このため，2次元以上の配列計算で使う時は，1次
元ごとの部分配列に代入するか，配列の形状を変換する組み込み関数 reshapeを使います．reshapeに
ついては 6.5節で説明します．

6.5 配列に関する組み込み関数
配列を引数にする組み込み関数も色々用意されています．まず，配列要素を利用した計算に関する関
数を表 6.1に示します．

表 6.1 配列要素の計算に関する組み込み関数
配列関数 引数の型 関数値の型 関数の意味
sum(a) 配列* 配列要素の型 全ての配列要素の和

product(a) 配列* 配列要素の型 全ての配列要素の積
minval(a) 配列* 配列要素の型 全ての配列要素の最小値
maxval(a) 配列* 配列要素の型 全ての配列要素の最大値

dot_product(a,b) 2個の 1次元配列36 配列要素の型 2個の 1次元配列の内積
matmul(a,b) 2個の 2次元配列37 2次元配列 2個の 2次元配列の行列積

例えば，

real a(5)
a(:) = (/(i, i=1,5)/)
print *,sum(a) ! 配列 aの全要素の合計

と書けば，配列 aの全要素の合計 (15.0)が出力されます．部分配列を利用すれば，指定した範囲の配列
要素だけを使った計算も可能です．例えば，

36aの要素数と bの要素数は等しくなければなりません．
37行列積の計算上，aの第 2要素数と bの第 1要素数は等しくなければなりません．なお，aか bのどちらかは 1次元配列

でも計算できます．この場合，結果は 1次元配列になります．

real a(10)
integer n
a(:) = (/(i, i=1,10)/)
n = 5
print *,product(a(1:n)) ! 部分配列の要素 a(1)～a(n)の積

と書けば，a(1)から a(n)までの積 (120.0)が出力されます．
表 6.1で引数の型に “ * ”の付いた関数を使う場合，引数配列の後に整数型の引数を追加すると，その
整数値が指定する次元に関してのみ計算をすることができます．例えば，

real b(3,4),x,y(4),z(3)
integer i,j
do j = 1, 4

b(:,j) = (/(sin(0.5*(i+j)), i=1,3)/)
enddo
x = maxval(b) ! 配列 bの全要素の最大値
y(:) = maxval(b,1) ! 配列 bの第 1次元方向の要素の最大値
z(:) = maxval(b,2) ! 配列 bの第 2次元方向の要素の最大値

と書けば，xには 2次元配列 bの全要素の最大値が代入されますが，1次元配列 yには，bの第 2次元を
固定して第 1次元方向に要素を比較したときの最大値がそれぞれ代入されます．また，1次元配列 zに
は，bの第 1次元を固定して第 2次元方向に要素を比較したときの最大値がそれぞれ代入されます．す
なわち，次元を指定すると，その次元を抜いた要素数の配列が結果になります．
次に，配列情報に関する関数を表 6.2に示します．

表 6.2 配列情報に関する組み込み関数
配列関数 引数の型 関数値の型 関数の意味
minloc(a) 配列 整数型配列 全ての配列要素の最小値の位置
maxloc(a) 配列 整数型配列 全ての配列要素の最大値の位置
lbound(a) 配列* 整数型配列 配列の各次元の最小要素番号リスト
ubound(a) 配列* 整数型配列 配列の各次元の最大要素番号リスト
shape(a) 配列 整数型配列 配列の各次元の要素数リスト
size(a) 配列* 整数 配列の全要素数

reshape(a,s) 配列と整数型配列 aの型の配列 配列 aを配列 sで指定された型の
配列に変換する

allocated(a) 配列 論理 配列 aが割り付けられていれば真，
さもなくば偽

　表 6.2で，引数の型に “ * ”の付いた関数は，引数配列の後に次元を示す整数型の引数を追加して，そ
の次元に関する値を取り出すこともできます．例えば，

real array(3,4) ! 3× 4の配列
integer m,n1,n2
m = size(array) ! 全要素数 3× 4=12
n1 = size(array,1) ! 第 1次元要素数 3
n2 = size(array,2) ! 第 2次元要素数 4

と書けば，mには，arrayの全要素数である 12が代入され，n1には arrayの第 1次元の要素数である 3
が，n2には arrayの第 2次元の要素数である 4が代入されます．
また，表 6.2で関数値の型が “整数型配列”になっている関数は，引数に与えた配列の次元を要素数に

持つ 1次元整数型配列が戻り値です．その際，その 1次元配列の各要素には，引数配列の各次元の情報
が代入されています．例えば，下限と上限を取得するための関数，lboundと uboundは以下のように使
います．

real bb(-10:10,4) ! 2次元配列
integer blow(2),bupp(2) ! 2次元なので要素数 2の配列
blow = lbound(bb)
bupp = ubound(bb)
print *,blow,bupp

ここで，2次元配列 bbは，第 1次元の下限が−10，上限が 10，第 2次元の下限が 1，上限が 4ですから，
配列構成子で書けば，blowは (/-10,1/)，buppは (/10,4/)になります．このとき，ubound(bb,2)の
ように，次元に関する引数を加えると，その結果は 1個の整数 (この例では 4)になります．
reshapeは，配列の形状を変換する関数です．そもそも，どんな次元の配列もメモリ上では 1次元的
に並んでいるので，2次元配列 a(3,4)と１次元配列 b(12)のように，全要素数が等しい配列は同じよう
に取り扱えるはずです．しかし，配列演算は次元および各次元の要素数が等しい同型の配列間でしか許
可されていません．そこで，a(3,4)と b(12)の間で演算をするには，形式上，同型になるような変換
が必要です．これを実行するのが関数 reshapeです．reshapeには，最初の引数に変換したい配列を与
え，2番目の引数に変換後の形状を表す 1次元配列を与えます．ここで形状を表す 1次元配列とは，変
換後の配列が，k1×k2×...×knの n次元配列であれば，

(/k1,k2,...,kn/)

で与えられる要素数 nの 1次元整数型配列のことです．例えば，6.4節最後の例文は，

real a(3,4)
integer i,j
a(:,:) = reshape((/((sin(0.5*(i+j)), i=1,3), j=1,4)/), (/3,4/))

と書くことができます．reshapeの第 1引数は do型定数なので 1次元配列ですが，これを第 2引数の
(/3,4/)で 3×4の 2次元配列に変換するよう指定しているわけです．
この配列の形状を表す 1次元配列は，関数 shapeを使って取得することができます．例えば，上記の

a(3,4)という宣言をした配列に対し，shape(a)は，1次元整数型配列，(/3,4/)になります．そこで，
上記の配列計算式は次のように書くことができます．

a(:,:) = reshape((/((sin(0.5*(i+j)),i=1,3),j=1,4)/),shape(a))

しかし，これではまだ do型変数における iや jの範囲指定が定数のままです．そこで，ここは先ほど説
明した関数 sizeで書き直すと良いでしょう．

a(:,:) = reshape((/((sin(0.5*(i+j)),i=1,size(a,1)),j=1,size(a,2))/),shape(a))

これなら，宣言文の要素数を変更するときに，この配列計算式を変更する必要はありません．

6.6 配列の動的割り付け
配列を宣言するときには，整数定数を与えて要素数を明示しなければなりません．これは，その情報に
基づいて，プログラムの動作開始時に計算で使用するメモリ量を確定するためです．しかし，最近のコ
ンピュータではプログラムの実行中に必要になったメモリを確保したり，不要になったメモリを解放す
るという動作が頻繁に行われています．このような，プログラムの実行中に必要に応じてメモリを確保
することを “動的割り付け ”といいます．Fortranも Fortran90から動的割り付けが可能になり，プログ
ラムの動作に必要な要素数を持つ配列を実行中に確保できるようになりました．動的割り付けを使うと，

メモリ効率の良い汎用性のあるプログラムを作成することができます．ここでその方法を紹介します．
まず，サブルーチンの中で必要に応じた要素数の配列を確保する簡単な方法として，サブルーチンの
引数を使った配列宣言があります．例えば，次のサブルーチンで real宣言されている 2次元配列 bが
これに相当します．

subroutine memory1(a,m,n)
implicit none
real a(m,n),b(m,n),x ! 引数 mと nを使って宣言する
integer m,n,i,j
x = 10.0
b(:,:) = x*reshape((/((i*j,i=1,m),j=1,n)/),shape(b))
a(:,:) = b(:,:)**3

end subroutine memory1

配列 bは，このサブルーチンがコールされた時点で，引数 mと nの値を使って確保されます．このとき，
b(m,n)のような単純な宣言だけではなく，b(0:2*m,n-1)のような，下限指定や計算式を使った宣言も
可能です．ここで，配列 aも引数を使って宣言されていますが，aは引数に含まれているので 3.5節で説
明した整合配列です．整合配列は，引数配列の取り扱いを決める仕組みにすぎず，配列自体はサブルー
チン側にありません．
これに対し，引数に含まれていない配列 bはローカル配列であり，宣言されたサブルーチンの一時メ
モリ領域に所属します．3.8.1節で説明したように，一時メモリ領域に所属するローカル変数は，サブ
ルーチンを呼び出した時点で生成されます．引数の値はサブルーチン開始時に確定しているので，これ
を使って配列に必要なメモリを確保することが可能なのです．ただし，これはあくまでもサブルーチン
で宣言する配列についてのみ使える機能であり，メインプログラムの配列やモジュール中のグローバル
配列としては使えません．
そこで，メモリの動的割り付けを明示的に行う仕組みが用意されています．Fortranで動的割り付け
を行うには，二つの手続きが必要です．一つは，割り付けたメモリを配列として使用するための配列名
の宣言です．これは，割り付けたメモリの先頭アドレスを保持するためのメモリを用意することだと考
えられます．もう一つは，その配列名に実際のメモリを割り付ける動作です．前者は宣言文なので非実
行文，後者はプログラム実行中に割り付けるので実行文です．
メモリを割り付ける予定の配列名は，allocatable属性を付けて型宣言をします38．このとき，配列
の次元情報をコロン “: ”を使って示す必要があります．配列の次元はコロンの数で決まり，実行時に変
更することはできません．例えば，

real, allocatable :: ab(:),z2(:,:)
integer, allocatable, dimension(:,:) :: km1, km2

のように宣言します．1行目のように，名前の後ろに次元情報を付加しても良いし，2行目のように，
dimension属性を使うことも可能です38．この例の場合，abは 1次元実数型配列，z2は 2次元実数型配
列，km1と km2は 2次元整数型配列として割り付けることができます．
配列の割り付けは allocate文で行います．allocate文は，次のように配列の要素指定をサブルーチ
ンのようにかっこで囲んで記述します．

allocate ( ab(100), z2(0:m,-m:m), km1(k-1,2*k) )

かっこ内は，配列宣言と同じ形式です．abや km1のように，要素数に整数だけを与えると，その次元の
下限は 1になるし，z2のように “:”を使って下限指定をすることも可能です．配列の数値型は型宣言の
際に確定しているので，この例のように，一つの allocate文で異なる数値型の配列を同時に割り付け
ることも可能です．allocate文は実行文なので，z2や km1のように整数型変数や整数式の計算結果を

38配列に属性を付加する方法と dimension属性については 3.8.1節参照．

使って割り付けることも可能です．一旦割り付けられた配列は，通常の宣言文で宣言した配列と同様に
使用することができます．
ただし，むやみに割り付けを繰り返すと，コンピュータで利用できる容量を超える “メモリオーバー”
になる可能性があります．そこで，不用になった配列のメモリ領域は deallocate文で解放することが
できます．deallocate文は，次のように配列名だけをかっこで囲んで指定します．

deallocate ( ab, km1 )

deallocate文も異なる数値型の配列を同時に解放することが可能です．例えば，先ほどのサブルーチ
ン memory1は，以下のように書き換えることができます．

subroutine memory1(a,m,n)
implicit none
real a(m,n),x
real, allocatable :: b(:,:) ! 割り付け用 2次元実数型配列
integer m,n,i,j
allocate (b(m,n)) ! m×nの 2次元配列を割り付け
x = 10.0
b(:,:) = x*reshape((/((i*j,i=1,m),j=1,n)/),shape(b))
a(:,:) = b(:,:)**3
deallocate (b) ! メモリの解放

end subroutine memory1

なお，引数変数を使って宣言したローカル配列は一時メモリ領域に所属しますが，allocate文で割り
付けた配列は固定メモリ領域に所属します39．サブルーチンにおける一時メモリ領域は固定メモリ領域
より使用可能容量が少ない可能性があるので，要素数の大きな配列を割り付けるときは allocate文を
使う方が良いでしょう．
allocate文を使った動的割り付けは，メインプログラムでも利用できるし，モジュールの中で配列名
宣言をして，グローバル配列として使用することも可能です．ただし，allocate文で割り付けられた配
列を，解放しないで再度 allocate文で割り付けようとすると実行時エラーになります．また，割り付け
る前にその配列を使用しようとしても実行時エラーになります．そこで，配列が割り付けられているか
否かを確認する関数が用意されています．これが表 6.2の最後にある関数 allocatedです．allocated

は論理型の値を返す関数で，allocatable属性を持つ配列名を引数に与えると，その配列がすでに割り
付けられていれば “真 ”，割り付けられていなければ “偽 ”を返します．論理型の関数は，そのまま if

文の条件として使うことができるので，例えば次のように利用することができます．

real, allocatable :: a(:)
integer i,n
.......

if (allocated(a)) then
a(:) = (/(100*i,i=1,n)/) ! 代入１

else
allocate ( a(n) )
a(:) = (/(200*i,i=1,n)/) ! 代入２

endif

この場合，配列aがすでに割り付けられていれば “代入 1”を実行し，割り付けられていなければ，allocate
文で割り付けてから “代入 2 ”を実行します．
メモリの動的割り付けや解放は，それほど時間がかかる処理ではありません．計算の途中で一時的に
利用する配列は，必要なときに割り付けて，不要になったら解放するようにしておけば，メモリの利用
効率が高くて汎用性のあるプログラムになります．

39ただし，サブルーチン終了後も数値を保持するために配列を固定メモリ領域に所属させる場合には，配列名の宣言に save
属性を加えて，配列名も固定メモリ領域に所属させなければなりません．

付録A gfortranを用いたコンパイルから実行までの手順
gfortranは Fortran95 の文法で書いたプログラムをコンパイルできる代表的なフリーの Fortranです．
多くの Linuxディストリビューションに付属しているし，Windows版やMac OS版も配布されている
ので，お手持ちのパソコンにインストールすれば，手軽に Fortranを利用することができます．ここで
は gfortranを使用したコンパイルから実行までの手順について述べます．
まず，Fortranプログラムを作るときには，ファイル名の最後に “ .f90 ”を付けます．つまり，

　　　　文字列.f90

という名前にします．このファイル名の最後に付加した “ドット (.)+文字”の部分を拡張子と呼びます．
作成したプログラムファイルを計算機で実行させるには “コンパイル ”と “リンク ”という二つの過程が
必要です．
コンパイルというのは，Fortranや C言語など，人間が理解できる言語で書かれたプログラムをコン
ピュータが理解できる機械語に翻訳することです．コンパイルするアプリケーションを “コンパイラ ”
といいます．フリー Fortranコンパイラの代表が gfortranです．

gfortranでプログラムをコンパイルをするときは，gfortranコマンドを使って，

　　　 gfortran プログラムファイル名

と入力します．例えば，test1.f90というファイル名のプログラムをコンパイルするときは

　　　 gfortran test1.f90

と入力します．コンパイル時に文法エラーなどが見つかると，エラーメッセージを出力して終了します．
gfortranコマンドは，コンパイルが成功すると，引き続きリンクを行います．プログラムはコンパイ
ルしただけでは実行できません．コンパイラはプログラミング言語を機械語に翻訳するだけであり，複
数のルーチンを結合してOS上で起動可能な形式にする作業までは行わないからです．この結合処理が
リンクです．単にプログラム中のルーチンのリンクだけではありません．入出力文の readや write，組
み込み関数の sinや log等は，標準ライブラリルーチンとして用意されているので，必要に応じてこれ
らのリンクも行います40．
リンクにも成功すると，最後に “ a.out ”という名前のファイルが作成されます．これがOS上で直接
実行させることができる機械語で書かれたファイルで，これを実行形式ファイルといいます．もしリン
クに失敗した場合は，エラーメッセージを出力して終了します．この時 a.outは作成されません．
実行形式ファイルは，一般のコマンドのようにファイル名を入力することでプログラムを実行するこ
とができます．ただし，次のように頭に “ ./ ”を付ける必要があります41．

　　　./a.out

プログラムが正常に動作すれば，それに書かれた一連の計算を行って終了します．計算の途中に print

文の記述があれば，結果を画面に出力するし，write文の記述があれば，指定したファイルに書き出し
ます．また，標準入力の read文の記述があれば，その時点でプログラムが一時停止して入力待ちの状
態になります．この状態で必要な数値をキーボードから入力すれば計算は再開します．
以上が gfortranを用いた，最も単純なコンパイルからプログラム実行までの手順ですが，gfortranコ
マンドにファイル名を与えただけの命令では，どんなプログラムをコンパイルしても a.outという同じ
名前の実行形式ファイルになってしまうので不便です．また，Fortranプログラムは計算速度が重要な

40コンパイラである gfortran自体はリンクする機能を持っていませんが，gfortran内部から OSに付属しているリンク用ア
プリケーション (リンカ)を呼び出すことができるので，リンクも実行することが可能なのです．

41“./”は “現在のディレクトリにあるファイル”という意味です．このあたりの詳細は，Linuxなどの解説書を見て下さい．

ので，最適化をしてできるだけ高速に処理する方が良いでしょう．さらに 1.2.2節で述べたように数値
計算をするときは倍精度実数を使うべきなので，自動倍精度化オプションも必要です．
このため，gfortranでプログラムをコンパイル・リンクする時は，以下のようなオプションを付ける
ことをお勧めします．

　　　 gfortran -O -fdefault-real-8 test1.f90 -o xtest1

-Oが最適化のオプション，-fdefault-real-8が自動倍精度化のオプションです．また，-o のオプショ
ンの次の文字列 (ここでは xtest1)は実行形式ファイル名指定です．大文字の-Oと小文字の-oを間違わ
ないようにして下さい．
この場合，コンパイルとリンクが正常に終了すると，-oオプションで指定した “ xtest1 ”という名の
実行形式ファイルが作成されます．このため，プログラムを実行するには，

　　　./xtest1

と入力することになります．
また，付録Gで紹介している数値計算ライブラリやグラフィックライブラリを使用したプログラムを
コンパイル・リンクする場合は，それらのライブラリファイルをオプションとして付加する必要があり
ます．例えば，FFTWを含んだプログラムの場合は，

　　　 gfortran -O -fdefault-real-8 test1.f90 -o xtest1 -L/usr/local/lib -lfftw3

のように入力します．ここで，-Lオプションが指定している “ /usr/local/lib ”は，FFTWライブラ
リ (libfftw3.a)の入ったディレクトリを示すオプションなので，使っている環境に応じて書き換える必要
があります．

付録B エラー・バグへの対処法
プログラムを開発する際の最大の問題は “エラー (Error) ”の発生です．コンパイル・リンク・実行と
いう一連の過程の中でそれぞれエラーが発生する可能性があります．エラーはプログラム開発にとって
付き物であり，避けては通れません．エラーの出ないように慎重にプログラムを書くことはもちろんで
すが，出たら出たでそれらに如何に素早く対処できるかが腕の見せ所です．ここではエラーをまとめて，
それぞれの対処法を示します．なお，具体的なエラーメッセージは gfortranの出力を利用していますが，
他のコンパイラでも形式が違うだけで出力情報の内容はあまり変わらないと思います．

(1) コンパイルエラー
コンパイルとは，作成したプログラムを文法に従って機械語に翻訳することです．このため，“コン
パイルエラー ”とは文法的な間違いのことです．プログラムに文法的間違いがあれば，コンパイラがエ
ラーメッセージを出力して終了します42．コンパイラのエラーメッセージ形式はコンパイラによって違
いますが，gfortranでは次のような形式で出力されます．この例は，test1.f90というプログラムファイ
ルをコンパイルしたときのエラー出力です．

test1.f90:15.7:
do i = 1, 100

1
Error: Symbol ’i’ at (1) has no IMPLICIT type

以下に，この出力の読み方を示します．

(1) 1行目はエラーのあるプログラムファイル名 (この例では，test1.f90)とエラーの位置
この例では，エラーの位置が 15.7となっていますが，15はプログラムの行番号，7は左から
数えた文字の位置を示す数値です．

(2) 2行目はエラーのあるプログラム文のコピー
(3) 3行目は “ 1 ”を上向き矢印のように使って，2行目の文のエラーが起きている位置を指定
(4) 4行目は 3行目が示した部分の具体的なエラーの説明
このメッセージを頼りにエラーを見つけてプログラムを修正します．

この例の 4行目のメッセージは「3行目の 1が示している変数 iは，暗黙の型がない」という意味です．
これは，変数 iが宣言されていないことが原因です．
コンパイルエラーは，エラーメッセージに従って修正すればいいので，それほどの手間ではありませ
ん．ただし，ある文に文法エラーがあると，その文が存在しない状態で引き続きコンパイルを行うため，
その波及効果で下方のプログラムがエラーになることがあります．例えば宣言文に誤りがあると，その
宣言文で宣言している変数がコンパイラにとっては宣言されていないことになり，その変数を使ってい
る文が全てエラーになります．エラーメッセージが多い時は，一度に全部チェックしないで，ある程度
修正したら再度コンパイルして，エラーが出るかどうかを確認した方が良いと思います．

42コンパイラが出すメッセージには “Warning ”(警告)と “Error ”(エラー)があります．Warningは「このままでも実行で
きるけど，ひょっとすると予期しない結果が起こる可能性があります」という意味なので，必ずしも修正する必要はありませ
ん．このため，メッセージがWarningだけのときは強制終了せずにリンクの作業に移ります．しかし，文法的には間違いが
なくても，作成者の意図とは違ったプログラムになっている可能性があるので，念のためにプログラムを確認した方が良いで
しょう．

(2) リンクエラー
リンクとは，別々に翻訳されたルーチンを結合して実行形式ファイルにする作業のことです．よって，
リンクエラーは用意されていないサブルーチンや関数を使ったときに起こります．例えば，

program test1
implicit none
real x,y
x = 5
y = 100
call subr(x,y) ! subrが用意されていないとリンクエラーになる
print *,x,y

end program test1

のようなプログラムを作って，これだけを単独でコンパイル・リンクすると

/tmp/ccy3lcdo.o: In function ‘MAIN__’:
test1.f90:(.text+0x6d): undefined reference to ‘subr_’
collect2: ld returned 1 exit status

のようなメッセージが出力されます．以下に，この出力の読み方を示します．

(1) 1行目はエラーが起こっているルーチンの表示
この例では “ MAIN ”，すなわちメインプログラム中で起こっていることが示されています．

(2) 2行目はエラーの原因
このメッセージを頼りにエラーを見つけてプログラムを修正します．

この例の 2行目のメッセージには「subrという名への参照が定義されていない」とあります．これは，メ
インプログラム中でコールしている subrという名のサブルーチンが用意されていないのが原因です43．
標準の組み込み関数が用意されていないことはまずないので，リンクエラーのほとんどはプログラム
中で呼び出したサブルーチン名や関数名の書き間違いが原因です．リンクエラーのメッセージにはコン
パイルエラーのようなエラー行の表示はありませんが，エラーメッセージの中に見つからなかったサブ
ルーチンや関数の名前が表示されているので，エディタの検索機能を使えば，見つけるのはさほど困難
ではありません．

(3) 実行時エラー
一番厄介なのが実行時のエラーです (こういうプログラムは “バグ (虫) ”があるといいます)．なぜな
ら，コンパイルエラーやリンクエラーのようなエラーの場所を特定する情報が出力されないので，どこで
エラーが起こっているのかを探す作業が大変だからです．実行時エラーを探すには，プログラムを詳細
にチェックするしかありません (この作業を「バグを取る」という意味で “デバッグする ”といいます)．
それでも，答えが合っているか否かは別として，プログラムがなんとか終了すれば良いのですが，バ
グによっては実行した後でうんともすんともいわなくなってしまう場合があります．これは無限ループ
に入ってしまったか，暴走したかのどちらかだと考えられます．通常，このような状態で実行を強制的
に終了させるには，コントロールキーを押しながらＣキーを押します．まず間違いなく止まります．
もう一つ，以下のようなメッセージを出して強制的に終了する場合があります．

Segmentation fault

これはプログラムではなくOSが出力しているメッセージで，主として実行中のプログラムがそれ以
外の実行中のプログラムのメモリ領域を破壊しようとした時に起こります．例えば，次のように配列宣

43メッセージ中では MAIN__とか，subr_のように名前の後ろにアンダースコアが付いていますが，これはコンパイラが自動
的に付加するものなので無視して下さい．リンクは gfortranがするのではなく，ldというリンカがするのですが，リンカは
OSによって異なるので，gfortranを使っていてもこの形式でエラーメッセージが出るとは限りません．

言の範囲外の要素に値を代入しようとすると起こることがあります44．

real a1(10)
integer i
do i = 1, 100000

a1(i) = i
enddo

この例では，10個しか用意されていない配列に，100000番目まで値を代入しようとしているのですか
ら，どこのメモリに値を代入しようとしているのか不明なのが問題なのです．
また，サブルーチンの引数に型の合っていない数値を与えたり，戻り値を与える引数に定数を与える，
などの不注意なサブルーチンコールでも Segmentation faultが起こる可能性があります．Segmentation
faultは，実行時に問題が出た段階で起こるので，適当に print文を入れて，どこまでが出力されて，ど
こからが出力されないかを確認することでエラーの起こっている場所を特定することができます．
しかし，このような実行時エラーはコンピュータの動作が正常でないという形で表面化するので，ま
だましです．原因を特定するのに時間がかかるかもしれませんが，修正しなければ動かないのだから実
害はありません．実を言うと，一番やっかいなエラーとは，ちゃんと実行してはいるのだけど，計算結
果が何となくおかしい，というものです．
筆者は何年も Fortranのプログラムを書いてきましたが，実行時のエラーほど見つけにくいものはあ
りません．それでも見つかれば良い方で，場合によってはバグの存在を知らずに結果を出し，学会直前
に気がついた，なんてこともありました．大規模なシミュレーションプログラムを作るとき，長いプロ
グラムを書くのに時間がかかるのはもちろんですが，取りあえず動いたプログラムの計算結果が正しい
かどうかを確認する作業にもかなりの時間が必要です．本書で紹介した “エラーの出にくいプログラム
の書き方 ”は，この確認作業にかかる時間を少しでも減らすためのものです．しかし，最後に必要なの
は “地道な努力と忍耐 ”なのです．

44実際にはこの程度の簡単なプログラムでは起こるとは限らないのですが，だからといってこんなプログラムを書いてはい
けません．

付録C 自動倍精度化オプション
コンピュータシミュレーションでは大量の数値を使って複雑な計算を繰り返し，必要に応じてそれら
を集積する作業を行います．このとき問題となるのが，1.4.3節で述べた “桁落ち ”です．桁落ちを減ら
すための工夫をいくつか紹介しましたが，究極的には演算精度を上げるしかありません．

Fortranで取り扱える実数には，単精度実数型と倍精度実数型がありますが，数値計算をするには倍
精度実数型を使うべきです．倍精度実数計算は単精度実数計算と比べてそれほど実行時間はかかりませ
ん．これは最近の CPUが倍精度実数計算用のハードウェアを装備しているからです．

Fortranの仕様では，デフォルトの実数型が単精度なので，real宣言をした実数は単精度実数型です．
このため，基本仕様のままで倍精度実数計算をするには変数宣言をする時に real*8や real(8)などの
8byte指定が必要です45．
例えば，デフォルト仕様のままでも，

real*8 x,a1(10)

と宣言すれば，変数 xや配列 a1は倍精度になります．しかし，宣言文を変更するだけでは不完全です．
1.0や 1.23e5などの実定数もデフォルト仕様では単精度なので，倍精度実定数に書き換える必要があ
ります．
そもそも倍精度で計算するのを基本にするのに，常に倍精度を意識した書式を利用しなければならな
いのは，面倒なだけでなく間違いに気がつかないまま精度の悪い計算をしている可能性もあります．そ
こで最近の Fortranコンパイラのほとんどが自動倍精度化オプションを装備していることを幸いに，本
書では単精度実数型と倍精度実数型を使い分ける文法の説明は省略しました45．自動倍精度化オプショ
ンを付けてコンパイルすれば，realだけの宣言文で倍精度実数変数を宣言できるし，1.0や 1.23e5の
ような定数はそのままで倍精度実定数を意味します．
表 C.1に筆者が使っている Fortranコンパイラの自動倍精度化オプションを示します．この他のコン
パイラについてはそれぞれのマニュアルをチェックして下さい．大抵は用意されていると思います．

表 C.1 コンパイラに応じた自動倍精度化オプション
コンピュータ コンパイラ ベンダー 自動倍精度化オプション
パソコン gfortran GNU -fdefault-real-8

パソコン ifort Intel -r8

パソコン f95 Absoft -N113

スパコン sxf90 NEC -Wf,-A dbl4

自動倍精度化は，コンパイルする時にオプションを加えれば良いだけなので，プログラムを変更する
ことから考えれば楽なものです．ぜひ利用して下さい．

45精度指定の詳細は付録 Dを参照．

付録D 数値型の精度指定
1.2.2節で，基本的な数値型には整数型と実数型があり，実数型には単精度実数型と倍精度実数型があ
るという話をしました．Fortranのデフォルト実数型は単精度ですが，コンパイラの自動倍精度化機能
(付録 C)を使えばデフォルトを倍精度実数型に変更できるので，本書では 2種類の実数型を区別せず，
単に “実数型”と表現しています．しかし，大量の実数型データをバイナリ形式で保存するときには，保
存容量を圧縮するために精度を落とすことが考えられますし，“ 4倍精度実数 ”という倍精度よりも有効
桁数の多い実数型を使って，より高精度の計算をすることも可能です46．ここでは，定数や変数の精度
を変更する書式について説明します．
変数の精度を指定するには，以下の 3種類の方法があります．ここでは，基本的な宣言文だけを紹介
しますが，3.8.1節で述べた属性や数値代入を含む拡張宣言も可能です．

型指定*精度数 変数 1，変数 2，...
型指定 (精度数) 変数 1，変数 2，...
型指定 (kind=精度数) 変数 1，変数 2，...

ここで，“精度数 ”のところには，数値型の byte数を整定数で指定します．実数型の場合，単精度なら
4，倍精度なら 8，4倍精度なら 16です．例えば，単精度実数型変数を宣言するには，

real*4 xs1,ys1,as1(100)
real(4) xs2,ys2,as2(100)
real(kind=4) xs3,ys3,ds3(100,100)

などと書きます．また，通常の整数型は 4byteですが，最近は “倍精度整数 ”という 8byteの整数型 (使
用可能な範囲は−263～263 − 1)を使うことができるので，これを利用するときには，

integer*8 n81,m81,k81(100)
integer(8) n82,m82,k82(100)
integer(kind=8) n83,m83,ka83(100,100)

などと書きます．
3種類の書式の中でどれが良いかは一概には言えません．一番目の “*”を使った書式は，古い Fortran
から使われているので，コンパイラが古い場合や，昔のプログラムを継承する可能性がある場合には覚
えておいた方が良いと思います．しかし，これからプログラムを書き始める場合には，二番目か，精度
数の意味を明記した三番目の方が良いと思います．これらは，次のように parameter変数 (3.8.2節)を
使って精度数を指定することができます．

integer, parameter :: kp=16
real(kp) xq2,yq2,aq2(100)
real(kind=kp) xq3,yq3,dq3(100,100)

これに対し，“ * ”を使った書式では必ず数字を使って指定しなければなりません．parameter変数で精
度指定をしておけば，計算機環境に応じて精度を変更する必要があるときに便利です．
ただし，高精度の計算をする場合は，定数も高精度にしなければなりません．例えば，“1.23”と書け
ばデフォルトの実数型になるので，本書の暗黙指定では倍精度実数型です．よって，このまま 4倍精度
の計算に使うと精度が落ちてしまいます．A×10Bで表される実数を 4倍精度で表現するときは “AqB ”
と書きます．このため，“ 1.23 ”は “ 1.23q0 ”と書かなければなりません47．
しかし，parameter変数で変数の精度を指定しているときに，eや qのような文字を使って定数の精度

46ただし，4倍精度実数型は，必ず実装しなければならない規格ではないようで，コンパイラによっては使えない場合があ
ります．gfortranでも，バージョン 4.8では使えますが，4.2では使えませんでした．

47ちなみに，デフォルト実数が単精度の時，倍精度実定数は “ AdB ”と書きます．例えば，“ 1.2 ”は “ 1.2d0 ”です．また，
実数化の組み込み関数 realの結果が単精度実数型になるので，倍精度実数化が必要なときには関数 dbleを使います．

を指定していると，精度変更の時に定数の変更が別途必要になります．そこで，数値の後にアンダースコ
ア “_”と精度数を付けて定数の精度を指定することができます．例えば，倍精度の “1.23”は “1.23_8”
と書け，4倍精度の “1.23q0 ”は，“1.23_16 ”と書くことができます．この精度指定には parameter変
数を使うことができるので，次のように書くことができます．

integer, parameter :: kp=16
real(kp) xx,yy
xx = 1.23_kp
yy = 1.2345e-15_kp*xx**3

この例の yyに代入している実数の指数指定は eを使っていますが，精度数が 16なので，4倍精度定数
になります．

付録E Big EndianとLittle Endian

大量のデータを精度を落とさずにファイルに保存するときは，4.5節で説明した書式なし write文を
使ってバイナリ形式で保存する方が良いと思います．これは計算機のメモリに保存された内部形式その
ままでファイルに保存するため，書式付き write文を使ったテキスト形式出力よりも出力データ量が小
さくなるからです．
しかし，バイナリ形式で保存するときは注意が必要です．テキスト形式で保存したファイルは，我々で
も直接読むことができるのですから，どんなコンピュータでも同じように読むことができます．このた
め，大型計算機で計算した結果をテキスト形式で出力しておけば，そのままパソコンのプログラムから読
み込んで処理をすることができます．しかしバイナリ形式で出力すると必ずしもうまくいきません．バ
イナリ形式は計算機が直接計算を実行するためのものですから，計算機のアーキテクチャによって違っ
ている可能性があるからです．しかし，現在はバイナリ形式も標準化されていて，ほとんどの計算機で
整数型も実数型も統一した形式が採用されています．このため，バイナリ形式を使って大型計算機が出
力したデータをパソコンのプログラムで処理することができます．
ただやはりいくつかの制約は残っています．その一つがEndian問題です．コンピュータが扱う数値の
内部形式は整数型で 4byte，倍精度実数型で 8byteですが，この byteの並びが上位の桁からのコンピュー
タと，下位の桁からのコンピュータがあるのです．前者を Big Endian，後者を Little Endianといいま
す．イメージ的に表現すれば，1234という数字を保存すると，Big Endianでは 01234の順番で保存す
るのに対し，Little Endianでは 43210の順番で保存するようなものです．これらの違いはCPUのアー
キテクチャによるものなのでユーザーが変更することはできません．
最近のほとんどのパソコンは，CPUに Intelかその互換チップを使っていますが，これらは Little

Endianです．これに対し，以前のMacintoshで使われていたMotorolaの CPUや NECのスパコンは
Big Endianです．このため，スパコンの Fortranで計算して書式なし write文で出力すると，そのまま
ではパソコンの Fortranで読むことができません．
しかし，要はデータ入出力時の形式の問題だけなので，最近の Fortranコンパイラの多くはコンパイ
ラオプションや環境変数などで Bigから Littleへ，あるいはその逆への変換をしてから入力または出力
をすることができるようになっています．例えば，NEC SXの場合にはプログラム実行前に環境変数
F_UFMTENDIANを指定することで変換できます．

setenv F_UFMTENDIAN 10,20 # C シェル系

環境変数で指定するのは書式なし write文の装置番号です．この例の場合，write(10)と write(20)

が Little Endianに変換して出力されます．この変換を使うと，メモリ上でBigな形式を Littleにして出
力するのですから若干計算時間にロスが出ます．しかし，byte並びを逆にするだけですからテキスト変
換よりは楽です．
これに対し，入力側で変換するやり方もあります．例えば，gfortranの場合には -fconvert=big-endian

というオプションを加えれば，書式なし read文で入力するときに Big Endianから Little Endianに変
換してくれます．最近の Fortranコンパイラならば大抵変換機能が付いているようなので，マニュアル
で確認すれば良いでしょう．
筆者はどちらかといえばスパコン側で変換する方が良いと思います．スパコンで行う計算は元々時間
がかかるものであり，それに比べれば出力時のデータ変換による時間増加はわずかです．これに対して
パソコン側で変換するのは時間もかかるし，変換するかどうかを問題に応じて使い分けるのも面倒です．
ただし，スパコンでの計算には時間制限があるので，データ変換のために CPU時間を削るのはもっ
たいないとも考えられます．スパコンはデータ変換に時間がかかるという話もありますので，どちらが
良いかはスパコン管理者に確認した方が良いでしょう．

付録F ASCIIコード
文字列の比較などに使われる半角英数字の文字コード (ASCIIコード)を表 F.1に示します．

表 F.1 ASCIIコード表
0 1 2 3 4 5 6 7 8 9 A B C D E F

2 t ! " # $ % & ’ ( ) * + , - . /

3 0 1 2 3 4 5 6 7 8 9 : ; < = > ?

4 @ A B C D E F G H I J K L M N O

5 P Q R S T U V W X Y Z [ \ ] ^ _

6 ‘ a b c d e f g h i j k l m n o

7 p q r s t u v w x y z { | } ~

半角英文字は 1byte(8bit)文字であり，左端の数字が上位 4bit，上端の数字 (16進数)が下位 4bitです．
例えば，“ T ”の文字は 16進数の 54，“ < ”は 3C，スペース “ t ”は 20です．1F以下と 7Fは制御コード
に割り当てられていて文字としては定義されていません．
なお，5Cのバックスラッシュ “\”は，日本語フォントでは “ Y ”に割り当てられています．

付録G 数値計算を補助するフリーのサブルーチンライブラリ
複雑な計算アルゴリズムを使ったプログラムを一から全て書くのは結構大変です．汎用性のあるアル
ゴリズムを使ったプログラムは，インターネット上にサブルーチンライブラリとして公開されているも
のがあるので，これらを利用してプログラムを作るのもスマートな方法です．ここでは筆者が使ったこ
とのあるライブラリの中からお勧めをいくつか紹介します．

G.1 高速フーリエ変換
偏微分方程式の解法やデータ解析で良く使う高速フーリエ変換 (Fast Fourier Transform, FFT)は，ア
ルゴリズムが複雑で，効率の良いプログラムを自作するのは大変です．筆者は最も簡単なレベルのプロ
グラムを作ったことはありますが，最近はネットで公開されているものを使っています．お勧めなのは，
FFTW(Fastest Fourier Transform in the West)です．

　　　　 http://www.fftw.org/

このライブラリには，Fortranから使用できるサブルーチンが入っています．

G.2 多倍長計算
Fortranでの倍精度実数は 8byteで，有効数字は 15桁程度です．4倍精度実数は 16byteで，有効数字
が 33桁程度に増加しますが，それ以上は無理だし，コンパイラによっては 4倍精度実数が使えないこと
もあります．筆者は多項式の解を計算するときに高精度計算が必要になったことがあります．数十次の
多項式を計算するときは，最大次数の項と最小次数の項との数値の差が非常に大きくて，桁落ちが深刻
な問題になるからです．
このときに便利なのが多倍長計算ライブラリです．これは，任意の精度で実数計算をすることができ
るものです．通常の計算より処理に時間がかかりますが，精度優先のときにはやむを得ません．ネット
上には何種類か公開されていますが，筆者が使って便利だったのは，FMLIBです．

　　　　 http://myweb.lmu.edu/dmsmith/FMLIB.html

これはFortranのモジュールになっていて，四則演算は「+ - * /」の記号を使った数式で書けるし，sin
や cosなどの関数も充実しています．本書では説明していない構造体の知識が少しだけ必要ですが，変
数宣言程度なのでそれほど難しくはありません．筆者は，倍精度計算用に作った多項式の解を計算する
プログラムを，それほど大きな変更をすることなく多倍長で動作させることができました．

G.3 長周期乱数発生
計算機シミュレーションには，時間発展の方程式を解くことで未来を予測する決定論的手法と，サイコ
ロを振って確率的に将来の可能性を探る確率論的手法があります．後者の確率論的手法で，サイコロに
相当する働きをするのが乱数です．Fortranには，乱数発生用の組み込みサブルーチン random_number

が用意されているので，

　　　　 call random_number(x)

のように書けば，乱数 xを得ることができます (問題 (3–10)参照)．しかし，乱数は周期長が問題で，周
期の短い乱数を使うと良いシミュレーション結果が得られません．random_numberでは周期長が不足す
るときにお勧めなのは，周期が非常に長い乱数として有名な，Mersenne Twisterです．これを考案した
松本眞さんが以下のサイトでプログラムを公開されています．

　　　　 http://www.math.sci.hiroshima-u.ac.jp/~m-mat/MT/mt.html

松本さんが公開されているのはＣ言語で書かれたものですが，リンク集の中にFortran用の公開先が入っ
ています．

G.4 数値計算ライブラリ
連立方程式の解法や数値積分など，汎用性のある数値計算ライブラリがあれば，数値計算のプログラ
ムを作るのが楽になります．ただ，残念ながら筆者はフリーの良いライブラリを知りません．ネットで
公開されていて使えそうなものに，GSL（GNU Scientific Library）があります．

　　　　 http://www.gnu.org/software/gsl/

ただし，基本的にＣ言語用なので，Fortranで使うには若干テクニックがいるようです．また，ちょっと
使ってみましたが，必ずしも速くありません．ベッセル関数などの特殊関数は，その高速バージョンを
公開しているサイトからダウンロードした方が良いと思います．筆者は，数値計算の教科書を読んで原
理から自作するか，参考文献 [5,6]などに掲載されているプログラムをコピーして使うことが多いです．
なお，これから数値計算を勉強しようという場合には，筆者が書いた「Fortranハンドブック」もお勧
めです [7]．

G.5 グラフィックライブラリ
計算機シミュレーションは物理的に起こっている現象を模擬するものですが，出力される数値だけで
現象を読み取るのは容易ではありません．そこで必要不可欠なのが可視化ソフトです．フリーの可視化
ソフトとして有名なものに gnuplotがあります．

　　　　 http://www.gnuplot.info/

プログラムの結果をファイルに保存しておけば，gnuplotからそれを読み込んで，様々なグラフに加工
して表示することができます．
最近のパソコンは基本操作が全てビジュアルであり，グラフィック出力機能はOSに組み込まれてい
ます．Fortranにはグラフィック出力に関する文法はありませんが，他の言語で書かれたグラフィック
ライブラリとリンクすれば，Fortranプログラムからサブルーチンをコールする形式で図を描くことが
可能です．その一つに，筆者が作った Framesがあります．

　　　　 http://www.pp.teen.setsunan.ac.jp/frames/

Framesは，Fortranで使うためのグラフィックツールとして開発してきたもので，本書で説明した知識
だけで簡単に 2次元や 3次元のグラフを描くことができます．また，描画した図形をイメージやアニメー
ション形式で出力してプレゼンテーションに使うことも可能です．良かったら試してみて下さい．

